

40 programmes en Basic pour la maison

MSX EN FAMILLE



Jean-François Séhan

MSX EN FAMILLE

Autres ouvrages relatifs au MSX

- 102 programmes pour MSX - Jacques Deconchat

A paraître:

- Basic MSX: 1 - Méthodes pratiques - Jacques Boisgontier
- Basic MSX: 2 - Programmes - Jacques Boisgontier
- Super jeux pour MSX - Jean-François Séhan
- Clefs pour MSX - Rémy Pineau
- Assembleur MSX - Alain Pinaud
- La découverte du MSX - Daniel-Jean David

Ouvrages d'initiation

- Visa pour l'informatique — Jean-Michel Jégo
- Visa pour le Basic — Jean-Michel Jégo
- Mon ordinateur — Jean-Claude Barbance
- Le logotron informatique — Jean-Pierre Petit
- L'ordinateur apprivoisé — François Picard et Danielle Shaw
- Minitel mode d'emploi — Dominique Schraen et Sibylle Rochas
- Macintosh le magnifique — Merl K. Miller et Mary A. Myers, adapté par Raymond Borraz

La loi du 11 mars 1957 n'autorisant, aux termes des alinéas 2 et 3 de l'article 41, d'une part, que les «copies ou reproductions strictement réservées à l'usage privé du copiste et non destinées à une utilisation collective», et, d'autre part, que les analyses et les courtes citations dans un but d'exemple et d'illustration, «toute représentation ou reproduction intégrale, ou partielle, faite sans le consentement de l'auteur ou de ses ayants droit ou ayants cause, est illicite» (alinéa 1^{er} de l'article 40).

Cette représentation ou reproduction, par quelque procédé que ce soit, constituerait donc une contrefaçon sanctionnée par les articles 425 et suivants du Code Pénal.

**40 programmes en Basic
pour la maison**

MSX EN FAMILLE

Jean-François SEHAN



**EDITIONS DU PSI
1984**

Du même auteur aux éditions du P.S.I.

- Etudes pour ZX-81 tome I
 - Etudes pour ZX-81 tome II
 - Suites pour PC-1500
 - Le TI-99/4A à l'affiche
 - Clefs pour ZX-81
 - Le VIC à l'affiche
 - Clefs pour ZX SPECTRUM
 - L'ORIC à l'affiche
 - Le SPECTRUM à l'affiche
 - Le COMMODORE 64 à l'affiche
 - Le MO5 et TO7/70 à l'affiche
 - MO5 et TO7/70 en famille
 - Le COMMODORE 64 en famille
-

AVERTISSEMENT

Si vous voulez utiliser correctement les programmes de ce livre, lisez attentivement les conseils suivants :

- Les programmes qui créent des fichiers peuvent aussi bien utiliser un magnétophone à cassettes qu'un lecteur de disquettes et ceci sans aucune modification.
- Seul le programme "Disposition des meubles" nécessite l'emploi d'un joystick.
- Pour gagner en rapidité d'exécution certaines instructions NEXT ne sont pas suivies d'une variable (gain de 10 à 20 %).
- Si vous n'obtenez pas les mêmes résultats que sur les exemples, vérifiez bien toutes les lignes et surtout celles contenant des DATAs.
- Dès que les programmes sont entrés dans le MSX, sauvez-les sur votre magnétophone. Pour les programmes très longs, il est vivement conseillé d'effectuer une première sauvegarde au milieu de leur frappe. N'attendez pas qu'ils soient entièrement entrés et testés... une coupure de courant est si vite arrivée !

Ce livre a été réalisé sur Yashica MSX, distribué en France par Segimex et sur Sanyo MSX, distribué en France par Sanyo France.

SOMMAIRE

Thème 1 - LES FINANCES

Compte bancaire	10
Placement chez l'écureuil	17
Budget familial	23
Coût annuel de votre voiture	28
Calcul d'impôts	32

Thème 2 - PÉDAGOGIE

Musique	40
Géographie	47
Apprendre à taper au clavier	54
Alphabet	58
Apprendre à compter	62
Tables de multiplication	67
Les quatre opérations	71
Questions-réponses	75
Conjugaisons	82
Suites logiques	88

Thème 3 - CUISINE ET TRICOT

Fichier de recettes	94
Recettes de cuisine	100
Menus aléatoires	106
Gestion d'un congélateur	111
Tricot	117

Thème 4 - LES JEUX NATIONAUX

Le Tiercé : pronostics	122
Le Loto : grilles aléatoires	127
Le Loto : statistiques	131

Thème 5 - LE TEMPS

Arbre généalogique	138
Astrologie	146
Astrologie chinoise	151
Espérance de vie	157

Biorythme	163
Jour de la semaine	167
Calendrier	172
Horloge	178
Agenda électronique	182

Thème 6 - VOTRE FORME


Poids idéal	190
Courbes poids/taille	194

Thème 7 - BRICOLAGE

Papier peint	200
Disposition des meubles	204

Thème 8 - UTILITAIRES

Cryptogrammes	212
Gestion des QSO	216
Gestion d'une cave	222
Répertoire téléphonique	228



THEME 1 LES FINANCES


COMPTE BANCAIRE

Plus qu'une simple gestion de compte bancaire, ce programme vous donne l'occasion de connaître vos dépenses en nourriture, loyer, vêtements, essence, etc., mais aussi le total des dépenses et des rentrées de l'année.

Si vous l'utilisez pour la première fois, le MSX vous demande le montant du compte actuellement ou au 1^{er} janvier si vous créez une nouvelle année. Puis un menu de six options s'affiche.

- 1 – **SAISIE** : vous devez entrer tous les mouvements en donnant des valeurs négatives aux sommes retirées (chèques, virements, carte de crédit). La date, le numéro et la description sont à titre indicatif, mais il est très intéressant de donner des noms par groupe pour faciliter les recherches. Par exemple si vous donnez les descriptions "ESSENCE VOITURE", "VIDANGE VOITURE", "CREDIT VOITURE", vous pourrez obtenir le total des dépenses pour votre voiture mais aussi les dépenses uniquement d'essence ou de vidanges ainsi que la somme de tous les crédits (pas uniquement celui de la voiture).
- 2 – **MODIFICATION** : si vous avez fait une erreur dans la saisie, il est toujours possible de la modifier en utilisant cette option. Vous devez donner le numéro de l'opération (si vous l'avez oublié utilisez les options 4 et 5 puis le numéro de l'élément à modifier). Quand vous ne désirez plus effectuer de modification, tapez "Ø" à la question "Numéro de l'opération".
- 3 – **SUPPRESSION** : supprime une des opérations déjà entrées. Le programme demande confirmation avant la suppression (répondre par "O" ou "N"). Comme pour la modification il faut répondre "Ø" pour revenir au menu.
- 4 – **RECHERCHE** : le programme recherche dans tout le fichier le nom, le numéro ou la date que vous lui donnez. Par exemple le nom "CREDIT" affiche et calcule le total des sommes concernant des crédits (crédits logement, voiture, etc.).
- 5 – **LISTING** : cette opération permet de lister sur l'écran la totalité des mouvements effectués sur le compte. Le listing s'arrête automatiquement en bas de page. Vous devez appuyer sur n'importe quelle touche pour continuer.
- 6 – **SAUVEGARDE** : dès que vous n'avez plus besoin du programme, cette option sauve sur la cassette ou sur la disquette la totalité du fichier.

COMPTE BANCAIRE



```

10 REM COMPTE BANCAIRE OU POSTAL
20 REM -----
30 CLEAR 5000
40 DIM D$(4,500)
50 L$(1)="MONTANT":L$(2)="DATE":L$(3)="NUMERO OU REPERE":L$(
4)="DESCRIPTION"
60 L(2)=8:L(3)=7:L(4)=7
70 SCREEN 0:COLOR 4,1,1:KEY OFF
80 INPUT"PREMIERE UTILISATION (O/N)";R$
90 IF R$<>"O" THEN GOTO 120
100 CLS:INPUT"MONTANT INITIAL DU COMPTE ";M
110 GOTO 300
120 CLS:PRINT"METTEZ LE MAGNETO K7 EN LECTURE"
130 OPEN "COMPTE" FOR INPUT AS #1
140 INPUT#1,N,C,D,M
150 FOR I=1 TO N
160 INPUT#1,D$(1,I),D$(2,I),D$(3,I),D$(4,I)
170 NEXT I
180 CLOSE
280 REM -----
290 REM MENU
300 CLS
310 PRINT"SOLDE : ";M+C-D:PRINT"CREDIT: ";C:PRINT"DEBIT : ";D
320 PRINT"NB D'OPERATIONS: ";N:PRINT:PRINT
330 PRINT"1 SAISIE"
340 PRINT"2 MODIFICATION"
350 PRINT"3 SUPPRESSION"
360 PRINT"4 RECHERCHE"
370 PRINT"5 LISTING"
380 PRINT"6 SAUVEGARDE"
390 PRINT:PRINT"VOTRE CHOIX ?";
400 R$=INPUT$(1):R=VAL(R$)
410 IF R<1 OR R>6 THEN 400
420 ON R GOTO 500,1000,1200,700,1500,1700
480 REM -----
490 REM SAISIE D'UNE OPERATION
500 CLS:PRINT"SAISIE D'UNE OPERATION":PRINT
510 IF N=500 THEN PRINT"FICHIER PLEIN";:R$=INPUT$(1):GOTO 30
0
520 N=N+1
530 FOR I=1 TO 4
540 PRINT L$(I);:INPUT D$(I,N)
550 IF I=1 AND VAL(D$(1,N))=0 THEN N=N-1:GOTO 300
560 IF I=2 AND LEN(D$(2,N))<>8 THEN 540
570 NEXT I
580 IF VAL(D$(1,N))<0 THEN D=D+ABS(VAL(D$(1,N))) ELSE C=C+VA
L(D$(1,N))
590 GOTO 500
680 REM -----
690 REM RECHERCHE
700 CLS:F=0:T=0
710 PRINT"RECHERCHE PAR : "
720 FOR I=1 TO 4:PRINT I:L$(I):NEXT
730 PRINT:PRINT"QUEL TYPE DE DONNEE ?":PRINT
740 R$="":R$=INPUT$(1):R=VAL(R$)
750 IF R<1 OR R>4 THEN 300
760 D$="":PRINT L$(R);:INPUT D$:PRINT
770 FOR I=1 TO N
780 IF INSTR(D$(R,I),D$)=0 THEN 840
790 F=1:PRINT USING"###";I;
800 FOR J=1 TO 4
810 IF J=1 THEN PRINT USING"#####.## ";VAL(D$(J,I)); ELSE P
RINT LEFT$(D$(J,I)+"",L(J));" ";

```

```

820 NEXT J:PRINT
830 T=T+VAL(D$(1,I))
840 NEXT I:PRINT"-----"
850 IF F=0 THEN PRINT"JE N'AI RIEN TROUVE" ELSE PRINT USING"
#####.##";T
860 R$=INPUT$(1):GOTO 700
980 REM -----
990 REM MODIFICATION
1000 CLS:PRINT"MODIFICATION":PRINT
1010 NU=0:INPUT"NUMERO DE L'OPERATION";NU
1020 IF NU<1 OR NU>N THEN 300
1030 FOR I=1 TO 4
1040 PRINT I;L$(I);": ";D$(I,NU)
1050 NEXT I
1060 R=0:PRINT:INPUT"MODIFICATION DU";R
1070 IF R<1 OR R>4 THEN 300
1080 IF R=1 THEN A=VAL(D$(1,NU))
1090 PRINT L$(R);:INPUT D$(R,NU)
1100 IF R=1 AND VAL(D$(1,NU))=0 THEN 1090
1110 IF R<>1 THEN 1000
1120 IF A<0 THEN D=D+A ELSE C=C-A
1130 IF VAL(D$(1,NU))<0 THEN D=D-VAL(D$(1,NU)) ELSE C=C+VAL(
D$(1,NU))
1140 GOTO 1000
1180 REM -----
1190 REM SUPPRESSION
1200 CLS:PRINT"SUPPRESSION":PRINT
1210 NU=0:INPUT"NUMERO DE L'OPERATION";NU
1220 IF NU<1 OR NU>N THEN 300
1230 PRINT
1240 FOR I=1 TO 4
1250 PRINT I;L$(I);": ";D$(I,NU)
1260 NEXT I:PRINT
1270 R$="":INPUT"VOUS ETES SUR DE SUPPRIMER CETTE OP.";R$
1280 IF R$<>"0" THEN 1200
1290 A=VAL(D$(1,NU))
1300 IF A<0 THEN D=D+A ELSE C=C-A
1310 N=N-1
1320 FOR I=NU TO N
1330 FOR J=1 TO 4
1340 D$(J,I)=D$(J,I+1)
1350 NEXT J,I
1360 GOTO 1200
1480 REM -----
1490 REM LISTING
1500 CLS
1510 FOR I=1 TO N
1520 PRINT USING"###";I;
1530 FOR J=1 TO 4
1540 IF J=1 THEN PRINT USING"#####.## ";VAL(D$(J,I)); ELSE
PRINT LEFT$(D$(J,I)+",L(J));" ";
1550 NEXT J:PRINT
1560 IF I/10=INT(I/10) THEN PRINT:PRINT" APPUYEZ SUR UNE TOU
CHE POUR LA SUITE";:R$=INPUT$(1):CLS
1570 NEXT I
1580 PRINT:PRINT" APPUYEZ SUR UNE TOUCHE POUR LE MENU";
1590 R$=INPUT$(1):GOTO 300
1680 REM -----
1690 REM SAUVEGARDE
1700 CLS:PRINT"APPUYEZ SUR RETURN QUAND LE MAGNETO K7 EST P
RET POUR L'ENREGISTREMENT"
1710 LINE INPUT R$
1720 OPEN "COMPTE" FOR OUTPUT AS #1

```

```
1730 PRINT#1,N,C,D,M
1740 FOR I=1 TO N
1750 FOR J=1 TO 4
1760 PRINT#1,D$(J,I)
1770 NEXT J,I
1780 CLOSE
1790 CLS:PRINT"A LA PROCHAINE OPERATION !"
1800 END
```

RECHERCHE PAR :

- 1 MONTANT
- 2 DATE
- 3 NUMERO OU REPERE
- 4 DESCRIPTION

QUEL TYPE DE DONNEE ?

DESCRIPTION? VOITURE

1	-187.00	08/01/84	123098	ESSENCE
2	-435.67	15/01/84	123112	REVISIO
3	-225.00	05/02/84	123125	ESSENCE
4	-218.00	10/02/84	123129	VIDANGE

= -1065.67

LE PROGRAMME

Lignes 30 à 180 : initialisation

On place dans le tableau L\$() les libellés des quatre données nécessaires à l'exploitation du fichier.

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 300 à 420 : menu

Ces lignes affichent le solde du compte, le total des crédits et des débits, puis ventilent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 500 à 590 : saisie d'une opération

On saisit par une boucle FOR/NEXT les quatre données du fichier et l'on met à jour le total crédit ou débit selon le cas (ligne 580).

Si aucun montant n'a été entré, on revient au menu (ligne 550).

Lignes 700 à 860 : recherche

Le programme demande d'abord sur quel type de donnée on désire effectuer la recherche, puis la donnée elle-même (variable D\$).

Les lignes 770 à 840 parcourent par une boucle FOR/NEXT l'intégralité du fichier, et affichent les enregistrements où l'on a trouvé la chaîne D\$. La variable T totalise les montants.

Lignes 1000 à 1140 : modification

Ces lignes modifient l'enregistrement choisi par l'utilisateur (variable NU). Si c'est le montant qui doit être changé, les lignes 1120 et 1130 mettent à jour le total des crédits et des débits.

Lignes 1200 à 1360 : suppression

On supprime du fichier l'enregistrement choisi par l'utilisateur (variable NU). Après l'avoir affiché, le programme demande confirmation. Les lignes 1320 à 1350 décalent le tableau vers le bas.

Lignes 1500 à 1590 : listing

On affiche par une boucle FOR/NEXT l'intégralité du fichier. Le programme est stoppé par la ligne 1560 dès que l'écran est plein.

Lignes 1700 à 1800 : sauvegarde

Ces lignes sauvent sur cassette ou sur disquette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

A	ancienne valeur du montant pour la modification
C	total des crédits
D	total des débits
D\$	donnée à rechercher
D\$(,)	tableau des données du fichier
F	drapeau. si F=0, aucun enregistrement n'a été trouvé
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
L()	longueur maximum d'affichage de chaque donnée
L\$()	libellés des données
M	montant initial du compte
N	nombre d'opérations dans le fichier
NU	numéro d'enregistrement choisi par l'utilisateur
R\$	choix de l'utilisateur
R	valeur numérique de R\$
T	total des montants dans la recherche

PLACEMENT CHEZ L'ECUREUIL

Que rapportent les économies que vous avez placées à la Caisse d'Epargne ? Laissez le soin à votre MSX d'en effectuer les calculs.

Le programme demande chaque quinzaine le total des dépôts et des retraits que vous avez effectués sur votre livret. En effet, les intérêts de la Caisse d'Epargne sont calculés tous les quinze jours (mais déposés sur le compte seulement le 31 décembre). Tous les retraits comptent à partir du premier jour de la quinzaine, les dépôts le premier jour de la quinzaine suivante. Il est préférable de se méfier pour les chèques "Reçu sous réserve d'encaissement" car ils prennent plusieurs jours avant d'être réellement déposés sur le livret. Un chèque déposé le 14 est encaissé après le 15, il comptera donc seulement dans les intérêts du mois suivant.

En plus des dépôts et des retraits, le programme demande aussi le taux d'intérêt, celui-ci pouvant être modifié d'une quinzaine à l'autre (aux bons soins de monsieur le ministre des Finances).

Ces formalités étant remplies, le MSX affiche par quinzaine le capital, les dépôts, les retraits, le capital sur lequel sont calculés les intérêts et les intérêts eux-mêmes.


DATE :CAPIT:DEPOT:RETR.:I SUR:INTERET

15/ 1:	12980:	0:	1000:	12980:	35.15
31/ 1:	12980:	0:	0:	12980:	35.15
15/ 2:	13880:	1200:	300:	12980:	35.15
31/ 2:	13880:	0:	0:	13880:	37.59
15/ 3:	12880:	1000:	2000:	12880:	34.88
31/ 3:	12380:	0:	500:	12380:	33.53
15/ 4:	12380:	0:	0:	12380:	33.53
31/ 4:	9880:	0:	2500:	9880:	26.76
15/ 5:	8880:	500:	1500:	8880:	24.05
31/ 5:	8880:	0:	0:	8880:	24.05
15/ 6:	8880:	0:	0:	8880:	24.05
31/ 6:	11225:	2345:	0:	8880:	24.05
15/ 7:	9991:	0:	1234:	9991:	27.06
31/ 7:	9991:	0:	0:	9991:	27.06
15/ 8:	8791:	0:	1200:	8791:	23.81
31/ 8:	8591:	0:	200:	8591:	23.27
15/ 9:	8591:	0:	0:	8591:	23.27

TOTAL INTERETS: 492.42

FIN (0/N) ?

PLACEMENT CHEZ L'ÉCUREUIL



```

10 REM CHEZ L'ECUREUIL
20 REM -----
30 DIM D(24),R(24),M$(12),C(24),T(24)
50 FOR I=1 TO 12:READ M$(I):NEXT I
60 DATA JANVIER,FEVRIER,MARS,AVRIL,MAI,JUIN
70 DATA JUILLET,AOUT,SEPTEMBRE,OCTOBRE,NOVEMBRE,DECEMBRE
80 SCREEN 0
90 INPUT"PREMIERE UTILISATION (O/N)";R$
100 IF R$="O" THEN GOTO 200
110 CLS:PRINT"METTEZ LE MAGNETO K7 EN LECTURE"
120 OPEN "ECUREU" FOR INPUT AS #1
130 INPUT#1,M,C
140 FOR I=1 TO 24
150 INPUT#1,D(I),R(I),T(I)
160 NEXT I
170 CLOSE
180 REM -----
190 REM DONNEES DU MOIS
200 CLS
210 IF M<>0 THEN 230
220 INPUT"CAPITAL AU 1er JANVIER (AVEC INTERETSDE L'ANNEE PR
ECEDENTE) ";C
230 C(0)=C
240 FOR I=1 TO M
250 C(I)=C(I-1)-R(I)+D(I)
260 NEXT I
270 IF M=24 THEN 500
280 M=M+1
290 PRINT
300 PRINT"CAPITAL ACTUEL: ";C(M-1)
310 PRINT"DONNEZ LES MOUVEMENTS DU : "
320 IF M/2=INT(M/2) THEN PRINT"15 AU 31 "; ELSE PRINT"1er AU
14 ";
330 PRINT M$(INT(.5+M/2))
340 PRINT
350 INPUT"DEPOT ";D(M)
360 INPUT"RETRAIT ";R(M)
370 INPUT"TAUX ";T$:IF T$="" THEN T(M)=T(M-1) ELSE T(M)=VAL(
T$)/100
380 IF INT(C(M-1)+D(M)-R(M))<0 THEN PRINT"IMPOSSIBLE: CAPITA
L < 0":GOTO 290
390 PRINT
400 C(M)=C(M-1)+D(M)-R(M)
480 REM -----
490 REM RESULTATS
500 TT=0
510 CLS:PRINT:PRINT
520 FOR I=1 TO M
530 IF R(I)>D(I) THEN R=R(I)-D(I) ELSE R=0
540 CA=C(I-1)-R
550 T=(CA/24)*T(I)
560 TT=TT+T
570 IF I/2=INT(I/2) THEN PRINT"31/"; ELSE PRINT"15/";
580 PRINT USING"###";INT(.5+I/2);
590 PRINT USING"#####";C(I),D(I),R(I),CA;
600 PRINT USING"####.##";T;
610 NEXT I
620 PRINT"-----";
630 PRINT TAB(15);"TOTAL INTERETS: ";:PRINT USING"####.##";TT
;
640 LOCATE 0,0:PRINT"DATE :CAPIT:DEPOT:RETR.:I SUR:INTERET";
650 PRINT"-----";
660 LOCATE 0,24:PRINT"FIN (O/N) ?";

```

```
670 R$=INPUT$(1): IF R$="N" THEN 200
680 REM -----
690 REM SAUVEGARDE
700 CLS:PRINT"APPUYEZ SUR RETURN QUAND LE MAGNETO K7 EST PR
ET POUR L'ENREGISTREMENT"
710 LINE INPUT R$
720 OPEN "ECUREU" FOR OUTPUT AS #1
730 PRINT#1,M,C
740 FOR I=1 TO 24
750 PRINT#1,D(I),R(I),T(I)
760 NEXT I
770 CLOSE
780 CLS:PRINT"A LA PROCHAINE QUINZAINE!"
790 END
```

LE PROGRAMME

Lignes 30 à 170 : initialisation

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 200 à 400 : données du mois

S'il s'agit de la première utilisation ($M=0$), la ligne 220 saisit le capital au 1^{er} janvier. Les lignes 240 à 260 placent dans le tableau C() le capital du mois (capital du mois précédent - les retraits + les dépôts du mois).

Si le fichier est déjà plein ($M=24$), la ligne 270 renvoie aux résultats sans passer par la saisie.

Le programme demande les dépôts, les retraits et le taux d'intérêts de la quinzaine. Si vous ne donnez aucune valeur pour ce dernier, on prend automatiquement le taux du mois précédent (ligne 370).

Lignes 500 à 670 : résultats

Ces lignes affichent un tableau récapitulatif des dépôts et des retraits et calculent le capital et les intérêts. La ligne 630 donne le total des intérêts annuels.

La ligne 670 renvoie à la saisie si l'on désire entrer les données de la quinzaine suivante avant la sauvegarde.

Lignes 700 à 790 : sauvegarde

On sauve sur cassette ou sur disquette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

D()	liste des dépôts
C	capital au 1 ^{er} janvier
C()	capital par quinzaine
CA	capital pour le calcul des intérêts
I	indice de boucle FOR/NEXT
M	numéro de la quinzaine en cours
M\$()	noms des 12 mois
R()	liste des retraits
R\$	réponse de l'utilisateur
T	intérêts de la quinzaine
T()	liste des taux par quinzaine
TT	total des intérêts

BUDGET FAMILIAL


A partir du total de vos dépenses et des dépenses par poste que vous lui donnez, le MSX vous indiquera le pourcentage du budget que vous consacrez aux impôts, au loyer, à la voiture, aux vacances, aux vêtements, à la nourriture, aux fêtes, aux loisirs et aux assurances.

Au dernier montant entré, l'ordinateur dessine l'histogramme de votre budget familial en identifiant chaque rectangle. Les pourcentages sont rappelés à droite de l'écran.


Au cas où vous voudriez modifier une somme, vous devez taper sur RETURN. Le questionnaire qui apparaît n'est vide qu'en apparence car il contient toujours les données précédentes.


Il suffit de taper un nouveau montant pour les modifications éventuelles ou de passer au poste suivant en tapant simplement RETURN s'il n'y a rien à changer. Puis, comme dans le premier cas, l'ordinateur redessine l'histogramme corrigé.

C'est un moyen rapide d'avoir une vue d'ensemble sur vos dépenses et vous aurez peut-être des surprises.



BUDGET FAMILIAL





```

10 REM BUDGET FAMILIAL
20 REM -----
30 DATA IMPOTS, LOYER, VOITURE, VACANCES, VETEMENTS, NOURRITURE
40 DATA FETE, LOISIRS, ASSURANCES, DIVERS
50 FOR I=1 TO 10
60 READ N$(I)
70 NEXT I
80 OPEN "GRP:" FOR OUTPUT AS #1
180 REM -----
190 REM SAISIE DES DEPENSES
200 SCREEN 0:KEY OFF:COLOR 4,1,1
210 R$="":INPUT"TOTAL DES DEPENSES ";R$
220 IF VAL(R$)<>0 THEN TD=VAL(R$)
230 IF TD=0 THEN 200
240 MX=0:DT=0
250 PRINT:PRINT:PRINT
260 FOR I=1 TO 9
270 PRINT N$(I);
280 R$="":INPUT R$
290 IF VAL(R$)<>0 THEN D(I)=VAL(R$)
300 IF D(I)<MX THEN 320
310 MX=D(I)
320 DT=DT+D(I)
330 IF DT<=TD THEN 360
340 PRINT"TOTAL > AUX DEPENSES"
350 TD=DT
360 NEXT I
370 D(10)=TD-DT
380 IF D(10)>MX THEN MX=D(10)
390 FOR I=1 TO 10
400 P(I)=INT((D(I)/(MX/100)))*1.6
410 NEXT I
480 REM -----
490 REM HISTOGRAMME
500 SCREEN 2:DRAW "BM17,5":PRINT#1,"^";TAB(15)"DEPENSES"
510 LINE (19,7)-(19,170):LINE -(214,170)
520 FOR I=1 TO 10
530 COLOR 6
540 FOR J=170 TO 170-P(I) STEP -1
550 LINE (12+I*10,J)-(18+I*10,J)
560 NEXT J
570 COLOR 15
580 DRAW"BM120,"+STR$(8*I+16):PRINT#1,USING"##-";I;
590 PRINT#1,N$(I):DRAW"BM224,"+STR$(8*I+16):PRINT#1,USING":#
#";INT(D(I)/(TD/100))
600 DRAW"BM"+STR$(13+I*10)+",176":PRINT#1,MID$(STR$(I),2,1)
610 DRAW"BM"+STR$(13+I*10)+",184":PRINT#1,MID$(STR$(I)+",",3
)
620 NEXT I
630 A$=INPUT$(1)
640 GOTO 200

```

LE PROGRAMME

Lignes 30 à 80 : initialisation

On place dans le tableau **N\$()** les libellés des dépenses. Ces derniers se trouvant placés dans les **DATAs** des lignes 30 et 40, il est possible de les modifier pour d'autres dépenses si vous n'avez pas de voiture ou si vous ne partez jamais en vacances.

Le fichier ouvert à la ligne 80 permet d'afficher des caractères sur l'écran graphique.

Lignes 200 à 410 : saisie des dépenses

La variable **MAX** garde en mémoire la valeur de la plus grande dépense effectuée. Elle servira de référence dans l'histogramme.

La variable **DT** garde la somme totale des 9 dépenses. Les dépenses diverses prendront alors la différence entre le total des dépenses (ce qui est réellement sorti de votre portefeuille) et la somme des 9 autres dépenses.

Dans le cas où **DT** devient supérieur à **TD**, on modifie **TD** pour le réajuster ; les dépenses diverses deviennent obligatoirement nulles.

On calcule pour chaque dépense un pourcentage pour donner une idée de la valeur que l'on vient de rentrer au clavier.

Lignes 500 à 640 : histogramme

Les lignes 540 à 560 affichent l'histogramme. L'ordonnée des dépenses est affichée aux lignes 600 et 610.

Dès qu'une touche est enfoncée, on revient à la ligne 200 pour une nouvelle saisie ou pour la modification des dernières données.

LISTE DES VARIABLES :

A\$	dernière touche appuyée
D()	tableau des dépenses
DT	dépenses totales
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
MAX	valeur maximale dans l'histogramme
N\$()	libellé des dépenses
P()	pourcentage de chaque dépense
R\$	dernière dépense tapée au clavier
TD	total des dépenses

COUT ANNUEL DE VOTRE VOITURE

Que coûte votre voiture chaque année ? Quelles sommes avez-vous dépensées pour l'entretien et l'essence ? Quel est le prix de revient d'un petit kilomètre ? Autant de questions auxquelles votre MSX répondra avec plaisir.


Le programme demande tous les mois le nombre de kilomètres au compteur et les dépenses en essence, en entretien et en mécanique ainsi que les remboursements éventuels d'un crédit (entrer "0" si vous n'avez pas de crédit).

Si vous utilisez le programme pour la première fois (ou une autre année), il demande aussi le prix de l'assurance et de la vignette, et le nombre de kilomètres au compteur au 1^{er} janvier.

Avec ces quelques renseignements, le MSX affiche pour chaque mois la distance parcourue, le total des dépenses, le prix du kilomètre et en supplément gratuit le prix annuel du kilomètre.

COUT DE VOTRE VOITURE						
ASSURANCE: 5600			VIGNETTE: 900			
MOIS	KM	ESS.	MECA	CRED.	TOT.	P/K
1	423	213	120	0	875	2.1
2	730	425	0	0	967	1.3
3	60	0	0	0	542	9.0
4	210	180	0	0	722	3.4
5	682	345	0	0	887	1.3
6	829	410	156	0	1108	1.3
7	364	223	0	0	765	2.1
8	0					
9	0					
10	0					
11	0					
12	0					
TOTAL/KM		ESS/F	MECA/F	CR/F	TOT/F	
		3298	1796	276	0	5864
PRIX MOYEN DU KM: 1.78						

COUT ANNUEL DE VOTRE VOITURE



```

10 REM COUT ANNUEL DE VOTRE VOITURE
20 REM -----
30 DIM E(12),M(12),K(12),C(12),Ms(12)
40 FOR I=1 TO 12:READ Ms(I):NEXT I
50 DATA DE JANVIER,DE FEVRIER,DE MARS,D'AVRIL,DE MAI,DE JUIN
60 DATA DE JUILLET,D'AOUT,DE SEPTEMBRE,D'OCTOBRE,DE NOVEMBRE
  ,DE DECEMBRE
70 SCREEN 0:COLOR 4,1,1:KEY OFF
80 INPUT"PREMIERE UTILISATION (O/N)";R$
90 IF R$="O" THEN GOTO 300
100 CLS:PRINT"METTEZ LE MAGNETO K7 EN LECTURE"
110 OPEN "VOITURE" FOR INPUT AS #1
120 INPUT#1,V,A,K
130 FOR I=1 TO 12
140 INPUT#1,E(I),M(I),K(I),C(I)
150 IF M=0 AND K(I)=0 THEN M=I
160 NEXT I
170 CLOSE
180 IF M=0 THEN M=12:K(0)=K:GOTO 500
280 REM -----
290 REM DONNEES DU MOIS
300 CLS
310 IF M<>0 THEN 370
320 INPUT"PRIX ANNUEL DE L'ASSURANCE ";A
330 INPUT"PRIX DE LA VIGNETTE ";V
340 INPUT"NOMBRE DE KMS AU 1ER JANVIER";K
350 M=1
360 PRINT
370 K(0)=K
380 PRINT"ENTREZ LES DONNEES DU MOIS ";Ms(M):PRINT
390 INPUT"ESSENCE ";E(M)
400 INPUT"ENTRETIEN/MECANIQUE ";M(M)
410 INPUT"NB DE KILOMETRES AU COMPTEUR ";K(M)
420 IF K(M)<K(M-1) THEN PRINT"KMS DU MOIS PRECEDENT:";K(M-1)
  :GOTO 410
430 INPUT"REMBOURSEMENT CREDIT ";C(M)
480 REM -----
490 REM RESULTATS
500 CLS:PRINT TAB(10)"COUT DE VOTRE VOITURE":PRINT
510 PRINT "ASSURANCE:";A;" VIGNETTE:";V
520 PRINT
530 PRINT"MOIS KM ESS. MECA CRED. TOT. P/K"
540 FOR I=1 TO 12
550 PRINT USING"###";I;
560 IF I>M THEN PRINT " 0":GOTO 610
570 T=E(I)+M(I)+C(I)+(A/12)+(V/12):TT=T/(K(I)-K(I-1))
580 PRINT USING"#####";K(I)-K(I-1),E(I),M(I),C(I),T;
590 PRINT USING"###.##";TT;
600 TE=TE+E(I):TK=K(I)-K:TM=TM+M(I):TC=TC+C(I):TA=TA+T
610 NEXT I
620 PRINT:PRINT"TOTAL/KM ESS/F MECA/F CR/F TOT/F":PRINT "
  ";
630 PRINT USING"#####";TK,TE,TM,TC,TA
640 PRINT:PRINT"PRIX MOYEN DU KM:";:PRINT USING"###.##";TA/TK
650 LINE INPUT R$
680 REM -----
690 REM SAUVEGARDE
700 CLS:PRINT"APPUYEZ SUR RETURN QUAND LE MAGNETO K7 EST PR
  ET POUR L'ENREGISTREMENT"
710 LINE INPUT R$
720 OPEN "VOITURE" FOR OUTPUT AS #1
730 PRINT#1,V,A,K
740 FOR I=1 TO 12
750 PRINT#1,E(I),M(I),K(I),C(I)
760 NEXT I
770 CLOSE
780 CLS:PRINT"AU MOIS PROCHAIN !"
790 END

```

LE PROGRAMME

Lignes 30 à 180 : initialisation

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 300 à 430 : données du mois

Le programme demande le prix de l'essence, de l'entretien, du crédit et le nombre de kilomètres au compteur. Cette dernière donnée est vérifiée avec celle du mois précédent. S'il s'agit de la première utilisation ($M=0$), les lignes 320 à 340 saisissent le prix de l'assurance et de la vignette, et le nombre de kilomètres au 1^{er} janvier.

Lignes 500 à 650 : résultats

Ces lignes affichent un tableau récapitulatif des dépenses et le prix moyen au kilomètre pour chaque mois et pour l'année entière.

Lignes 700 à 790 : sauvegarde

On sauve sur cassette ou sur disquette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

A	prix de l'assurance
E()	liste des dépenses d'essence
C()	liste des remboursements du crédit
I	indice de boucle FOR/NEXT
K	nombre de kilomètres au 1 ^{er} janvier
K()	liste des kilomètres au compteur
M	mois en cours
M()	liste des dépenses d'entretien et de mécanique
M\$()	liste des noms des mois
R\$	réponse de l'utilisateur
T	total mensuel des dépenses
TA	total annuel des dépenses
TE	total annuel des dépenses d'essence
TC	total annuel des remboursements du crédit
TK	total annuel des kilomètres parcourus
TM	total annuel des dépenses d'entretien et de mécanique
TT	prix moyen du kilomètre du mois
V	prix de la vignette

CALCUL D'IMPOTS

Pour ne pas avoir de mauvaise surprise en recevant votre avis d'imposition, ce programme vous donnera plus de détails sur son calcul.

Comment l'utiliser : vous devez en priorité répondre à quatre questions. Si vous êtes marié (répondre par oui ou non), le nombre de personnes à charge qui ont des revenus, votre nombre de parts et pour finir les charges déductibles (vos achats de valeurs françaises par exemple).

Puis, pour vous, votre conjoint et les personnes à charge vous devez donner le revenu imposable (ne pas déduire les 10 et 20 %) et le taux de la déduction supplémentaire si vous y avez droit (par exemple limeur de cadres de bicyclette du département de la Loire (20 %), brodeur de la région lyonnaise utilisant des métiers pantographes (20 %), journaliste (30 %), speakerine de la télévision française (20 %), scaphandrier (10 %), mannequin d'une grande maison de couture parisienne (10 %), etc.).


Le programme tient compte des mesures fiscales de l'année 83 (majoration de l'impôt au-dessus de 20 000 et 30 000 francs, contribution sociale). Vous serez peut-être amené à modifier le programme pour les années à venir. Il faut dans tous les cas modifier les tranches d'imposition situées à la fin du programme, sinon votre impôt risque d'être inférieur à la réalité.


-VOUS-	
REVENU IMPOSABLE :	123070
DEDUCTION 10% :	-12307
DEDUCTION 25 % :	-27690.75
ABATTEMENT 20 % :	-16614.45
NET IMPOSABLE :	66457.8

-VOTRE CONJOINT-	
REVENU IMPOSABLE :	78210
DEDUCTION 10% :	-7821
ABATTEMENT 20 % :	-14077.8
NET IMPOSABLE :	56311.2

TOTAL NET IMPOSABLE :	122760
NOMBRE DE PARTS :	2
QUOTIENT FAMILIAL :	61380
TRANCHE D'IMPOSITION :	40 %
MAJORATION 5 % :	1134.4
CONTRIBUTION SOCIALE 1% :	1227.6
IMPOT :	25050

CALCUL D'IMPOT





```

10 REM CALCUL D'IMPOT
20 REM -----
30 SCREEN 0:COLOR 6,1,1:KEY OFF
40 DIM T(14):GOSUB 1000
80 REM -----
90 REM DONNEES DE CALCUL
100 CLS
110 PRINT TAB(10)"CALCUL D'IMPOTS":PRINT:PRINT
120 INPUT"ETES-VOUS MARIE (O/N)";RS
130 IF RS="O" THEN M=2 ELSE M=1
140 INPUT"NOMBRE DE PERSONNES A CHARGE";P
150 IF P>0 THEN M=M+P
160 INPUT"NOMBRE DE PARTS";N
170 IF N<1 THEN 160
180 INPUT"CHARGES DEDUCTIBLES";C
190 FOR I=1 TO M
200 CLS:IF I>2 THEN PRINT D$(3);I-2 ELSE PRINT D$(I)
210 PRINT
220 INPUT"REVENU IMPOSABLE";R(I)
230 INPUT"DEDUCTION SUPPLEMENTAIRE (O/N)";RS
240 IF RS<>"O" THEN 260
250 INPUT"TAUX";TS(I):TS(I)=TS(I)/100
260 NEXT I
380 REM -----
390 REM CALCUL
400 CLS
410 FOR I=1 TO M
420 IF I>2 THEN PRINT "-";D$(3);I-2; "-" ELSE PRINT "-";D$(I)
; "-"
430 PRINT"REVENU IMPOSABLE : ";TAB(24);R(I)
440 D1=R(I)*.1
450 IF D1>S1 THEN D1=S1
460 IF D1<I1 THEN D1=0
470 R(I)=R(I)-D1
480 PRINT"DEDUCTION 10% : ";TAB(24);-D1
490 IF TS(I)=0 THEN 540
500 DS=R(I)*TS(I)
510 IF DS>SS THEN DS=SS
520 R(I)=R(I)-DS
530 PRINT"DEDUCTION";TS(I)*100; "% : ";TAB(24);-DS
540 D2=R(I)*.2
550 IF D2>S2 THEN D2=S2
560 IF D2<I2 THEN D2=0
570 R(I)=R(I)-D2
580 PRINT"ABATTEMENT 20 % : ";TAB(24);-D2
590 PRINT"NET IMPOSABLE : ";TAB(24);R(I):PRINT
600 R=R+R(I)
610 NEXT I
680 REM -----
690 REM RESULTATS
700 R=R-C:R=INT(R/10)*10:QF=R/N:CS=R/100
710 FOR I=1 TO 14
720 D=D+T(I-1)*.05
730 IF QF<=T(I) THEN 760
740 T=T+.05
750 NEXT I
760 PRINT
770 PRINT"TOTAL NET IMPOSABLE : ";TAB(24);R
780 PRINT"NOMBRE DE PARTS : ";TAB(24);N
790 PRINT"QUOTIENT FAMILIAL : ";TAB(24);QF
800 PRINT"TRANCHE D'IMPOSITION : ";TAB(24);T*100; "%
810 R=(R*T)-(D*N)
820 IF R>20000 AND R<30000 THEN ME=R*.05:TM=5

```

```

830 IF R>=30000 THEN ME=R*.08:TM=8
840 IF ME>0 THEN PRINT"MAJORATION";TM;"%";TAB(24);
ME
850 PRINT"CONTRIBUTION SOCIALE 1%";TAB(24);CS
860 PRINT"IMPOT";TAB(24);INT(R+CS+ME)
870 R$=INPUT$(1)
880 RUN
980 REM -----
990 REM TRANCHES
1000 D$(1)="VOUS":D$(2)="VOTRE CONJOINT":D$(3)="PERSONNE A C
HARGE"
1010 FOR I=1 TO 14
1020 READ T(I)
1030 NEXT I
1040 REM TRANCHES D'IMPOSITION
1050 DATA 13770,14390,17070,26990,34700,43610,52760
1060 DATA 60870,101430,139500,165010,187700,212750,999999
1070 REM MINI-MAX DES 10,20% ET SUPPL.
1080 I1=1800:S1=50900!:S2=92000!
1090 SS=50000!
1100 RETURN

```

PROGRAMME

Lignes 100 à 260 : données de calcul

Après l'entrée des quatre premières questions, le programme saisit par une boucle FOR/NEXT le revenu et le taux de déduction supplémentaire pour tous les membres de la famille ayant des revenus.

Lignes 400 à 610 : calcul

Pour chaque revenu on déduit les 10 et 20 %, et s'il y a lieu la déduction supplémentaire.

Lignes 700 à 880 : résultats

Après l'addition de tous les revenus et déduction des charges, on recherche la tranche d'imposition dans le tableau T à partir du quotient familial (revenu imposable divisé par le nombre de parts). La réduction d'impôt (variable D) est calculée en additionnant 5 % de toutes les tranches précédentes.

Les lignes 820 à 840 calculent la majoration de l'impôt (5 % entre 20 000 et 30 000, et 8 % au-dessus de 30 000), et les lignes 700 et 850 la contribution sociale de 1%. Vous devrez modifier ces lignes les années à venir en fonction des nouvelles mesures fiscales.


Lignes 1000 à 1100 : tranches

Vous devez modifier chaque année les DATAs des lignes 1050 et 1060 avec les nouvelles tranches d'imposition que l'on trouve généralement en page 7 de la notice explicative.

Les variables I1 et S1 correspondent au plancher et au plafond de la déduction de 10 %, S2 à celle des 20 % (pas de plancher) et SS à celle de la déduction supplémentaire.

LISTE DES VARIABLES :

C	montant des charges déductibles
CS	montant de la contribution sociale
D	déduction de la tranche d'impôt
D1	montant de la déduction de 10 %
D2	montant de l'abattement de 20 %
DS	montant de la déduction supplémentaire
DS()	noms des membres de la famille
I	indice de boucle FOR/NEXT
I1	plancher de la déduction de 10 %
M	nombre de personnes ayant des revenus
ME	montant de la majoration exceptionnelle
N	nombre de parts
P	nombre de personnes à charge
QF	quotient familial
R	total des revenus imposables
R()	revenu imposable des membres de la famille
RS	réponse du contribuable
S1	plafond de l'abattement de 10 %
S2	plafond de l'abattement de 20 %
SS	plafond de la déduction supplémentaire
T	taux de la tranche d'imposition
TM	taux de la majoration exceptionnelle
TS()	taux des déductions supplémentaires



THEME 2 PEDAGOGIE

MUSIQUE

Voilà pour exercer vos talents artistiques un programme de musique.

Après son lancement, deux portées se dessinent sur l'écran ainsi qu'une flèche. Celle-ci se déplace avec les touches droite et gauche et pointe l'endroit où se positionne la note entrée à l'aide des touches de la première rangée du clavier (du DO de la première octave au LA de la seconde. exemple : 1=DO, 2=RE, 3=MI, etc.).

En dessous des portées plusieurs options vous sont proposées :


- **DUREE** : en tapant sur B, N ou C vous obtenez des blanches, des noires, ou des croches.
- **SON** : en tapant sur S puis sur une des deux flèches droite ou gauche vous pouvez la faire varier de 1 à 15.
- **TEMPO** : il varie comme l'attaque, mais après avoir tapé sur la lettre T.


Pour modifier une note il suffit de faire venir la flèche en dessous et d'effectuer le changement.

Pour effacer ou insérer une note, il faut la désigner avec la flèche puis appuyer sur DEL ou INS.

Quand votre mélodie est entièrement écrite ou pour faire des essais, la touche M la fait **exécuter** par le MSX.

MUSIQUE





```

10 REM MUSIQUE
20 REM -----
30 KEY OFF: SCREEN 1: GOSUB 1200
40 DIM N(54), D(54)
50 F$="CNBTS"+CHR$(18)+CHR$(127)+"M1234567890-="
60 F(1)=148: F(2)=140: F(3)=144
70 N$(1)="C": N$(2)="D": N$(3)="E": N$(4)="F": N$(5)="G": N$(6)="
A": N$(7)="B"
80 LF$=CHR$(10)+CHR$(8)+CHR$(8)
90 FOR I=128 TO 138 STEP 2
100 C$=C$+CHR$(I)+CHR$(I+1)+LF$
110 NEXT I
120 GOSUB 1000
130 I=0: X=2: Y=1
140 D=1: T=10: S=11
280 REM -----
290 REM SAISIE DES NOTES
300 LOCATE X,Y+6: PRINT "^": LOCATE 7,21: PRINT CHR$(F(D))
310 LOCATE 0,22: PRINT "DUREE: "; CHR$(F(D)+1): PRINT "SON: "; S; "
TEMPO: "; T;
320 A$=INPUT$(1)
330 LOCATE X,Y+6: PRINT " "
340 IF A$=CHR$(28) AND I<53 THEN I=I+1: X=X+1: IF X=29 THEN X=
2: Y=Y+8
350 IF A$=CHR$(29) AND I>0 THEN I=I-1: X=X-1: IF X=1 THEN X=28
: Y=Y-8
360 Z=INSTR(F$,A$): IF Z=0 THEN 300
370 IF Z>8 THEN Z=Z-8: GOTO 530
380 ON Z GOTO 390,390,390,400,430,460,490,520
390 D=Z: GOTO 300
400 A$=INPUT$(1): IF A$=CHR$(29) AND T>1 THEN T=T-1
410 IF A$=CHR$(28) AND T<20 THEN T=T+1
420 GOTO 300
430 A$=INPUT$(1): IF A$=CHR$(29) AND S>1 THEN S=S-1
440 IF A$=CHR$(28) AND S<15 THEN S=S+1
450 GOTO 300
460 FOR J=52 TO I STEP -1: N(J+1)=N(J): D(J+1)=D(J): NEXT J
470 N(I)=0: D(I)=0: X1=X: Y1=Y: I1=I: GOSUB 1000: X=X1: Y=Y1: I=I1
480 GOTO 300
490 FOR J=I TO 53: N(J)=N(J+1): D(J)=D(J+1): NEXT J
500 N(53)=0: D(53)=0: X1=X: Y1=Y: I1=I: GOSUB 1000: X=X1: Y=Y1: I=I1
510 GOTO 300
520 GOSUB 800: GOTO 300
530 N(I)=Z
540 D(I)=D
550 GOSUB 1050: IF X=0 THEN X=2
560 J=I: PLAY "S11T100": GOSUB 850
570 IF I<53 THEN I=I+1 ELSE X=28: Y=9
580 GOTO 300
780 REM -----
790 REM MUSIQUE
800 PLAY "S"+MID$(STR$(S),2)+"T"+MID$(STR$(10*T+32),2)
810 FOR J=0 TO I
820 GOSUB 850
830 NEXT J
840 RETURN
850 IF N(J)>7 THEN N=N(J)-7: O$="04" ELSE N=N(J): O$="03"
860 IF D(J)=1 THEN L$="L8": GOTO 880
870 IF D(J)=3 THEN L$="L2" ELSE L$="L4"
880 PLAY O$+L$+N$(N)+G$
890 RETURN
980 REM -----
990 REM DESSIN DE LA PORTEE

```

```

1000 X=0:Y=1
1010 FOR I=0 TO 53
1020 GOSUB 1050
1030 NEXT I
1040 RETURN
1050 IF X=0 THEN LOCATE 0,Y:PRINT C$:X=X+2
1060 LOCATE X,Y-1:PRINT " ":LOCATE X,Y+5:PRINT " "
1070 FOR J=Y TO Y+4:LOCATE X,J:PRINT CHR$(152):NEXT
1080 IF N(I)=0 THEN 1130
1090 IF N(I)/2=INT(N(I)/2) THEN P=2 ELSE P=0
1100 IF D(I)<2 THEN N=8:GOTO 1120
1110 IF D(I)>2 THEN N=4 ELSE N=0
1120 LOCATE X,Y+4-INT((N(I)-1)/2):PRINT CHR$(140+P+N):LOCATE
X,Y+5-INT((N(I)-1)/2):PRINT CHR$(141+P+N)
1130 X=X+1:IF X=29 THEN X=0:Y=Y+8
1140 RETURN
1180 REM -----
1190 REM GRAPHIQUES
1200 FOR I=0 TO 199
1210 READ A:VPOKE 1024+I,A
1220 NEXT I
1230 RETURN
1240 DATA 1,2,4,4,4,255,4,4
1250 DATA 128,64,32,32,32,255,32,96
1260 DATA 4,4,4,4,5,255,3,6
1270 DATA 96,96,192,192,128,255,0,0
1280 DATA 14,14,26,50,50,255,99,199
1290 DATA 0,0,0,0,0,255,192,224
1300 DATA 207,217,209,209,209,255,200,196
1310 DATA 240,48,24,24,8,255,136,136
1320 DATA 64,32,32,16,15,255,0,0
1330 DATA 144,144,160,192,192,255,64,64
1340 DATA 0,0,14,31,31,30,15,0
1350 DATA 64,64,64,64,64,64,128,0
1360 DATA 0,0,0,0,2,255,2,2
1370 DATA 2,2,2,62,126,255,126,60
1380 DATA 2,2,2,2,2,255,2,62
1390 DATA 126,126,126,60,0,255,0,0
1400 DATA 0,0,0,0,2,255,2,2
1410 DATA 2,2,2,62,66,255,66,60
1420 DATA 2,2,2,2,2,255,2,62
1430 DATA 66,66,66,60,0,255,0,0
1440 DATA 0,0,0,0,6,255,5,5
1450 DATA 5,4,4,60,124,255,124,56
1460 DATA 6,7,5,5,5,255,4,60
1470 DATA 124,124,124,56,0,255,0,0
1480 DATA 0,0,0,0,0,255,0,0

```

LE PROGRAMME

Lignes 30 à 140 : initialisation

Après la définition des caractères graphiques par le sous-programme en 1200, on place dans le tableau F() les pointeurs des caractères représentant les 3 durées et dans le tableau N\$() le nom des 7 notes pour l'instruction PLAY.

Le sous-programme en 1000 affiche la portée et les clefs.

Lignes 300 à 670 : saisie des notes

La saisie est divisée en 11 parties :

ligne 340	: déplace le curseur à droite
ligne 350	: déplace le curseur à gauche
ligne 390	: modifie la durée de la prochaine note
lignes 400 à 420	: modifient le tempo de la mélodie
lignes 430 à 450	: modifient le son de la mélodie
lignes 460 à 480	: insèrent une note sous le curseur
lignes 490 à 510	: suppriment la note sous le curseur
ligne 520	: joue la mélodie (sous-programme en 800)
lignes 530 à 580	: placent une note dans N() et D().

Lignes 800 à 900 : musique

Ces lignes jouent toutes les notes du tableau N() pour les durées contenues dans le tableau D().

La ligne 850 modifie l'octave si le numéro de la note est supérieur à 7 (variable O\$). Les lignes 860 et 870 placent dans la variable L\$ la durée de la note.


Lignes 1000 à 1140 : dessin de la portée

Le sous-programme affiche sur l'écran les portées avec les clefs de sol, ainsi que toutes les notes du tableau N().

Cette partie du programme est aussi utilisée pour supprimer ou ajouter une note (option INS et DEL).

LISTE DES VARIABLES :

A\$	dernière touche appuyée
C\$	dessin de la clef de sol
D	durée en cours
D()	liste des durées
F()	pointeurs des caractères graphiques pour chaque durée
F\$	liste des fonctions
I	position du curseur dans le tableau N()
I1	sauvegarde de I
J	indice de boucle FOR/NEXT
L\$	durée de la note à jouer
LF\$	1 flèche vers le bas + 2 flèches vers la gauche
N	2 ^e pointeur des graphiques de la note à afficher
N()	liste des notes
N\$()	noms des notes pour l'instruction PLAY
O\$	octave de la note à jouer
P	1 ^{er} pointeur des graphiques de la note à afficher
S	type du son en cours
T	tempo en cours
X et Y	position du curseur sur l'écran
X1 et Y1	sauvegardes des variables X et Y
Z	numéro de la note en cours




GEOGRAPHIE

Comment devenir un as en géographie grâce à votre ordinateur ?

Celui-ci vous dessine d'abord les cartes de la France et de la Corse. Puis il vous demande soit la préfecture d'un département, soit le numéro du département d'une préfecture. A chaque bonne réponse il affiche définitivement la ville, sinon il l'efface et vous la reproposera plus tard.

Vous pouvez compter sur l'infinie patience de votre MSX. Il s'arrêtera seulement quand toutes les villes seront rétablies sur la carte.

GEOGRAPHIE



```

10 REM GEOGRAPHIE
20 REM -----
30 DEFINT A-Z: DIM X(95), Y(95), N$(95), D$(95), G(95)
40 COLOR 4, 1, 1: SCREEN 2: KEY OFF
50 P=RND(-(TIME/10) MOD 100)
60 PSET(90, 4)
70 FOR I=1 TO 123
80 READ X, Y: LINE -(X*.7+16, Y)
90 NEXT I
100 LINE (16, 0)-(147, 190), 4, B
110 LINE (142, 64)-(147, 128), 4, B: LINE (147, 65)-(147, 127), 1
120 PSET(27, 150)
130 FOR I=1 TO 24
140 READ X, Y: LINE -(X*.7+16, Y)
150 NEXT I
160 LINE (16, 144)-(34, 190), 4, B
170 FOR I=1 TO 95
180 READ X(I), Y(I), N$(I), D$(I): X(I)=X(I)*.7+16
190 NEXT I
200 OPEN "GRP:" FOR OUTPUT AS #1
210 PAINT (100, 100), 4
220 PAINT (25, 175), 4: PAINT (28, 153)
280 REM -----
290 REM NOM DES VILLES
300 P=INT(RND(1)*95)+1: IF G(P)<>0 THEN 300
310 COLOR 1: FOR I=1 TO 6: DRAW"BM144,"+STR$(I*8+64): PRINT#1, S
  TRING$(15, 219): NEXT: COLOR 4
320 IF RND(1)>.5 THEN 380
330 DRAW"BM144, 72": PRINT#1, "VILLE: "
340 DRAW"BM144, 104": PRINT#1, "DEPARTEMENT: ": DRAW"BM144, 112": P
  RINT#1, D$(P)
350 GOSUB 600
360 IF R$<>LEFT$(N$(P), LEN(R$)) OR R$="" THEN 300
370 GOTO 430
380 DRAW"BM144, 72": PRINT#1, "DEPARTEMENT: "
390 DRAW"BM144, 104": PRINT#1, "VILLE: ": DRAW"BM144, 112": PRINT#1
  , N$(P)
400 GOSUB 600
410 IF LEN(R$)=1 THEN R$="0"+R$
420 IF R$<>D$(P) THEN 300
430 PSET (X(P), Y(P)), 1
440 G(P)=1: G=G+1
450 COLOR 1: DRAW"BM150, 180": PRINT#1, STRING$(10, 219)
460 COLOR 4: DRAW"BM150, 180": PRINT#1, G; "VILLES";
470 IF G<95 THEN 300
480 A$=INPUT$(1): RUN
580 REM -----
590 REM SAISIE DES REPONSES
600 R$="": DRAW"BM144, 80"
610 LINE (X(P), Y(P))-(X(P), Y(P)+1), 4, B
620 A$=INKEY$: IF A$="" THEN 680
630 IF A$=CHR$(13) THEN RETURN
640 IF A$=CHR$(8) OR A$=CHR$(127) OR A$=CHR$(29) THEN IF LEN
  (R$)>0 THEN R$=LEFT$(R$, LEN(R$)-1): GOTO 670 ELSE 670
650 R$=R$+A$
660 IF LEN(R$)=16 THEN RETURN
670 COLOR 1: DRAW"BM144, 80": PRINT#1, STRING$(15, 219): COLOR 4: D
  RAW"BM144, 80": PRINT#1, R$
680 LINE (X(P), Y(P))-(X(P), Y(P)+1), 1, B
690 GOTO 610
980 REM -----
990 REM FRANCE METROPOLITAINE
1000 DATA 93, 8, 91, 25, 73, 33, 72, 37, 74, 38

```

1010 DATA 68,41,57,39,54,40,52,36,53,34
 1020 DATA 53,32,44,32,45,39,47,41,48,52
 1030 DATA 50,53,50,55,45,54,44,52,38,54
 1040 DATA 37,53,32,56,26,48,19,49,19,52
 1050 DATA 16,50,3,54,2,57,7,57,8,57
 1060 DATA 5,60,8,61,8,62,2,63,6,66
 1070 DATA 7,69,9,70,12,67,25,76,29,74
 1080 DATA 29,77,33,78,33,81,38,84,38,86
 1090 DATA 39,87,38,91,42,99,51,103,52,110
 1100 DATA 50,115,58,125,51,120,44,160,40,165
 1110 DATA 45,166,44,172,48,172,59,177,62,176
 1120 DATA 66,179,79,180,77,177,90,182,93,182
 1130 DATA 95,185,100,188,102,186,105,188,108,188
 1140 DATA 110,186,114,186,116,184,114,181,115,172
 1150 DATA 129,162,131,163,138,164,140,167,143,164
 1160 DATA 158,170,165,169,184,153,185,146,178,147
 1170 DATA 172,142,174,134,168,131,168,127,174,124
 1180 DATA 174,118,170,113,172,109,169,105,169,100
 1190 DATA 162,100,160,105,156,106,159,97,172,79
 1200 DATA 176,79,178,61,186,45,164,41,162,37
 1210 DATA 157,35,154,37,150,35,147,35,138,30
 1220 DATA 138,22,135,24,135,27,129,27,128,19
 1230 DATA 122,20,119,15,115,15,115,12,113,9
 1240 DATA 110,12,107,7,106,4
 1250 REM -----
 1260 REM CORSE
 1270 DATA 16,156,14,155,6,160,5,163,7,165
 1280 DATA 5,165,5,167,8,169,6,173,8,173
 1290 DATA 7,176,10,177,8,180,13,182,13,183
 1300 DATA 15,184,18,177,18,172,20,168,20,159
 1310 DATA 18,156,19,154,18,149,16,150
 1320 REM -----
 1330 REM VILLES
 1340 DATA 127,62,TROYES,10,175,64,COLMAR,68
 1350 DATA 161,63,EPINAL,88,142,66,CHAUMONT,52
 1360 DATA 29,73,VANNES,56,120,73,AUXERRE,89
 1370 DATA 168,73,BELFORT,90,158,75,VESOUL,70
 1380 DATA 88,77,BLOIS,41,61,79,ANGERS,49
 1390 DATA 79,81,TOURS,37,46,83,NANTES,44
 1400 DATA 142,83,DIJON,21,157,83,BESANCON,25
 1410 DATA 103,88,BOURGES,18,115,90,NEVERS,58
 1420 DATA 93,94,CHATEAUX,36,48,95,LA ROCHE S/YON,85
 1430 DATA 150,95,LONS-LE-SAUNIER,39,74,99,POITIERS,86
 1440 DATA 118,99,MOULINS,03,62,103,NIORT,79
 1450 DATA 52,106,LA ROCHELLE,17,139,103,MACON,71
 1460 DATA 145,105,BOURG EN BRESSE,01,159,110,ANNECY,74
 1470 DATA 112,14,LILLE,59,108,20,ARRAS,62
 1480 DATA 102,30,AMIENS,80,135,30,MEZIERES,08
 1490 DATA 121,36,LAON,02,86,39,ROUEN,76
 1500 DATA 99,39,BEAUVAIS,60,65,43,CAEN,14
 1510 DATA 54,45,ST LO,50,30,56,ST BRIEUC,22
 1520 DATA 10,65,QUIMPER,29,46,66,RENNES,35
 1530 DATA 85,47,EVREUX,27,157,44,METZ,57
 1540 DATA 179,54,STRASBOURG,67,131,48,CHALONS S/MARNE,51
 1550 DATA 103,50,PARIS,75,142,52,BAR LE DUC,55
 1560 DATA 156,53,NANCY,54,70,60,ALENCON,61
 1570 DATA 90,60,CHARTRES,28,107,57,MELUN,77
 1580 DATA 58,66,LAVAL,53,73,68,LE MANS,72
 1590 DATA 97,70,ORLEANS,45,96,107,GUERET,23
 1600 DATA 96,113,LIMOGES,87,114,115,CLERMONT-FERRAND,63
 1610 DATA 140,114,LYON,69,134,122,ST ETIENNE,42
 1620 DATA 157,118,CHAMBERY,73,70,117,ANGOULEME,16
 1630 DATA 80,126,PERIGUEUX,24,95,125,TULLE,19

1640 DATA 104,133,AURILLAC,15,120,141,MENDE,48
1650 DATA 159,140,GAP,05,125,130,LE PUY,43
1660 DATA 137,135,PRIVAS,07,142,132,VALENCE,26
1670 DATA 153,126,GRENOBLE,38,112,180,PERPIGNAN,66
1680 DATA 150,165,MARSEILLE,13,91,174,FOIX,09
1690 DATA 58,134,BORDEAUX,33,60,155,MONT-DE-MARSAN,40
1700 DATA 62,166,PAU,64,68,168,TARBES,65
1710 DATA 103,169,CARCASSONNE,11,166,160,DRAGUIGNAN,83
1720 DATA 179,155,NICE,06,126,160,MONTPELLIER,34
1730 DATA 77,159,AUCH,32,89,160,TOULOUSE,31
1740 DATA 77,148,AGEN,47,89,151,MONTAUBAN,82
1750 DATA 100,154,ALBI,81,133,150,NIMES,30
1760 DATA 142,153,AVIGNON,84,162,149,DIGNE,04
1770 DATA 90,142,CAHORS,46,106,144,RODEZ,12
1780 DATA 7,172,AJACCIO,20,105,52,CRETEIL,94
1790 DATA 101,49,NANTERRE,92,104,48,BOBIGNY,93
1800 DATA 105,55,EVRY,91,100,52,VERSAILLES,78
1810 DATA 100,45,PONTOISE,95

LE PROGRAMME

Lignes 30 à 220 : initialisation

Après le dimensionnement des divers tableaux (voir liste des variables), le programme initialise les nombres aléatoires (ligne 50).

Les lignes 60 à 150 tracent la carte de France et de Corse avec les DATAs des lignes 1000 à 1310.

On place les coordonnées, le nom et le département des villes dans les tableaux X(), Y(), N\$() et D\$() avec les DATAs des lignes 1340 à 1810.

Lignes 300 à 470 : nom des villes

Le nombre aléatoire contenu dans la variable P pointe une des 95 villes du fichier. Si cette ville a déjà été trouvée ($G(P) < > 0$) on en essaye une autre (ligne 300).

Le nombre aléatoire de la ligne 320 détermine si l'on doit demander le nom ou le numéro du département.

La saisie de la réponse s'effectue à l'aide du sous-programme des lignes 600 à 690.

Si la réponse est juste, on place le chiffre 1 dans le tableau G() pour ne pas demander plusieurs fois la même ville.


Lignes 600 à 690 : saisie des réponses

On saisit caractère par caractère la réponse du joueur, ceci pour éviter l'instruction INPUT qui pourrait détruire le dessin si l'on tape trop de lettres ou de chiffres.

On sort du sous-programme quand le joueur a appuyé sur la touche RETURN ou si la réponse contient au moins 16 caractères.

LISTE DES VARIABLES :

A\$	dernière touche appuyée
D\$()	liste des départements
G	nombre de villes déjà trouvées
G()	liste des villes déjà trouvées
I	indice de boucle FOR/NEXT
N\$()	liste des villes
P	pointe la prochaine ville à trouver
R\$	réponse du joueur
X() et Y()	coordonnées des villes sur l'écran




APPRENDRE A TAPER AU CLAVIER


Si vous désirez apprendre à taper ou améliorer votre vitesse de frappe au clavier, le programme qui suit devrait vous être très utile.

Après le dessin du clavier, le MSX affiche une lettre au hasard sur la touche correspondante. Vous devez bien sûr taper immédiatement cette même lettre pendant le temps très court qui vous est imparti. Chaque réponse correcte vous donne un point. Si vous obtenez plus de 17/20 le niveau de rapidité augmente. Pour les niveaux supérieurs à 5 il n'est plus possible de regarder le clavier, vous devrez donc le connaître par cœur !

Paradoxalement, vous devrez taper le programme avant de pouvoir vous exercer !



APPRENDRE A TAPER AU CLAVIER



```

10 REM APPRENDRE A TAPER AU CLAVIER
20 REM -----
30 SCREEN 0
40 DIM C$(46),X(46),Y(46)
50 E$=CHR$(1):FOR I=1 TO 6:G$(I)=E$+CHR$(85+I):NEXT
60 LF$=CHR$(10)+CHR$(8)+CHR$(8)+CHR$(8)
70 B$=G$(3)+G$(2)+G$(4)+LF$+G$(1)+" "+G$(1)+LF$+G$(5)+G$(2)+
G$(6)
80 DATA 1,2,3,4,5,6,7,8,9,0,-,=
90 DATA Q,W,E,R,T,Y,U,I,O,P,[,]
100 DATA A,S,D,F,G,H,J,K,L,;, '
110 DATA Z,X,C,V,B,N,M,"",",",.,/, " "
120 Y=3:X=1
130 FOR I=1 TO 46
140 READ C$(I)
150 X(I)=X:Y(I)=Y
160 X=X+3
170 IF X>35 THEN Y=Y+3:X=INT(Y/3)
180 NEXT I
190 A=RND(-(TIME/10) MOD 100)
200 N=0
280 REM -----
290 REM DESSIN DU CLAVIER
300 CLS
310 FOR I=1 TO 45
320 LOCATE X(I)-1,Y(I)-1:PRINT B$
330 NEXT I
340 LOCATE 9,14:PRINT G$(3);:FOR I=1 TO 22:PRINT G$(2);:NEXT
PRINT G$(4)
350 PRINT TAB(9);G$(1);TAB(32);G$(1)
360 PRINT TAB(9);G$(5);:FOR I=1 TO 22:PRINT G$(2);:NEXT:PRIN
T G$(6)
370 X(46)=20:Y(46)=15
380 REM -----
390 REM APPRENTISSAGE
400 P=0
410 LOCATE 0,20:PRINT CHR$(24);"NIVEAU: ";N; " MOYE
NNE: "
420 FOR K=1 TO 20
430 C=INT(RND(1)*46)+1
440 IF C=46 THEN LOCATE X(C),Y(C):PRINT CHR$(219):GOTO 460
450 LOCATE X(C),Y(C):PRINT C$(C)
460 FOR I=1 TO 350-N*20
470 A$=INKEY$:IF A$<>" " THEN 490
480 NEXT I
490 IF A$=C$(C) THEN P=P+1
500 LOCATE 30,20:PRINT STR$(P);"/";MID$(STR$(K),2)
510 FOR I=1 TO 350-N*20:Z$=INKEY$:NEXT I
520 LOCATE X(C),Y(C):PRINT " "
530 NEXT K
540 IF P>17 THEN N=N+1:PLAY"CDEFG" ELSE PLAY"GFEDC"
550 FOR I=1 TO 3000:NEXT I
560 IF N>9 THEN N=9
570 GOTO 300

```

LE PROGRAMME

Lignes 30 à 200 : lecture des DATAs

On place dans le tableau C\$() les lettres qui composent le clavier et dans les tableaux X() et Y() leurs coordonnées sur l'écran. La barre d'espace est remplacée par un caractère blanc.

La ligne 190 initialise les nombres aléatoires.

Lignes 300 à 370 : dessin du clavier

Chaque touche du clavier est représentée par un carré sur l'écran. La touche d'espace étant de forme différente, ce sont les lignes 340 à 370 qui l'affichent.

Lignes 400 à 570 : apprentissage

On affiche vingt fois de suite une lettre au hasard aux coordonnées pointées par les tableaux X et Y.

Pour la barre d'espacement, on affiche le caractère graphique 219. La ligne 470 saisit la réponse du joueur pendant un temps précis défini par le niveau N. Si on a plus de 17 réponses exactes, on augmente automatiquement le niveau (ligne 540).

LISTE DES VARIABLES :

A\$	réponse tapée au clavier
B\$	dessin d'une touche
C	numéro de la lettre choisie au hasard
C\$()	tableau des lettres du clavier
G\$	caractères graphiques composant le dessin d'une touche
I	indice de boucle FOR/NEXT
K	indice de boucle FOR/NEXT
LF\$	1LF + 4 BS
N	niveau de rapidité
P	nombre de points
X et Y	coordonnées temporaires
X() et Y()	coordonnées des lettres sur l'écran

ALPHABET

Si, depuis que vous avez acheté votre MSX, vos charmants enfants vous le réclament par curiosité, proposez-leur ce programme qui les occupera pendant que vous lirez la suite de ce livre.


L'alphabet entier se dessine sur l'écran. Il y a une lettre à trouver et elle est désignée par un rectangle noir.

Si le caractère frappé par l'enfant en réponse est le bon, il entend une petite musique bien connue. Dans le cas contraire, c'est une gamme descendante qui se fait entendre.

Il n'y a qu'un seul essai à chaque coup. Par contre, l'ordinateur demandera inlassablement une nouvelle lettre...



ALPHABET



```

10 REM ALPHABET
20 REM -----
30 SCREEN 1
40 FOR I=0 TO 31
50 READ A:VPOKE 1024+I,A
60 NEXT
70 DATA 0,30,18,255,128,128,255,60
80 DATA 63,33,33,255,1,1,255,60
90 DATA 63,32,47,41,47,32,255,30
100 DATA 255,1,61,37,61,1,255,30
110 A=RND(-(TIME/10) MOD 100)
120 PLAY"OST200"
130 REM -----
140 REM AFFICHE LE JEU
150 CLS
160 LOCATE 8,6
170 L=INT(RND(1)*26)+1
180 FOR I=1 TO 26
190 IF I=L THEN PRINT CHR$(219);:GOTO 260
200 PRINT CHR$(64+I);
210 IF I=13 THEN LOCATE 8,10
220 NEXT I
230 LOCATE 5,22:PRINT"QUELLE EST LA LETTRE
E ?";
240 COLOR 1
250 P=10:D=26
260 FOR I=10 TO 26 STEP 2
270 LOCATE I,18:PRINT CHR$(130);CHR$(131)
280 NEXT I
290 REM -----
300 REM LE JEU
310 FOR I=1 TO 20
320 A$=INKEY$:IF A$<>" " THEN 490
330 NEXT I
340 LOCATE P,18:PRINT CHR$(130);CHR$(131)
350 P=P-2:IF P<2 THEN P=26
360 D=D-2:IF D<2 THEN D=26
370 LOCATE P,18:PRINT CHR$(128);CHR$(129)
380 LOCATE D,18:PRINT " "
390 GOTO 400
400 IF A$<"A" OR A$>"Z" THEN 410
410 LOCATE 18,23:PRINT A$;
420 IF A$=CHR$(64+L) THEN PLAY"CCDCEDCEDDC":GOTO 530
430 PLAY"BAGFEDC"
440 FOR I=1 TO 2000:NEXT I
450 GOTO 200

```

QUI MANQU

LE PROGRAMME

Lignes 30 à 90 : initialisation

Initialisation des caractères graphiques et du registre son du MSX.

Lignes 200 à 330 : affichent le jeu

Ces lignes affichent l'alphabet et un carré noir sur la lettre à trouver. Elles affichent aussi un petit train en bas de l'écran.

Lignes 400 à 540 : le jeu

Les lignes 400 à 420 testent vingt fois de suite si une touche a été enfoncée. Si c'est le cas, on va en ligne 490 pour la vérifier. Dans le cas contraire on déplace le train d'une case.

LISTE DES VARIABLES :

A\$	dernière touche appuyée
D	dernière position du train
I	indice de boucle FOR/NEXT
L	lettre à trouver
P	position de départ du train


APPRENDRE A COMPTER

Une idée originale pour apprendre à compter !


Un nombre différent de pommes, cow-boys, téléphones, etc., se dessine à l'écran. Votre bambin doit tout simplement les compter et donner au professeur (le plus patient des professeurs !) le nombre qu'il pense avoir trouvé.

Le MSX ne donne pas la réponse, mais indique musicalement si elle est exacte.

On recommence ?



APPRENDRE A COMPTER



```

10 REM APPRENDRE A COMPTER
20 REM -----
30 DEFINT A-Z
40 OPEN "GRP:" FOR OUTPUT AS #1
50 DIM N$(22)
60 SCREEN 2,3:GOSUB 400
70 I=RND(-(TIME/10) MOD 100)
80 I=INT(RND(1)*22)+1
90 CLS
100 N=INT(RND(1)*9)+1
110 NS=0
120 FOR J=1 TO N
130 IF XX(J)>X(J) THEN NS=NS+1:GOTO 150
140 XX(J)=XX(J)+4
150 IF YY(J)>Y(J) THEN NS=NS+1:GOTO 170
160 YY(J)=YY(J)+4
170 PUT SPRITE J,(XX(J),YY(J)),J+1,I
180 NEXT J
190 IF NS<N*2 THEN 110
200 DRAW"BM96,160":PRINT#1,"NOMBRE ";
210 IF LEFT$(N$(I),1)="A" OR LEFT$(N$(I),1)="H" THEN PRINT#1
,"D" ELSE PRINT#1,"DE"
220 PRINT#1,TAB(16-LEN(N$(I))/2;N$(I);" ?"
230 R$=INPUT$(1)
240 R=VAL(R$):IF R=0 THEN 230
250 IF R=N THEN PLAY"T20002CDEFGAB03CDEF6AB04C" ELSE PLAY"T2
0003BAGFEDC02BAGFEDC"
260 FOR I=1 TO 9
270 PUT SPRITE I,(0,209):XX(I)=0:YY(I)=0
280 NEXT I
290 GOTO 80
300 REM -----
390 REM GRAPHIQUES
400 DATA 7,5,7,1,15,31,51,99
410 DATA 111,7,7,3,3,3,3,15
420 DATA 240,176,240,192,248,252,230,227
430 DATA 227,224,96,96,96,96,96,120
440 DATA 0,0,7,4,4,31,16,19
450 DATA 19,16,16,19,18,18,16,31
460 DATA 0,192,224,32,34,250,10,202
470 DATA 202,10,10,206,72,72,8,248
480 DATA 0,3,13,16,32,32,64,97
490 DATA 65,33,32,16,13,3,0,0
500 DATA 0,128,96,16,56,72,132,12
510 DATA 4,8,8,16,96,128,0,0
520 DATA 0,3,12,16,33,35,69,65
530 DATA 65,65,33,33,16,12,3,0
540 DATA 0,192,48,8,4,116,66,98
550 DATA 66,66,4,4,8,48,192,0
560 DATA 0,63,196,159,228,9,18,36
570 DATA 40,40,36,34,33,32,63,24
580 DATA 0,252,35,249,39,144,72,36
590 DATA 20,20,100,100,148,4,252,24
600 DATA 127,64,84,92,72,64,64,66
610 DATA 67,65,64,84,92,72,64,127
620 DATA 252,4,84,116,36,4,4,132
630 DATA 132,4,4,84,116,36,4,252
640 DATA 255,68,68,255,17,17,255,68
650 DATA 68,255,17,17,255,68,68,255
660 DATA 255,68,68,255,17,17,255,68
670 DATA 68,255,17,17,255,68,68,255
680 DATA 63,32,36,32,32,36,32,63
690 DATA 32,32,32,32,36,32,63,0
700 DATA 252,4,36,4,4,36,4,252
710 DATA 4,36,4,4,4,4,252,0
720 DATA 3,6,10,18,34,74,130,146
730 DATA 131,164,137,144,164,192,255,0
740 DATA 255,1,73,1,73,1,73,1
750 DATA 255,2,36,8,144,32,192,0
760 DATA 1,64,200,48,25,141,95,35
770 DATA 83,139,11,3,3,3,3,7
780 DATA 32,162,68,136,52,227,128,241
790 DATA 202,164,146,168,128,128,128,192
800 DATA 0,0,12,30,63,127,127,63
810 DATA 30,12,0,0,0,0,0,0
820 DATA 7,26,100,132,8,136,144,16
830 DATA 24,60,126,255,255,126,60,24
840 DATA 0,0,3,1,1,1,1,1

```

```

850 DATA 7,3,7,31,11,23,44,120
860 DATA 128,128,224,64,192,64,192,64
870 DATA 240,96,240,252,104,244,26,15
880 DATA 63,96,192,135,140,152,248,152
890 DATA 24,48,96,192,255,1,3,3
900 DATA 252,6,3,225,49,25,31,25
910 DATA 24,12,6,3,255,128,192,192
920 DATA 0,1,19,31,255,127,63,31
930 DATA 31,31,31,63,63,127,31,0
940 DATA 192,224,252,255,254,254,254,252
950 DATA 252,252,254,254,255,252,128,0
960 DATA 0,0,0,0,0,7,31,127
970 DATA 127,255,255,255,255,255,127,62
980 DATA 0,7,24,44,206,199,131,129
990 DATA 128,192,192,192,128,128,0,0
1000 DATA 1,7,15,63,255,80,76,69
1010 DATA 65,65,73,68,35,17,8,7
1020 DATA 0,192,224,248,254,20,102,70
1030 DATA 6,6,38,70,138,21,39,197
1040 DATA 0,0,1,3,7,15,15,31
1050 DATA 63,63,127,255,3,3,3,0
1060 DATA 0,0,0,128,192,224,224,240
1070 DATA 248,248,252,254,128,128,128,0
1080 DATA 3,3,1,7,15,11,19,11
1090 DATA 6,3,2,2,3,2,21,10
1100 DATA 128,128,0,192,224,160,192,160
1110 DATA 148,168,212,160,192,128,0,0
1120 DATA 7,31,7,7,3,31,63,103
1130 DATA 127,55,7,6,6,6,6,7
1140 DATA 192,240,64,192,128,240,248,216
1150 DATA 223,204,192,192,192,192,192,249
1160 DATA 63,127,240,127,63,3,3,3
1170 DATA 3,3,3,31,3,15,3,31
1180 DATA 252,254,15,254,252,192,192,192
1190 DATA 192,192,192,192,192,192,192,192
1200 DATA 0,0,0,15,63,115,247,231
1210 DATA 255,255,255,127,31,7,0,0
1220 DATA 8,30,55,34,240,252,254,255
1230 DATA 255,255,255,255,254,248,0,0
1240 DATA 12,14,27,53,106,213,170,213
1250 DATA 107,53,26,14,4,0,0,0
1260 DATA 0,0,0,128,192,96,240,152
1270 DATA 12,6,3,1,0,0,0,0
1280 FOR I=1 TO 22
1290 A$=""
1300 FOR J=1 TO 32
1310 READ A:A$=A$+CHR$(A)
1320 NEXT J
1330 SPRITE$(I)=A$
1340 NEXT I
1350 FOR I=1 TO 22:READ N$(I):NEXT
1360 FOR I=1 TO 9:READ X(I),Y(I):NEXT
1370 RETURN
1380 DATA ROBOTS,MACHINES,HORLOGES,PIECES,TELEPHONES,CARTES,
MURS
1390 DATA DOMINOS,DES,ARBRES,CERISES,TOURS EIFFEL,CLOCHES,CA
RTES DE FRANCE
1400 DATA POIRES,CHINOIS,SAPINS,SKIEURS,COW-BOYS
1410 DATA CLEFS,POMMES,DRAPEAUX
1420 DATA 96,64,162,64,128,24,64,24,32,64,192,24,128,104,64,
104,192,104

```

LE PROGRAMME

Lignes 30 à 100 : initialisation

L'appel du sous-programme en 400 place dans le tableau N\$() les noms des objets qui peuvent apparaître à l'écran, définit le dessin de ces objets dans des sprites et leurs coordonnées sur l'écran (tableaux X et Y).

On place dans les variables N et I des nombres aléatoires pour définir respectivement le nombre et le type de caractères graphiques à afficher.

Lignes 110 à 290 : le jeu

Après l'affichage des N sprites, on saisit la réponse du joueur dans la variable R\$. Si la réponse est juste, on joue une gamme montante, dans le cas contraire une gamme descendante.

LISTE DES VARIABLES :

I	numéro de l'objet à afficher
J	indice de boucle FOR/NEXT
NS	nombre de sprites en place
N\$()	noms des objets
R	valeur décimale de R\$
R\$	réponse du joueur
X() et Y()	coordonnées des sprites sur l'écran
XX() et YY()	coordonnées actuelles des sprites


TABLES DE MULTIPLICATION

Attention : programme réservé aux moins de dix ans !

Vos enfants vont enfin pouvoir réviser tranquillement leurs tables de multiplication.

L'ordinateur demande juste la table que vous voulez obtenir (jusqu'à 22).


Aucun doute, le MSX ne se trompe jamais !



A computer monitor is shown with a thick black border. Inside the screen, on the left, is a label 'TABLE DE 9' with an arrow pointing to the right. To the right of the label is a table of multiplication results for the number 9, ranging from 1 to 12.

1	X	9	=	9
2	X	9	=	18
3	X	9	=	27
4	X	9	=	36
5	X	9	=	45
6	X	9	=	54
7	X	9	=	63
8	X	9	=	72
9	X	9	=	81
10	X	9	=	90
11	X	9	=	99
12	X	9	=	108

TABLES DE MULTIPLICATION



```

10 REM TABLES DE MULTIPLICATION
20 REM -----
30 OPEN "GRP:" FOR OUTPUT AS #1
40 SCREEN 0:COLOR 6,1,1:KEY OFF
50 LOCATE 10,10:PRINT "QUELLE TABLE ";
60 INPUT T$
70 T=VAL(T$)
80 IF T<1 OR T>22 THEN 40
180 REM -----
190 REM TABLE
200 SCREEN 2
210 FOR I=1 TO 12
220 DRAW"BM112,"+STR$(I*8+32)
230 PRINT#1,USING"## x";I;
240 PRINT#1,USING"## =";T;
250 PRINT#1,USING"###";I*T;
260 NEXT I
270 LINE (100,8)-(204,170),,B
280 DRAW"BM24,88":PRINT#1,"TABLE"
290 PRINT#1," DE ";
300 PRINT#1,USING"##";T
310 LINE (16,80)-(70,80)
320 LINE -(85,95):LINE -(69,111)
330 LINE -(16,111):LINE -(16,80)
340 T$=INPUT$(1)
350 RUN

```

LE PROGRAMME

Lignes 30 à 80 : saisie

Ces lignes saisissent la table de votre choix. La ligne 80 vérifie que votre réponse est comprise entre 1 et 22.

Lignes 200 à 350 : édition de la table

On affiche avec une boucle FOR/NEXT les 12 premières multiplications de la table choisie.

Les lignes 270 à 330 dessinent le cadre et le numéro de la table.

LISTE DES VARIABLES :


I	indice de boucle FOR/NEXT
T	numéro de la table
T\$	réponse du joueur

LES QUATRE OPERATIONS


Vous n'êtes pas très fort en calcul mental ? Le MSX vous sera d'un grand secours.

Il vous propose une des quatre opérations composées de nombres tirés au hasard. Vous devez entrer votre solution comme si vous l'écriviez sur une feuille de papier (de gauche à droite, les chiffres se décalent automatiquement sur l'écran).

Si vous avez du mal à résoudre ces problèmes, reportez-vous au programme précédent !



LES QUATRE OPERATIONS



```

10 REM LES QUATRE OPERATIONS
20 REM -----
30 O$(1)="ADDITION":O$(2)="SOUSTRACTION"
40 O$(3)="MULTIPLICATION":O$(4)="DIVISION"
50 S$(1)="+":S$(2)="-"
60 S$(3)="X"
70 SCREEN 0:FOR I=0 TO 23
80 READ A:VPOKE 2048+128*8+I,A
90 NEXT I
100 DATA 0,0,0,255,255,0,0,0
110 DATA 24,24,24,24,24,24,24,24
120 DATA 24,24,24,31,31,24,24,24
130 CLS:O=RND(-(TIME/10) MOD 100)
140 O=INT(RND(1)*4)+1
150 A=INT(RND(1)*99)+1
160 B=INT(RND(1)*99)+1
170 IF O>2 AND B>9 THEN 160
180 IF O=4 AND A/B<>INT(A/B) THEN 150
190 IF O=2 AND A<B THEN 150
200 ON O GOTO 210,220,230,240
210 R=A+B:GOTO 300
220 R=A-B:GOTO 300
230 R=A*B:GOTO 300
240 R=A/B:GOTO 400
280 REM -----
290 REM OPERATIONS + - *
300 PRINT TAB(18-LEN(O$(O))/2);O$(O)
310 LOCATE 18,9:PRINT USING"##";A
320 LOCATE 14,12:PRINT S$(O); " ";
330 PRINT USING"##";B
340 LOCATE 14,15:PRINT STRING$(6,128)
350 LOCATE 14,18:PRINT"= "
360 GOTO 500
380 REM -----
390 REM DIVISION
400 PRINT TAB(18-LEN(O$(O))/2);O$(O)
410 LOCATE 13,12:PRINT USING"## " ";A,B
420 FOR I=10 TO 20
430 LOCATE 16,I:PRINT CHR$(129)
440 NEXT I
450 LOCATE 16,15:PRINT CHR$(130)STRING$(3,128)
480 REM -----
490 REM SAISIE REPONSE
500 R$=""
510 A$=INPUT$(1)
520 IF A$<"0" OR A$>"9" THEN 510
530 R$=R$+A$
540 IF LEN(R$)>LEN(STR$(R))-1 THEN 700
550 LOCATE 17,18:PRINT USING"###";VAL(R$)
560 IF VAL(R$)=R THEN 710
570 IF VAL(R$)>R THEN 700
580 GOTO 510
680 REM -----
690 REM GAGNE ET PERDU
700 PLAY"04C03BAGFEDC":GOTO 720
710 PLAY"03CDEFGAB04C"
720 LOCATE 17,18:PRINT USING"###";R
730 A$=INPUT$(1)
740 GOTO 130

```

LE PROGRAMME

Lignes 30 à 450 : initialisation et opérations

On place dans le tableau O\$() le nom des quatre opérations et dans le tableau S\$() leurs signes.

Les lignes 70 à 120 définissent trois caractères graphiques.

La variable O définissant le type d'opération à effectuer (nombre aléatoire) et A et B les opérandes, on les affiche après le calcul du résultat R (lignes 210 à 240).

Lignes 500 à 740 : saisie réponse et résultats

La ligne 510 saisit votre réponse et la ligne 550 l'affiche par un PRINT USING.

Si votre réponse est juste, la ligne 700 envoie une gamme montante, sinon la ligne 710 envoie une gamme descendante.

LISTE DES VARIABLES :

A	première opérande
A\$	dernière touche appuyée
B	deuxième opérande
I	indice de boucle FOR/NEXT
O	opération choisie
O\$()	noms des opérations
R	résultat de l'opération
R\$	réponse du joueur
S\$()	signes des opérations

QUESTIONS/REPONSES

Le jeu des questions-réponses est composé de deux programmes. Le premier permet de constituer un fichier de mots liés par une question. L'exemple ci-dessous est celui des réalisateurs de cinéma. Il demande en priorité le nom du fichier. Pour notre exemple nous avons répondu "CINEMA" (six lettres maximum).

Si vous l'utilisez pour la première fois, le MSX demande la question (toujours pour notre exemple nous avons répondu "Quel est le réalisateur du film").

Puis le programme propose un menu de trois options :


- 1 - **SAISIE** : vous devez donner le mot de la question suivi de la réponse juste, par exemple "Pickpocket" et "Bresson". Quand vous utiliserez le deuxième programme, il demandera alors : "Quel est le réalisateur de Pickpocket", et la réponse devra être "Bresson".
- 2 - **LISTING** : liste sur l'écran la totalité du fichier. Vous pouvez arrêter cette liste en appuyant sur n'importe quelle touche (idem pour continuer).
- 3 - **SAUVEGARDE** : sauve le fichier sur cassette ou sur disquette.

Le deuxième programme est le jeu proprement dit. Il demande comme le premier le nom du fichier. Puis il propose au hasard une des X questions du fichier.

Quand vous désirez arrêter, tapez le mot "FIN" comme réponse, les résultats s'affichent automatiquement.

Ces deux programmes permettent de créer une multitude de fichiers dans des domaines très variés : littérature, peinture, histoire, géographie, musique, sport, animaux, plantes, langues étrangères, informatique, etc., et bien d'autres que vous ne manquerez pas de découvrir.

CREATION QUESTIONS/REPONSES



```

10 REM CREATION QUESTIONS/REPONSES
20 REM -----
30 CLEAR 9000: DIM Q$(1000,2)
40 E$=CHR$(5)
50 SCREEN 0: COLOR 15,1,1: KEY OFF
60 INPUT "NOM DU FICHIER "; N$
70 IF LEN(N$)>8 THEN 60
80 INPUT "PREMIERE UTILISATION (O/N)"; R$
90 IF R$="O" THEN GOTO 170
100 CLS: PRINT "METTEZ LE MAGNETO K7 EN LECTURE"
110 OPEN N$ FOR INPUT AS #1
120 INPUT #1, Q$
130 IF EOF(1) THEN CLOSE: GOTO 200
140 N=N+1
150 INPUT #1, Q$(N,1), Q$(N,2)
160 GOTO 130
170 INPUT "LA QUESTION "; Q$
180 REM -----
190 REM MENU
200 CLS: PRINT TAB(18)"MENU"
210 PRINT TAB(13)"1 SAISIE 1": PRINT TAB(13)"2 LISTING
220 PRINT TAB(13)"3 SAUVEGARDE 3"
230 PRINT: PRINT TAB(13)"VOTRE CHOIX ?";
240 R$=INPUT$(1): R=VAL(R$)
250 IF R<1 OR R>3 THEN 230
260 ON R GOTO 300,400,500
280 REM -----
290 REM SAISIE
300 CLS: PRINT: PRINT TAB(9)"SAISIE D'UNE QUESTION": PRINT
310 PRINT: PRINT: PRINT " QUESTION: "; Q$
320 LOCATE 10,10: PRINT E$; "NOT PROPOSE : ";: LINE INPUT A$
330 IF A$="" THEN 200
340 N=N+1: Q$(N,1)=A$
350 LOCATE 10,12: PRINT E$; "REPONSE : ";: LINE INPUT Q$(N,
2)
360 GOTO 300
380 REM -----
390 REM LISTING
400 CLS
410 FOR I=1 TO N
420 PRINT Q$(I,1); " --> "; Q$(I,2)
430 IF INKEY$<>" " THEN A$=INPUT$(1)
440 NEXT I
450 A$=INPUT$(1)
460 GOTO 200
480 REM -----
490 REM SAUVEGARDE
500 CLS: PRINT "APPUYEZ SUR RETURN QUAND LE MAGNETO K7 EST PR
ET POUR L'ENREGISTREMENT"
510 LINE INPUT R$
520 OPEN N$ FOR OUTPUT AS #1
530 PRINT #1, Q$
540 FOR I=1 TO N
550 PRINT #1, Q$(I,1); ", "; Q$(I,2)
560 NEXT I
570 CLOSE
580 CLS: PRINT "AU REVOIR !"
590 END

```

PROGRAMME "CREATION QUESTIONS/REPONSES"

Lignes 30 à 170 : initialisation

Le programme demande en priorité le nom du fichier puisque l'on peut créer plusieurs jeux "Questions/réponses".

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement, sinon l'utilisateur doit donner la phrase qui correspond à la question.

Lignes 200 à 250 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 300 à 360 : saisie

Le programme saisit le mot proposé puis la réponse. Si le mot proposé est une chaîne vide (A\$=""), on retourne au menu.

Lignes 400 à 460 : listing

On affiche à l'écran par une boucle FOR/NEXT l'intégralité du fichier. La ligne 430 teste le clavier pour permettre à l'utilisateur d'arrêter l'édition à son gré.


Lignes 500 à 590 : sauvegarde

Ces lignes sauvent sur cassette ou sur disquette le fichier dès que la touche RETURN a été appuyée.

LISTE DES VARIABLES :

A\$	dernier mot proposé
E\$	caractère d'effacement d'une ligne (code A 0115)
I	indice de boucle FOR/NEXT
N	nombre de questions dans le fichier
N\$	nom du fichier
Q\$	la question
Q\$(,)	liste des mots proposés et des réponses
R	valeur numérique de R\$
R\$	choix de l'utilisateur

JEU QUESTIONS/REPONSES



```

10 REM JEU QUESTIONS/REPONSES
20 REM -----
30 CLEAR 10000: DIM Q$(1000,2)
40 E$=CHR$(5)
50 SCREEN 0:CLS
60 INPUT "NOM DU FICHIER ";N$
70 IF LEN(N$)>8 THEN 60
80 CLS:PRINT "METTEZ LE MAGNETO K7 EN LECTURE"
90 OPEN N$ FOR INPUT AS #1
100 INPUT #1,Q$
110 IF EOF(1) THEN CLOSE:GOTO 200
120 N=N+1
130 INPUT #1,Q$(N,1),Q$(N,2)
140 GOTO 110
180 REM -----
190 REM QUESTIONS
200 CLS
210 Q=INT(RND(1)*N)+1
220 LOCATE 0,10:PRINT Q$;" ";Q$(Q,1);"? "
230 LOCATE 0,14:LINE INPUT "REPONSE :";R$
240 IF R$="FIN" THEN 400
250 P=P+1
260 IF R$=Q$(Q,2) THEN 290
270 LOCATE 0,14:PRINT E$;"REPONSE :";Q$(Q,2)
280 GOTO 310
290 LOCATE 0,18:PRINT "REPONSE JUSTE"
300 B=B+1
310 A$=INPUT$(1)
320 GOTO 200
380 REM -----
390 REM FIN
400 CLS
410 LOCATE 8,10:PRINT B;"REPONSES JUSTES SUR";P
420 PRINT:PRINT:PRINT:END

```

PICKPOCKET --> BRESSON
 AMERICAN GRAFFITI --> LUCAS
 METROPOLIS --> LANG
 ANNIE HALL --> ALLEN
 LES LUMIERES DE LA VILLE --> CHAPLIN
 LE PARRAIN --> COPPOLA
 LE LIVRE DE LA JUNGLE --> DISNEY
 UN TAXI MAUVE --> BOISSET
 ZAZIE DANS LE METRO --> MALLE
 MAX ET LES FERRAILLEURS --> SAUTET
 PEUR SUR LA VILLE --> VERNEUIL
 LE SALON DE MUSIQUE --> RAY
 MORT A VENISE --> VISCONTI
 KAGEMUSHA --> KUROSAWA
 CHINATOWN --> POLANSKI
 LE CUIRASSE POTEMKINE --> EISENSTEIN

LE PROGRAMME "JEU QUESTIONS/REPONSES"

Lignes 30 à 140 : lecture du fichier

Après la saisie du nom, le programme lit le fichier sur la cassette ou sur la disquette.

Lignes 200 à 320 : question

Le programme propose la question pointée par la variable Q (nombre aléatoire).

La ligne 230 saisit la réponse du joueur. S'il s'agit du mot "FIN", le programme saute aux lignes 400 pour la fin du jeu.

En cas de réponse juste, la ligne 300 incrémente le nombre de points B.

Lignes 400 à 420 : fin

Ces lignes affichent le nombre de questions posées et le nombre de réponses justes.

LISTE DES VARIABLES :

A\$	dernière touche appuyée
B	nombre de réponses justes
E\$	caractère d'effacement d'une ligne (code ASCII 5)
I	indice de boucle FOR/NEXT
N	nombre de questions dans le fichier
N\$	nom du fichier
P	nombre de questions posées
Q	pointe la prochaine question
Q\$	la question
Q\$ (,)	liste des mots proposés et des réponses
R\$	réponse du joueur

CONJUGAISONS

Vos enfants ont-ils du mal avec leurs conjugaisons ?


Pour les aider, voilà un programme qui leur permettra d'obtenir le listing des conjugaisons des verbes réguliers des deux premiers groupes pour les quatre premiers temps.


Pour cela, il suffit de donner à votre MSX le verbe désiré à l'infinitif pour obtenir sa conjugaison au présent, à l'imparfait, au passé simple et au futur simple.

Si par erreur un verbe d'un autre groupe est demandé, l'ordinateur répond "verbe inconnu".

VERBE MANGER			
PRESENT		PASSE SIMPLE	
JE MANGE		JE MANGEAI	
TU MANGES		TU MANGEAS	
IL MANGE		IL MANGEA	
ELLE MANGE		ELLE MANGEA	
NOUS MANGEONS		NOUS MANGEAMES	
VOUS MANGEZ		VOUS MANGEATES	
ILS MANGENT		ILS MANGERENT	
ELLES MANGENT		ELLES MANGERENT	
IMPARFAIT		FUTUR SIMPLE	
JE MANGEAIS		JE MANGERAI	
TU MANGEAIS		TU MANGERAS	
IL MANGEAIT		IL MANGERA	
ELLE MANGEAIT		ELLE MANGERA	
NOUS MANGIONS		NOUS MANGERONS	
VOUS MANGIEZ		VOUS MANGEREZ	
ILS MANGEAIENT		ILS MANGERONT	
ELLES MANGEAIENT		ELLES MANGERONT	

CONJUGAISONS





```

10 REM CONJUGAISONS
20 REM -----
30 DIM A$(56)
40 GOSUB 400
50 SCREEN 0
60 INPUT "VERBE "; V$
70 IF V$="" THEN 60
80 IF RIGHT$(V$,2)="ER" THEN G=1:GOTO 100
90 IF RIGHT$(V$,2)="IR" THEN G=2 ELSE PRINT "VERBE INCONNU ..
  ":GOTO 60
100 Z$=LEFT$(V$,1):A$(1)="JE"
110 IF Z$="A" OR Z$="E" OR Z$="I" OR Z$="O" OR Z$="U" OR Z$=
  "Y" THEN A$(1)="J'"
120 CLS:PRINT TAB(10); "VERBE "; V$
130 V$=LEFT$(V$,LEN(V$)-2):V1$=LEFT$(V$,LEN(V$)-1)
140 IF G=2 THEN U$="I" ELSE U$=""
150 X=0:Y=2
160 FOR T=1 TO 4
170 LOCATE X,Y:PRINT T$(T):Y=Y+1
180 P=G*6+T*12-9
190 FOR I=1 TO 8
200 Y=Y+1:LOCATE X,Y
210 X$=LEFT$(A$(P),1)
220 IF G=1 AND RIGHT$(V$,1)="G" AND (X$="A" OR X$="O") THEN
  PRINT A$(I); " "; V$; "E"; A$(P);:GOTO 260
230 IF G=1 AND T=1 AND RIGHT$(V$,1)="Y" AND (I<5 OR I>6) THE
  N PRINT A$(I); " "; V1$; "I"; A$(P);:GOTO 260
240 IF G=1 AND T=4 AND RIGHT$(V$,1)="Y" THEN PRINT A$(I); " "
  ; V1$; "I"; A$(P);:GOTO 260
250 PRINT A$(I); " "; V$; U$; A$(P);
260 IF I<>3 AND I<>7 THEN P=P+1
270 NEXT I
280 Y=Y+3:IF Y>24 THEN X=X+20:Y=2
290 NEXT T
300 A$=INPUT$(1)
310 GOTO 50
380 REM -----
390 REM LECTURE DES DATAS
400 FOR I=1 TO 4
410 READ T$(I)
420 NEXT I
430 FOR I=1 TO 56
440 READ A$(I)
450 NEXT I
460 RETURN
470 DATA PRESENT, IMPARFAIT, PASSE SIMPLE, FUTUR SIMPLE
480 DATA JE, TU, IL, ELLE, NOUS, VOUS, ILS, ELLES
490 DATA E, ES, E, ONS, EZ, ENT
500 DATA S, S, T, SSONS, SSEZ, SSENT
510 DATA AIS, AIS, AIT, IONS, IEZ, AIENT
520 DATA SSAIS, SSAIS, SSAIT, SSIONS, SSIEZ, SSAIENT
530 DATA AI, AS, A, AMES, ATES, ERENT
540 DATA S, S, T, MES, TES, RENT
550 DATA ERAI, ERAS, ERA, ERONS, EREZ, ERONT
560 DATA RAI, RAS, RA, RONS, REZ, RONT
570 RETURN

```

LE PROGRAMME

Lignes 30 à 110 : saisie du verbe

Le verbe à étudier est saisi dans la variable V\$. Si la terminaison à l'infinitif est différente de ER ou IR, on affiche le message "Verbe inconnu".

Dans le cas où la première lettre de celui-ci est une voyelle, on remplace le JE de A\$(1) par J'.

Lignes 120 à 310 : édition

Les verbes en YER prennent soit un I soit un Y dans leurs terminaisons, et cela suivant le temps étudié. On place donc, dans la variable V1\$, le verbe à éditer en enlevant le dernier caractère. Les verbes du deuxième groupe (G=2) comportant toujours un I pour premier caractère de leur terminaison, on place le caractère I dans la variable U\$.

On édite avec une boucle FOR/NEXT (variable T) le verbe au présent, à l'imparfait, au passé simple et au futur simple. Pour chaque temps, on calcule le pointeur P du tableau des terminaisons (ligne 180).

La ligne 220 tient compte des verbes en GER. On ajoute donc la lettre E devant la terminaison dans le cas où un "O" ou un "A" suivent le G (exemple : mangons → mangeons).

Les lignes 230 et 240 tiennent compte, elles, des verbes en YER. On utilise à ce moment-là la variable V1\$ citée plus haut.

Lignes 400 à 570 : lecture des DATAs

On place dans le tableau T\$ la liste des pronoms et dans A\$ celle des terminaisons qui se trouvent dans les DATAs des lignes 470 à 560.

LISTE DES VARIABLES :

A\$	dernière touche appuyée
A\$()	liste des terminaisons
G	groupe du verbe à étudier
I	indice de boucle FOR/NEXT
P	pointeur du tableau des terminaisons
T	indice de boucle FOR/NEXT (temps à éditer)
T\$()	liste des pronoms
U\$	U\$="I" si verbe du deuxième groupe
V\$	verbe à éditer
V1\$	verbe à éditer sans le dernier caractère
X\$	contient le 1 ^{er} caractère de la terminaison
Z\$	premier caractère du verbe à éditer
X et Y	coordonnées sur l'écran du verbe à éditer

SUITES LOGIQUES

Sans vouloir se comparer à un test poussé de QI, ce programme de suites logiques peut constituer un bon entraînement au raisonnement.

Il consiste à trouver le nombre découlant d'une suite. Pour cela, il propose trois nombres dans un ordre donné, obéissant à un algorithme utilisant l'addition, la soustraction ou la multiplication.

Exemple : suite 1, 14, 92


Vous devez donner votre première solution. Si elle est fausse, le MSX vous donne le quatrième élément de la suite et vous demande votre deuxième solution. Si elle est encore fausse, l'ordinateur affiche la "bonne solution".

Le jeu a trois niveaux de difficulté : il commence au niveau 1 et passe directement aux niveaux 2 puis 3 si votre première solution est bonne.

Si vous ne trouvez le bon résultat qu'au deuxième coup, le niveau de jeu ne change pas. Dans le cas où vous ne trouvez pas du tout la solution, le jeu redescendra automatiquement au niveau précédent.

Je n'explique pas ici la méthode utilisée pour programmer la suite logique ; cela rendrait le jeu trop facile ; les tricheurs peuvent toujours décortiquer le programme !

SUITES LOGIQUES



```

10 REM SUITES LOGIQUES
20 REM -----
30 SCREEN 0:COLOR 6,1,1:KEY OFF
40 N=1:A(1)=1
50 K=RND(-(TIME/10) MOD 100)
80 REM -----
90 REM LE JEU
100 I=INT(RND(1)*10)+1:J=INT(RND(1)*10)+1
110 K=INT(RND(1)*10)+1:L=INT(RND(1)*10)+1
120 E=E+1
130 ON N GOTO 210,200,300
180 REM -----
190 REM NIVEAU 1 & 2
200 IF K>5 THEN J=-J
210 FOR U=2 TO 5
220 A(U)=A(U-1)*I+J
230 NEXT U
240 GOTO 500
280 REM -----
290 REM NIVEAU 3
300 IF K>5 THEN J=-J
310 IF L>5 THEN 360
320 FOR U=2 TO 5
330 A(U)=A(U-1)*I+J
340 NEXT U
350 GOTO 500
360 FOR U=2 TO 5
370 A(U)=(A(U-1)+J)*I
380 NEXT U
480 REM -----
490 REM LA SUITE
500 FOR U=1 TO 2
510 CLS:PRINT"PARTIE: ";E,"NIVEAU: ";N:PRINT
520 FOR V=1 TO 2+U
530 PRINT "LE CHIFFRE";V;"EST ";A(V)
540 NEXT V:PRINT
550 PRINT "LE CHIFFRE";3+U;"EST ";:INPUT S
560 IF S=A(3+U) THEN 700
570 NEXT U
580 CLS:PRINT"LA SOLUTION: ":PRINT
590 FOR U=1 TO 5
600 PRINT U;"=",A(U)
610 NEXT U
620 P=P-1-N
630 IF N>1 THEN N=N-1
640 GOTO 800
680 REM -----
690 REM GAGNE
700 PRINT:PRINT "VOUS AVEZ TROUVE A L'ESSAI";U
710 IF U=2 THEN 750
720 P=P+1+N
730 IF N<3 THEN N=N+1
740 GOTO 800
750 P=P+N
780 REM -----
790 REM FIN
800 PRINT
810 PRINT"POINTS: ";P,"NB DE PARTIES: ";E
820 PRINT:PRINT"VOUS ETES MAINTENANT AU NIVEAU";N
830 IF INKEY$="" THEN 830
840 GOTO 100

```

LE PROGRAMME

Lignes 30 à 130 : initialisation

La suite logique étant placée dans le tableau A(1) à A(5), A(1) est mis dès le début du jeu à 1.

Les lignes 100 et 110 placent dans les variables I, J, K et L quatre nombres aléatoires qui détermineront la suite logique.

La ligne 130 renvoie sur les calculs de suites, suivant le niveau de jeu (variable N).

Lignes 200 à 380 : calcul de la suite logique

Ces lignes calculent pour chaque niveau respectif la suite logique correspondante (5 chiffres au total). Puis chacune de ces lignes renvoie en 500 pour afficher les trois premiers chiffres.

Lignes 500 à 610 : le jeu

Le programme saisit vos deux réponses par une boucle FOR/NEXT (variable U).

Si votre proposition est bonne, la ligne 560, après un test, envoie en ligne 700 pour un message de félicitations.

Dans le cas contraire, la même suite est affichée avec un chiffre supplémentaire.

Si vous n'avez pas trouvé, les lignes 580 à 610 affichent la bonne solution.


Lignes 620 à 840 : gagné-perdu

Les lignes 620 à 750 calculent votre nombre de points suivant le niveau et le nombre d'essais qu'il vous a fallu pour trouver la bonne solution. Si vous n'avez pas trouvé du tout, votre capital diminue de la même manière.

Les lignes 800 à 840 affichent les résultats de la dernière partie.

LISTE DES VARIABLES :

A()	contient la suite logique
E	nombre de parties
I, J, K et L	nombres aléatoires pour le calcul de la suite
N	niveau du jeu
P	nombre de points
S	solution proposée
U	indice de boucle FOR/NEXT
V	indice de boucle FOR/NEXT



PARTIE: 2 NIVEAU: 2

LE CHIFFRE 1 EST 1

LE CHIFFRE 2 EST 2


LE CHIFFRE 3 EST 7

LE CHIFFRE 4 EST ? 32

VOUS AVEZ TROUVE A L'ESSAI 1

POINTS: 5 NB DE PARTIES: 2

VOUS ETES MAINTENANT AU NIVEAU 3



THEME 3

CUISINE ET TRICOT

FICHER DE RECETTES

Comment préparer vos tournedos pour ce soir ? Quelle recette choisir pour utiliser les tomates de votre jardin ? Le mieux est de demander à votre MSX.

Le programme propose les options suivantes :

- **SAISIE** : vous devez entrer tous les renseignements d'une recette : nom, ingrédients et préparation.

Si vous ne voulez plus entrer de recette, tapez uniquement sur "RETURN" à la question "NOM:".

- **MODIFICATION** : si vous désirez modifier une recette, entrez son numéro (si vous avez oublié ce numéro, utilisez l'option 4 ou 5). Le programme vous demande alors le type de donnée que vous voulez modifier. Si vous n'avez plus de modification à apporter, tapez "Ø" en réponse à la question "NUMERO DE LA RECETTE:".


- **SUPPRESSION** : si vous désirez enlever une recette du fichier, cette option permet de la supprimer définitivement. Comme pour la modification, tapez "Ø" pour revenir au menu.

- **RECHERCHE** : cette partie recherche le mot que vous lui donnez dans une des trois données d'une recette. Cela vous permet de retrouver les recettes dont le nom contient le mot "Bœuf", ou les recettes utilisant comme ingrédient des "Tomates". Il n'est pas nécessaire de donner un nom entier : le mot "Bœuf" suffit à retrouver la recette du "Bœuf en Daube", du "Bœuf Bourguignon" ou du "Bœuf à la Flamande".

- **LISTING** : édite simplement sur l'écran le fichier dans son intégralité.

- **SAUVEGARDE** : le programme sauve le fichier sur cassette ou sur disquette.

FICHIER DE RECETTES



```

10 REM FICHER DE RECETTES
20 REM -----
30 CLEAR 5000
40 DIM D$(3,100)
50 L$(1)="NOM":L$(2)="INGREDIENTS":L$(3)="RECETTE"
60 SCREEN 0
70 INPUT"PREMIERE UTILISATION (O/N)";R$
80 IF R$="0" THEN 200
90 CLS:PRINT"METTEZ LE MAGNETO K7 EN LECTURE"
100 OPEN "RECETTE" FOR INPUT AS #1
110 INPUT#1,N
120 FOR I=1 TO N
130 FOR J=1 TO 3
140 LINE INPUT#1,D$(J,I)
150 NEXT J,I
160 CLOSE
180 REM -----
190 REM MENU
200 CLS
210 PRINT"NB DE RECETTES: ";N:PRINT:PRINT
220 PRINT"1 SAISIE"
230 PRINT"2 MODIFICATION"
240 PRINT"3 SUPPRESSION"
250 PRINT"4 RECHERCHE"
260 PRINT"5 LISTING"
270 PRINT"6 SAUVEGARDE"
280 PRINT:PRINT"VOTRE CHOIX ?";
290 R$=INPUT$(1):R=VAL(R$)
300 IF R<1 OR R>6 THEN 290
310 ON R GOTO 400,900,1100,600,1300,1500
380 REM -----
390 REM SAISIE D'UNE RECETTE
400 CLS:PRINT"SAISIE D'UNE RECETTE":PRINT
410 IF N=100 THEN PRINT"FICHER PLEIN";:R$=INPUT$(1):GOTO 20
420 N=N+1
430 FOR I=1 TO 3
440 PRINT L$(I);": ";:LINE INPUT D$(I,N)
450 IF I=1 AND D$(1,N)="" THEN N=N-1:GOTO 200
460 NEXT I
470 GOTO 400
580 REM -----
590 REM RECHERCHE
600 CLS:F=0
610 PRINT"RECHERCHE PAR : "
620 FOR I=1 TO 3:PRINT I;L$(I):NEXT
630 PRINT:PRINT"QUEL TYPE DE DONNEE ?":PRINT
640 R$="":R$=INPUT$(1):R=VAL(R$)
650 IF R<1 OR R>3 THEN 200
660 PRINT L$(R);": ";:LINE INPUT D$:PRINT
670 FOR I=1 TO N
680 IF INSTR(D$(R,I),D$)=0 THEN 740
690 F=1:PRINT USING"### ";I;:PRINT"*";D$(1,I);"*"
700 FOR J=2 TO 3
710 FOR K=1 TO LEN(D$(J,I)) STEP 32
720 PRINT " ";MID$(D$(J,I),K,32)
730 NEXT K,J:PRINT
740 NEXT I:PRINT
750 IF F=0 THEN PRINT"JE N'AI RIEN TROUVE"
760 R$=INPUT$(1):GOTO 600
880 REM -----
890 REM MODIFICATION
900 CLS:PRINT"MODIFICATION":PRINT

```

```

910 NU=0:INPUT"NUMERO DE LA RECETTE";NU
920 IF NU<1 OR NU>N THEN 200
930 FOR I=1 TO 3
940 PRINT I;L$(I);": ";D$(I,NU)
950 NEXT I
960 R=0:PRINT:INPUT"MODIFICATION DU";R
970 IF R<1 OR R>3 THEN 200
980 PRINT L$(R);": ";:LINE INPUT D$(R,NU)
990 GOTO 900
1080 REM -----
1090 REM SUPPRESSION
1100 CLS:PRINT"SUPPRESSION":PRINT
1110 NU=0:INPUT"NUMERO DE LA RECETTE";NU
1120 IF NU<1 OR NU>N THEN 200
1130 PRINT
1140 FOR I=1 TO 3
1150 PRINT I;L$(I);": ";D$(I,NU)
1160 NEXT I:PRINT
1170 R$="":INPUT"VOUS ETES SUR DE SUPPRIMER CETTE REC
ETTE";R$
1180 IF R$<>"O" THEN 1100
1190 N=N-1
1200 FOR I=NU TO N
1210 FOR J=1 TO 3
1220 D$(J,I)=D$(J,I+1)
1230 NEXT J,I
1240 GOTO 1100
1280 REM -----
1290 REM LISTING
1300 CLS:F=0
1310 FOR I=1 TO N
1320 PRINT USING"### ";I;:PRINT"*";D$(1,I);"*":F=F+1
1330 FOR J=2 TO 3
1340 FOR K=1 TO LEN(D$(J,I)) STEP 32
1350 PRINT " ";MID$(D$(J,I),K,32):F=F+1
1360 NEXT K,J:PRINT:F=F+1
1370 IF F>15 THEN PRINT:PRINT"APPUYEZ SUR UNE TOUCHE POUR LA
SUITE";:R$=INPUT$(1):CLS:F=0
1380 NEXT I
1390 PRINT:PRINT"APPUYEZ SUR UNE TOUCHE POUR LE MENU";
1400 R$=INPUT$(1):GOTO 200
1480 REM -----
1490 REM SAUVEGARDE
1500 CLS:PRINT"APPUYEZ SUR RETURN QUAND LE MAGNETO K7 EST P
RET POUR L'ENREGISTREMENT"
1510 LINE INPUT R$
1520 OPEN "RECETTE" FOR OUTPUT AS #1
1530 PRINT#1,N
1540 FOR I=1 TO N
1550 FOR J=1 TO 3
1560 PRINT#1,D$(J,I)
1570 NEXT J,I
1580 CLOSE
1590 CLS:PRINT"A LA PROCHAINE !"
1600 END

```

LE PROGRAMME

Lignes 30 à 160 : initialisation

On place dans le tableau L\$() les libellés des trois données nécessaires à l'exploitation du fichier.

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 200 à 310 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 400 à 470 : saisie d'une opération

On saisit par une boucle FOR/NEXT les trois données du fichier. Si le nom de la recette est une chaîne vide, on revient au menu (ligne 450).

Lignes 600 à 760 : recherche

Le programme demande d'abord sur quel type de donnée on désire effectuer la recherche, puis la donnée elle-même (variable D\$).

Les lignes 670 à 740 parcourent par une boucle FOR/NEXT l'intégralité du fichier, et affichent les enregistrements où l'on a trouvé la chaîne D\$.

Lignes 900 à 990 : modification

Ces lignes modifient l'enregistrement choisi par l'utilisateur (variable NU).

Lignes 1100 à 1240 : suppression

On supprime du fichier l'enregistrement choisi par l'utilisateur (variable NU). Après l'avoir affiché, le programme demande confirmation. Les lignes 1200 à 1230 décalent le tableau vers le bas.

Lignes 1300 à 1400 : listing

On affiche par une boucle FOR/NEXT l'intégralité du fichier. Le programme est stoppé par la ligne 1370 dès que l'écran est plein.

Lignes 1500 à 1600 : sauvegarde

Ces lignes sauvent sur cassette ou sur disquette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

D\$	donnée à rechercher
D\$(,)	tableau des données du fichier
F	drapeau. Si F=0, aucun enregistrement n'a été trouvé.
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
K	indice de boucle FOR/NEXT
L\$()	libellés des données
N	nombre de recettes dans le fichier
NU	numéro d'enregistrement choisi par l'utilisateur
R\$	choix de l'utilisateur
R	valeur numérique de R\$

NOM: TOURNEDOS

- 1 *TOURNEDOS*
4 TOURNEDOS, 4 CROUTOS, 100G DE
BEURRE
FAIRE DORER LES CROUTONS DANS LA
MOITIE DU BEURRE. FAIRE GRILLER
LES TOURNEDOS DANS LE RESTE DU
BEURRE. CUIRE DE CHAQUE COTE 5MN
. SALER, PLACER CHAQUE TOURNEDOS
SUR UN CROUTON AVANT DE SERVIR.
- 2 *TOURNEDOS A LA LIMOUSINE*
4 TOURNEDOS, 4 BEAUX CEPES, 4 CR
OUTONS RISSOLES AU BEURRE.
FAIRE CUIRE 5 MN DE CHAQUE COTE
ET ENLEVER. FAIRE FRIRE LES CEPE
S DANS L'HUILE. ASSAISONNER D'AI
L, ECHALOTTES, SEL ET POIVRE. SE
RVIR EN COURONNE TOURNEDOS, CROU
TONS ET CEPES.

RECETTES DE CUISINE

Voici le moyen infallible de faire des gâteaux extraordinaires tous différents les uns des autres.

Très simple ! Vous tapez RUN et en quelques secondes votre ordinateur affiche à l'écran une recette inédite de gâteau au chocolat et aux noix.


Vous n'avez plus qu'à réunir tous les ingrédients dont la liste vous est donnée et à suivre les instructions de fabrication.

Dès que votre gâteau sera cuit, vous pourrez le déguster avec la boisson choisie par le MSX, vin ou cidre...


Un petit conseil cependant : pour déjouer les farces éventuelles de votre ordinateur, vérifiez que la combinaison de tous les ingrédients de la recette soit vraisemblable.

De toute façon, il suffit d'appuyer sur n'importe quelle touche pour obtenir une nouvelle recette.

Bon appétit !



RECETTE DE CUISINE



```

10 REM CUISINE
20 REM -----
30 DIM C$(24), Q(24), U(24), A(24)
40 DATA FARINE, 100, 2, 3, SUCRE, 50, 2, 2, EAU, 10, 4, 2, LAIT, 10, 4, 2, 0
EUF, 1, 1, 3
50 DATA VANILLE, .5, 5, 2, BLANC OEUF, 1, 1, 3, HUILE, 1, 3, 2, BEURRE, 5
0, 2, 2, LEVURE, .5, 5, 2
60 DATA RAISINS S., 50, 2, 2, RHUM, .5, 6, 1, COGNAC, .5, 6, 1, NOIX, 50,
2, 2, NOISETTES, 50, 2, 2
70 DATA CHOCOLAT, 50, 2, 2, JAUNE OEUF, 1, 1, 2, OEUF NEIGE, 1, 1, 3, PO
MMES, 200, 2, 2
80 DATA CERISES, 100, 2, 2, SEL, 1, 7, 2, AMANDES, 50, 2, 2, FRUITS C., 5
0, 2, 2, CREME, 25, 2, 2
90 DATA , GRAM., CUIL., CLS, SACHET, VERRE, PINC.
100 DATA VIN ROUGE, VIN BLANC, CIDRE, CHAMPAGNE, JUS D'ORANGE, VI
N CUIT
110 A=RND(-(TIME/100) MOD 100)
180 REM -----
190 REM LECTURE INGREDIENTS
200 FOR I=1 TO 24
210 READ C$(I), Q(I), U(I), A(I)
220 NEXT I
230 FOR I=1 TO 7
240 READ U$(I)
250 NEXT I
260 FOR I=1 TO 6
270 READ B$(I)
280 NEXT I
380 REM -----
390 REM ELEMENTS BASE
400 SCREEN 0:PRINT:PRINT TAB(14); "RECETTE":PRINT:PRINT
410 NE=1:GOSUB 1000
420 NE=2:GOSUB 1000
430 NL=INT(RND(1)*2)+1
440 FOR I=1 TO NL
450 NE=INT(RND(1)*3)+3
460 IF I=1 THEN 480
470 IF NE=L(1) THEN 450
480 GOSUB 1000
490 L(I)=NE
500 NEXT I
580 REM -----
590 REM ELEMENTS GATEAU
600 NS=INT(RND(1)*3)+3
610 FOR I=1 TO NS
620 NE=INT(RND(1)*19)+6
630 IF I=1 THEN 670
640 FOR J=1 TO I-1
650 IF NE=S(J) THEN 620
660 NEXT J
670 GOSUB 1000
680 S(I)=NE
690 NEXT I
780 REM -----
790 REM RECETTE
800 PRINT:PRINT:PRINT "MELANGER FARINE, SUCRE";
810 IF NL=1 THEN 840
820 PRINT ", "; C$(L(1)); " ET "; C$(L(2));
830 GOTO 850
840 PRINT " ET "; C$(L(1));
850 PRINT " POUR OBTENIR LA PATE. ";
860 PRINT "AJOUTER "; C$(S(1));
870 FOR I=2 TO NS-1

```

```

880 PRINT",",C$(S(I));
890 NEXT I
900 PRINT" ET ",C$(S(NS));".";
910 PRINT"BEURRER UN MOULE ET Y METTRE LA PATE OBTENUE.";
920 PRINT"PLACER CE MOULE DANS UN FOUR A",100+(10*INT(RND(1)
*6));"DEGRES.";
930 PRINT"LAISSER CUIRE",30+(5*INT(RND(1)*7));"MINUTES.";
940 PRINT"DEGUSTER CE GATEAU AVEC DU ",B$(INT(RND(1)*6)+1);"
.";
950 A$=INPUT$(1)
960 GOTO 400
980 REM -----
990 REM CALCUL INGREDIENTS
1000 QU=Q(NE)*INT(RND(1)*A(NE)+1)
1010 IF QU=.5 THEN 1040
1020 PRINT TAB(5);C$(NE);TAB(21);": ";QU;U$(U(NE))
1030 RETURN
1040 PRINT TAB(5);C$(NE);TAB(21);": 1/2 ";U$(U(NE))
1050 RETURN

```

LE PROGRAMME

Lignes 30 à 280 : lecture des ingrédients

Tous les éléments des DATAs des lignes 40 à 100 sont placés dans leurs tableaux respectifs (voir identification des variables).

Lignes 400 à 500 : éléments de base

La farine et le sucre sont les deux éléments indispensables à la pâte. On appelle donc toujours le sous-programme des lignes 1000 à 1050 pour ces deux éléments.

On ajoute, pour compléter cette pâte, un ou deux éléments liquides. Le nombre de liquides est déterminé par la variable NL. Dans le cas où deux éléments liquides sont utilisés, on vérifie qu'il ne s'agit pas deux fois des mêmes.

Lignes 600 à 690 : éléments du gâteau

Puis c'est au tour des éléments solides. On utilise le même procédé que pour les liquides. Chaque solide est placé dans le tableau S pour la vérification des doublons. Ce tableau sera aussi utilisé pour l'édition du texte de la recette (lignes 860 à 880).

Lignes 800 à 960 : recette

Chaque élément liquide et solide est repris dans le texte de la recette. Deux nombres aléatoires donnent la température et le temps de cuisson du gâteau. Puis on affiche un des éléments du tableau des boissons.

Lignes 1000 à 1050 : sous-programme

La variable NE correspond au numéro de l'élément à éditer. La quantité à utiliser est comprise entre une base (tableau Q) et cette base multipliée par le nombre aléatoire maximum (tableau A). Dans le cas où l'on obtient la valeur 0.5, la ligne 1040 affiche le message "1/2".

LISTE DES VARIABLES :

A()	valeur maximale aléatoire pour chaque ingrédient
AS	dernière touche appuyée
B\$()	liste des boissons
C\$()	liste des composants
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
L()	liste des éléments liquides
NE	numéro de l'élément à éditer
NL	nombre de liquides
NS	nombre de solides
Q()	liste des quantités de base de chaque élément
QU	quantité à utiliser
S()	liste des solides
U()	liste des unités
U\$()	nom des unités

RECETTE

FARINE	:	200 GRAM.
SUCRE	:	100 GRAM.
EAU	:	10 CLS
OEUFS	:	3
POMMES	:	200 GRAM.
CREME	:	25 GRAM.
NOIX	:	100 GRAM.

MELANGER FARINE, SUCRE, EAU ET OEUFS POUR OBTENIR LA PATE. AJOUTER POMMES, CREME ET NOIX. BEURRER UN MOULE ET Y METTRE LA PATE OBTENUE. PLACER CE MOULE DANS UN FOUR A 150 DEGRES. LAISSER CUIRE 45 MINUTES. DEGUSTER CE GATEAU AVEC DU CIDRE.

MENUS ALEATOIRES


Si vous manquez d'idées de menu pour le repas de ce soir, le MSX vous propose une infinité de combinaisons, composées bien sûr de vos plats préférés.

Nul besoin de longues explications. Exécutez le programme, un menu s'affiche automatiquement. Il suffit d'appuyer sur "RETURN" pour obtenir d'autres repas.

Bon appétit !



MENUS ALEATOIRES



```

10 REM MENUS ALEATOIRES
20 REM -----
30 SCREEN 0
40 DIM E$(158)
50 E=RND(-(TIME/10) MOD 100)
60 FOR I=1 TO 158
70 READ E$(I)
80 NEXT I
180 REM -----
190 REM LES MENUS
200 CLS:PRINT
210 PRINT TAB(14); "MENU DU JOUR"
220 LOCATE 16,6:PRINT "ENTREE:"
230 E=INT(RND(1)*46)+1
240 LOCATE 19.5-LEN(E$(E))/2,8:PRINT E$(E)
250 LOCATE 17,11:PRINT "PLAT:"
260 E=INT(RND(1)*76)+47
270 IF RIGHT$(E$(E),1)="*" THEN LOCATE 20-LEN(E$(E))/2,13:PR
INT LEFT$(E$(E),LEN(E$(E))-1):GOTO 310
280 LOCATE 19.5-LEN(E$(E))/2,13:PRINT E$(E)
290 E=INT(RND(1)*12)+123
300 LOCATE 19.5-LEN(E$(E))/2,14:PRINT E$(E)
310 LOCATE 16,17:PRINT "DESSERT:"
320 E=INT(RND(1)*24)+135
330 LOCATE 19.5-LEN(E$(E))/2,19:PRINT E$(E)
340 IF INKEY$="" THEN 340
350 GOTO 200
980 REM -----
990 REM LES ENTREES
1000 DATA SOUPE JULIENNE, SOUPE A L'OIGNON, SOUPE POIREAUX/POM
MES DE TERRE
1010 DATA SARDINES, SAUMON, MAQUEREAUX VIN BLANC, HARENGS, THON
MAYONNAISE, COQUILLES ST JACQUES
1020 DATA PATE DE CAMPAGNE, PATE DE FOIE, PATE DE CANARD, TERRI
NE DE LIEVRE, RILLETES
1030 DATA JAMBON CRU, JAMBON CUIT, MUSEAU VINAIGRETTE, SAUCISSO
N SEC, SAUCISSON A L'AIL, ANDOUILLE
1040 DATA SALADE MIXTE, SALADE DE CONCOMBRES, SALADE DE MAIS, O
EUFS EN GELEE, OEUFS MAYONNAISE, OEUFS A LA COQUE, SALADE DE TO
MATES, MELON PORTO, RADIS, RADIS NOIR, ARTICHAUTS VINAIGRETTE
1050 DATA CHOU ROUGE, CHOU-FLEUR, BOUCHEES A LA REINE, QUICHE L
ORRAINE, TARTE AUX OIGNONS, PIZZA, CREPES AU JAMBON, CREPES AUX
CHAMPIGNONS
1060 DATA ASPERGES VINAIGRETTE, SALADE D'ENDIVES, POIREAUX VIN
AIGRETTE, CAROTTES RAPEES, FONDS D'ARTICHAUTS, LENTILLES VINAIG
RETTE, CHAMPIGNONS A LA GRECQUE
1070 REM LES PLATS
1080 DATA BIFTECK FRITES*, BOEUF BOURGUIGNON*, BOEUF EN DAUBE*
, BIFTECK HACHE PUREE*, POT-AU-FEU*, ENTRECOTE FRITES*, TOURNEDO
S
1090 DATA FOIE GRILLE PUREE*, LANGUE DE BOEUF JARDINIERE*, ENT
RECOTES, GRAS-DOUBLE PETITS POIS*
1100 DATA ESCALOPES CHAMPIGNONS*, ESCALOPES A LA CREME*, COTEL
ETTES DE VEAU, ROTI DE VEAU, VEAU MARENGO*, RIS DE VEAU*, ROGNON
S DE VEAU PUREE*
1110 DATA TETE DE VEAU, FOIE DE VEAU
1120 DATA COTELETTES DE MOUTON, EPAULE DE MOUTON, GIGOT DE MOU
TON FLAGEOLETS*, RAGOUT DE MOUTON*, NAVARIN*
1130 DATA BROCHETTES D'AGNEAU RIZ*, EPAULE D'AGNEAU, BLANQUETT
E DE VEAU*
1140 DATA COTELETTES DE PORC, ROTI DE PORC, JAMBONNEAU, ANDOUIL
LETES FRITES*, BOUDIN, BOUDIN BLANC, PIEDS DE PORC, SAUCISSES A
U CHOU*

```

1150 DATA POULET ROTI, POULET CURRY RIZ*, POULE AU POT, POULE A
U RIZ*, COQ AU VIN*, OIE ROTIE, CONFIT D'OIE*
1160 DATA DINDE AUX MARRONS*, ROTI DE DINDONNEAU, CANARD A L'O
RANGE*, CANETONS ROTIS
1170 DATA CIVET DE LIEVRE, CIVET DE LAPIN
1180 DATA FILET DE BARBUE, CABILLAUD FRIT, DAURADE GRILLEE, LIM
ANDE, LOTTE, MERLANS FRITS, BRANDADE DE MORUE*, RAIE BEURRE NOIR
1190 DATA FILETS DE SOLE, SOLE MEUNIERE, TURBOT, MOULES MARINIE
RE RIZ*, TRUITES AUX AMANDES*
1200 DATA QUENELLES DE POISSON*, ESCARGOTS DE BOURGOGNE*, CUIS
SES DE GRENOUILLES*
1210 DATA TOMATES FARCIES*, CHOUROUTE GARNIE*, CASSOULET*, AUB
ERGINES FARCIES*, COUSCOUS*, PAELLA*
1220 DATA OMELETTE AUX CHAMPIGNONS*, OMELETTE AUX POMMES DE T
ERRE*, ENDIVES AU JAMBON*, LASAGNES*, CANNELLONIS*
1230 REM LEGUMES
1240 DATA PUREE, POMMES DE TERRE, HARICOTS VERTS, PETITS POIS, T
OMATES PROVENCALES, EPINARDS, SALSIFIS, LENTILLES
1250 DATA PATES, SPAGHETTIS, NOUILLES, RIZ
1260 REM DESSERT
1270 DATA GATEAU DE RIZ, BEIGNETS AUX POMMES, CREME AU CARAMEL
, MOUSSE AU CHOCOLAT, PUDDING
1280 DATA CHARLOTTE, FLAN, CREPES, CAKE, CLAFOUTIS, TARTE, PAIN PE
RDU, QUATRE-QUARTS, CREME GLACEE, PATISSERIE
1290 DATA BANANES FLAMBEES, MELON, SALADE DE FRUITS, GATEAU MAI
SON, FRUITS DE SAISON
1300 DATA FROMAGE, CEMBERT, FROMAGE BLANC, YAOURT

LE PROGRAMME

Lignes 30 à 80 : initialisation

Le programme lit dans les DATAs des lignes 1000 à 1300 tous les éléments qui vont composer vos menus. Cette liste n'est pas exhaustive. Le lecteur peut ajouter des éléments dans les DATAs entre les lignes 1060 et 1070 pour les entrées, entre les lignes 1220 et 1230 pour les plats, entre les lignes 1250 et 1260 pour les légumes et après la ligne 1300 pour les desserts. Il faut bien sûr modifier le nombre 158 de la ligne 60 (il y a 158 éléments pour l'instant) mais aussi la ligne 230 (il y a 46 entrées, la première est en 1), la ligne 260 (il y a 76 plats, le premier est en 47), la ligne 290 (il y a 12 légumes, le premier est en 123) et pour finir la ligne 320 (il y a 24 desserts, le premier est en 135).

Tous les plats qui n'ont pas besoin d'un légume doivent être suivis d'une étoile.

Lignes 200 à 350 : les menus

Le programme affiche quatre types d'éléments au hasard. Si le nom du plat se termine par une étoile, on ne tient pas compte des légumes pour éviter les "couscous nouilles" et les "choucroutes garnies spaghettis".

LISTE DES VARIABLES :

E	pointe l'élément choisi au hasard
ES()	liste des éléments
I	indice de boucle FOR/NEXT

GESTION D'UN CONGELATEUR

Vous avez du mal à gérer le contenu de votre congélateur ? Le MSX va vous donner un petit coup de main !

Il vous demande en priorité si le fichier a déjà été créé. Si c'est le cas, tapez uniquement sur la touche "RETURN" pour relire les données. Puis le programme affiche un menu pour les options suivantes :

- 1 - **SAISIE** : vous devez entrer tous les renseignements sur l'aliment que vous venez de congeler. Les dates doivent être saisies sous la forme année, mois et jour pour faciliter les recherches de l'option 4, et comporter 6 chiffres (exemple : le 09/03/84 s'écrit ici 840309).

Si vous n'avez plus rien à rentrer, appuyez sur "RETURN" à la question "Date de congélation :".

- 2 - **MODIFICATION** : si vous désirez modifier un élément, entrez son numéro (si vous avez oublié ce numéro, utilisez l'option 4 ou 5). Le programme vous demande alors le type de donnée que vous voulez modifier. Si vous n'avez plus de modification à ajouter, tapez "Ø" en réponse à la question "Numéro de l'élément :".

- 3 - **SUPPRESSION** : dès qu'un aliment est sorti du congélateur, vous pouvez le supprimer par cette option.


Comme pour la modification, tapez Ø pour revenir au menu.

- 4 - **RECHERCHE** : elle permet d'afficher tous les aliments qui seront périmés entre deux dates précises, de connaître toutes les sortes de crêpes qui vous attendent impatiemment dans votre congélateur, ou tous les petits plats que vous y avez mis durant l'année 84. Le simple mot "Crêpe" permet de retrouver aussi bien les crêpes au jambon que celles aux champignons.

- 5 - **LISTING** : édite simplement sur l'écran le fichier dans son intégralité.

- 6 - **SAUVEGARDE** : le programme sauve le fichier sur cassette ou sur disquette.

GESTION DU CONGELATEUR



```

10 REM GESTION DU CONGELATEUR
20 REM -----
30 CLEAR 2000:KEY OFF
40 DIM D$(3,100)
50 L$(1)="DATE DE CONGELATION(AAMMJJ)":L$(2)="DATE DE PEREMP
TION (AAMMJJ)":L$(3)="NATURE DU PRODUIT"
60 SCREEN 0
70 INPUT"PREMIERE UTILISATION (O/N)";R$
80 IF R$="O" THEN 200
90 CLS:PRINT"METTEZ LE MAGNETO K7 EN LECTURE"
100 OPEN "CONGEL" FOR INPUT AS #1
110 INPUT#1,N
120 FOR I=1 TO N
130 FOR J=1 TO 3
140 LINE INPUT#1,D$(J,I)
150 NEXT J,I
160 CLOSE
180 REM -----
190 REM MENU
200 CLS
210 PRINT"NB D'ELEMENTS: ";N:PRINT:PRINT
220 PRINT"1 SAISIE"
230 PRINT"2 MODIFICATION"
240 PRINT"3 SUPPRESSION"
250 PRINT"4 RECHERCHE"
260 PRINT"5 LISTING"
270 PRINT"6 SAUVEGARDE"
280 PRINT:PRINT"VOTRE CHOIX ?";
290 R$=INPUT$(1):R=VAL(R$)
300 IF R<1 OR R>6 THEN 290
310 ON R GOTO 400,900,1100,600,1300,1500
380 REM -----
390 REM SAISIE D'UN ELEMENT
400 CLS:PRINT"SAISIE D'UN ELEMENT":PRINT
410 IF N=100 THEN PRINT"FICHIER PLEIN";:R$=INPUT$(1):GOTO 200
420 N=N+1
430 FOR I=1 TO 3
440 PRINT L$(I);": ";:LINE INPUT D$(I,N)
450 IF I=1 AND VAL(D$(1,N))=0 THEN N=N-1:GOTO 200
460 IF I<3 AND LEN(D$(I,N))<>6 THEN 440
470 NEXT I
480 GOTO 400
580 REM -----
590 REM RECHERCHE
600 CLS:F=0
610 PRINT"RECHERCHE PAR : "
620 FOR I=1 TO 3:PRINT I;L$(I):NEXT
630 PRINT:PRINT"QUEL TYPE DE DONNEE ?":PRINT
640 R$=INPUT$(1):R=VAL(R$)
650 IF R<1 OR R>3 THEN 200
660 IF R=3 THEN 700
670 INPUT"DATE DEBUT (AAMMJJ) ";DD
680 INPUT"DATE FIN (AAMMJJ) ";DF
690 GOTO 710
700 PRINT L$(R);": ";:LINE INPUT D$:PRINT
710 PRINT
720 FOR I=1 TO N
730 IF R=3 THEN IF INSTR(D$(R,I),D$)=0 THEN 800 ELSE 750
740 IF VAL(D$(R,I))<DD OR VAL(D$(R,I))>DF THEN 800
750 F=1:PRINT USING"### ";I;
760 FOR J=1 TO LEN(D$(3,I)) STEP 35
770 PRINT TAB(4);MID$(D$(3,I),J,35)
780 NEXT J

```

```

790 PRINT "   CONGELE LE ";D$(1,I);" PERIME LE ";D$(2,I);:PR
INT
800 NEXT I:PRINT
810 IF F=0 THEN PRINT"JE N'AI RIEN TROUVE"
820 R$=INPUT$(1):GOTO 600
880 REM -----
890 REM MODIFICATION
900 CLS:PRINT"MODIFICATION":PRINT
910 INPUT"NUMERO DE L'ELEMENT";NU
920 IF NU<1 OR NU>N THEN 200
930 FOR I=1 TO 3
940 PRINT I;L$(I);": ";D$(I,NU)
950 NEXT I
960 PRINT:INPUT"MODIFICATION DU";R
970 IF R<1 OR R>3 THEN 200
980 PRINT L$(R);": ";:LINE INPUT D$(R,NU)
990 GOTO 900
1080 REM -----
1090 REM SUPPRESSION
1100 CLS:PRINT"SUPPRESSION":PRINT
1110 INPUT"NUMERO DE L'ELEMENT";NU
1120 IF NU<1 OR NU>N THEN 200
1130 PRINT
1140 FOR I=1 TO 3
1150 PRINT I;L$(I);": ";D$(I,NU)
1160 NEXT I:PRINT
1170 INPUT"VOUS ETES SUR DE SUPPRIMER CET          ELEMENT";
R$
1180 IF R$<>"O" THEN 1100
1190 N=N-1
1200 FOR I=NU TO N
1210 FOR J=1 TO 3
1220 D$(J,I)=D$(J,I+1)
1230 NEXT J,I
1240 GOTO 1100
1280 REM -----
1290 REM LISTING
1300 CLS:F=0
1310 FOR I=1 TO N
1320 PRINT USING"###";I;
1330 FOR J=1 TO LEN(D$(3,I)) STEP 35
1340 PRINT TAB(4);MID$(D$(3,I),J,35):F=F+1
1350 NEXT J
1360 PRINT "   CONGELE LE ";D$(1,I);" PERIME LE ";D$(2,I);:P
RINT:F=F+2
1370 IF F>16 THEN PRINT:PRINT" APPUYEZ SUR UNE TOUCHE POUR L
A SUITE";:R$=INPUT$(1):CLS:F=0
1380 NEXT I
1390 PRINT:PRINT" APPUYEZ SUR UNE TOUCHE POUR LE MENU"
1400 R$=INPUT$(1):GOTO 200
1480 REM -----
1490 REM SAUVEGARDE
1500 CLS:PRINT"APPUYEZ SUR RETURN QUAND LE MAGNETO K7 EST P
RET POUR L'ENREGISTREMENT"
1510 LINE INPUT R$
1520 OPEN "CONGEL" FOR OUTPUT AS #1
1530 PRINT#1,N
1540 FOR I=1 TO N
1550 FOR J=1 TO 3
1560 PRINT#1,D$(J,I)
1570 NEXT J,I
1580 CLOSE
1590 CLS:PRINT"A LA PROCHAINE CONGELATION !"
1600 END

```

LE PROGRAMME

Lignes 30 à 160 : initialisation

On place dans le tableau L\$() les libellés des trois données nécessaires à l'exploitation du fichier.

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 200 à 310 : menu

Ces lignes dispatchent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 400 à 480 : saisie d'une opération

On saisit par une boucle FOR/NEXT les trois données du fichier. Si aucune date de congélation n'a été entrée, on revient au menu (ligne 450).

Lignes 600 à 820 : recherche

Le programme demande d'abord sur quel type de donnée on désire effectuer la recherche, puis la donnée elle-même (variable D\$ pour la nature du produit, et date de début DD et date de fin DF pour une période).

Les lignes 720 à 800 parcourent par une boucle FOR/NEXT l'intégralité du fichier, et affichent les enregistrements où l'on a trouvé la chaîne D\$ ou comprise entre les deux dates DD et DF.

Lignes 900 à 990 : modification

Ces lignes modifient l'enregistrement choisi par l'utilisateur (variable NU).

Lignes 1100 à 1240 : suppression

On supprime du fichier l'enregistrement choisi par l'utilisateur (variable NU). Après l'avoir affiché, le programme demande confirmation. Les lignes 1200 à 1230 décalent le tableau vers le bas.

Lignes 1300 à 1400 : listing

On affiche par une boucle FOR/NEXT l'intégralité du fichier. Le programme est stoppé par la ligne 1370 dès que l'écran est plein.

Lignes 1500 à 1600 : sauvegarde

Ces lignes sauvent sur cassette ou sur disquette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

D\$	nature du produit à rechercher
D\$(,)	tableau des données du fichier
DD	date de début des recherches
DF	date de fin des recherches
F	drapeau. Si F=0, aucun enregistrement n'a été trouvé.
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
L\$()	libellés des données
N	nombre de produits dans le fichier
NU	numéro d'enregistrement choisi par l'utilisateur
RS	choix de l'utilisateur
R	valeur numérique de RS

RECHERCHE PAR :

- 1 DATE DE CONGELATION(AAMMJJ)
- 2 DATE DE PEREMPTION (AAMMJJ)
- 3 NATURE DU PRODUIT

QUEL TYPE DE DONNEE ?

DATE DEBUT (AAMMJJ) ? 840101
DATE FIN (AAMMJJ) ? 841231

1 HARICOTS VERTS
CONGELE LE 841212 PERIME LE 850712

2 CREPES AU JAMBON
CONGELE LE 840901 PERIME LE 850901

3 DINDE AUX MARRONS
CONGELE LE 840615 PERIME LE 841015

TRICOT

Indispensable à toute tricoteuse, ce petit programme permet de connaître rapidement et sans calcul le nombre de mailles ou de rangs à monter et à tricoter, et d'adapter ainsi à toutes les tailles les explications des tricotés aux mesures standardisées.

Il suffit de rentrer le nombre de mailles et de rangs d'un échantillon tricoté de 10 cm sur 10 cm, les mesures diverses du dos et du devant (largeur à la base, largeur après diminutions des emmanchures, hauteur sous les bras, hauteur totale) et les mesures des manches (largeur poignet, largeur à l'épaule, hauteur à l'épaule, hauteur totale).

Dès que tous ces renseignements sont entrés dans votre MSX, le programme affiche le nombre de rangs à monter et le nombre de mailles à tricoter de toutes les parties du tricot.

MONTAGE DOS OU DEVANT

- MONTER 124 MAILLES, SOIT 48 CMS.
- TRICOTER 136 RANGS, SOIT 40 CMS.
- A 40 CMS, RABATTRE DE CHAQUE COTE 10 MAILLES, EN PLUSIEURS FOIS, POUR LES EMMANCHURES.
- TRICOTER ENCORE 61 RANGS, SOIT 18 CMS.
- A 58 CMS, FERMER SELON L'ENCOLURE.

MONTAGE DES MANCHES

- MONTER 57 MAILLES, SOIT 22 CMS.
- TRICOTER 146 RANGS, SOIT 43 CMS. EN AUGMENTANT REGULIEREMENT CHAQUE COTE DE 18 MAILLES.
- A 43 CMS, RABATTRE LES 93 MAILLES RESTANTES REGULIEREMENT, SOIT 46 MAILLES DE CHAQUE COTE.
- A 56 CMS, FERMER LE TRICOT.

TRICOT




```

10 REM TRICOT
20 REM -----
30 SCREEN 0:PRINT"ECHANTILLON 10 CM SUR 10 CM":PRINT
40 INPUT"NOMBRE DE MAILLES";M1
50 INPUT"NOMBRE DE RANGS";R1
60 M2=M1/10
70 R2=R1/10
80 REM -----
90 REM MESURES DEVANT ET DOS
100 CLS:PRINT"MESURES DEVANT ET DOS":PRINT
110 INPUT"LARGEUR DOS/DEVANT A LA BASE";L1
120 INPUT"LARGEUR POITRINE";L2
130 INPUT"HAUTEUR SOUS LES BRAS";H1
140 INPUT"HAUTEUR TOTALE DOS/DEVANT";H2
150 N1=INT(M2*L1)
160 N2=INT(M2*L2)
170 N3=INT(R2*H1)
180 N4=INT(R2*H2)
190 D=INT((N1-N2)/2)
280 REM -----
290 REM MESURES MANCHES
300 CLS:PRINT"MESURES MANCHES":PRINT
310 INPUT"LARGEUR POIGNET";P1
320 INPUT"LARGEUR MANCHE A L'EPAULE";E1
330 INPUT"HAUTEUR MANCHE A L'EPAULE";H3
340 INPUT"HAUTEUR TOTALE MANCHE";H4
350 P2=INT(M2*P1)
360 E2=INT(M2*E1)
370 N5=INT(R2*H3)
380 N6=INT(R2*H4)
390 A=INT((E2-P2)/2)
480 REM -----
490 REM RESULTATS DOS ET DEVANT
500 CLS:PRINT"  MONTAGE DOS OU DEVANT":PRINT
510 PRINT"-MONTER";N1;"MAILLES, SOIT";L1;"CMS."
520 PRINT"-TRICOTER";N3;"RANGS, SOIT";H1;"CMS."
530 PRINT"-A";H1;"CMS, RABATTRE DE CHAQUE COTE"
540 PRINT D;"MAILLES, EN PLUSIEURS FOIS,"
550 PRINT" POUR LES EMMANCHURES."
560 N7=N4-N3
570 C=H2-H1
580 PRINT"-TRICOTER ENCORE";N7;"RANGS, SOIT"
590 PRINT C;"CMS."
600 PRINT"-A";H2;"CMS, FERMER SELON L'ENCOLURE."
680 REM -----
690 REM RESULTATS MANCHES
700 PRINT:PRINT:PRINT"  MONTAGE DES MANCHES":PRINT
710 PRINT"-MONTER";P2;"MAILLES, SOIT";P1;"CMS."
720 PRINT"-TRICOTER";N5;"RANGS, SOIT";H3;"CMS."
730 PRINT" EN AUGMENTANT REGULIEREMENT"
740 PRINT" CHAQUE COTE DE";A;"MAILLES."
750 N8=(P2+(A*2))
760 N9=INT(N8/2)
770 PRINT"-A";H3;"CMS, RABATTRE LES";N8;"MAILLES"
780 PRINT" RESTANTES REGULIEREMENT, SOIT";N9
790 PRINT" MAILLES DE CHAQUE COTE."
800 PRINT"-A";H4;"CMS, FERMER LE TRICOT."

```

LISTE DES VARIABLES :

A	augmentation des manches de chaque côté
C	cms entre l'épaule et le total (dos ou devant)
D	diminutions des emmanchures de chaque côté
E1	largeur à l'épaule
E2	nombre de mailles à l'épaule
H1	hauteur sous le bras
H2	hauteur totale dos ou devant
H3	hauteur à l'épaule
H4	hauteur totale
I	indice de boucle FOR/NEXT
L1	largeur dos ou devant
L2	largeur à la poitrine
M1	nombre de mailles pour 10 cm
M2	nombre de mailles pour 1 cm
N1	nombre de mailles largeur dos ou devant
N2	nombre de mailles largeur poitrine
N3	nombre de rangs hauteur sous bras
N4	nombre de rangs hauteur totale dos ou devant
N5	nombre de rangs à l'épaule
N6	nombre de rangs hauteur totale
N7	nombre de rangs entre l'épaule et le total dos ou devant
N8	nombre de mailles à l'épaule de la manche
N9	nombre de mailles à rabattre de chaque côté de la manche
P1	largeur poignet
P2	nombre de mailles au poignet
R1	nombre de rangs pour 10 cm
R2	nombre de rangs pour 1 cm



THEME 4 LES JEUX NATIONAUX

LE TIERCE : PRONOSTICS

Vous jouez au tiercé et vous vous lancez dans des calculs fastidieux pour faire la synthèse des avis de vos pronostiqueurs favoris ?

Ne vous trompez plus dans vos numéros ; demandez tout simplement l'aide de votre MSX !

Entrez le nombre de chevaux de la course, le nombre de pronostics en votre possession et le nombre de chevaux minimum commun à tous les pronostics.

Un court temps de réflexion et l'ordinateur vous donne les chevaux qui ont le plus de chance d'arriver dans les quatre premiers, ainsi que le total des points obtenus dans chaque pronostic d'après l'ordre de classement.

Il vous révélera aussi quatre tiercés possibles d'après ses résultats.

Alors pourquoi ne pas essayer... au moins une fois !

PRONOSTICS


PLACE CHEVAL POINTS

1	16	8
2	9	8
3	12	7
4	14	3
5	8	2
6	4	2

VOUS POUVEZ JOUER LE :

16	9	12
9	12	14
12	14	16
14	16	9

TIERCE



```

10 REM PRONOSTICS
20 REM -----
30 DIM P(30),R(30)
40 SCREEN 0
50 INPUT "NOMBRE DE CHEVAUX";NC
60 IF NC>30 THEN 40
70 FOR I=1 TO NC
80 R(I)=I
90 NEXT I
100 PRINT
110 INPUT "NB DE PRONOSTICS";N
120 PRINT
130 INPUT "NB DE CHX/PRONO.";CP
140 CLS
150 FOR I=1 TO N
160 PRINT "PRONOSTICS NO";I
170 FOR J=CP TO 1 STEP -1
180 PRINT "NO DU CHEVAL";CP+1-J;
190 INPUT NP
200 IF NP>NC THEN 180
210 P(NP)=P(NP)+J
220 NEXT J
230 PRINT:PRINT:PRINT
240 NEXT I
280 REM -----
290 REM TRI
300 FOR I=1 TO NC-1
310 FOR J=I+1 TO NC
320 IF P(J)<P(I) THEN 350
330 SWAP P(J),P(I)
340 SWAP R(J),R(I)
350 NEXT J
360 NEXT I
480 REM -----
490 REM RESULTATS
500 CLS:PRINT TAB(13);"PRONOSTICS":PRINT:PRINT
510 PRINT "        PLACE CHEVAL POINTS"
520 FOR I=1 TO NC
530 IF P(I)=0 THEN 550
540 PRINT TAB(7);I;TAB(15);R(I);TAB(23);P(I)
550 NEXT I
560 PRINT:PRINT
570 PRINT "        VOUS POUVEZ JOUER LE :":PRINT
580 FOR I=1 TO 4
590 FOR J=0 TO 2
600 IF J+I>4 THEN 630
610 PRINT TAB(7+J*8);R(J+I);
620 GOTO 640
630 PRINT TAB(7+J*8);R(J+I-4);
640 NEXT J
650 PRINT
660 NEXT I
670 AS=INPUT$(1)
680 RUN

```

LE PROGRAMME

Lignes 30 à 240 : saisie des paramètres

Après la saisie du nombre de chevaux, du nombre de pronostics et du nombre de chevaux par pronostic, le programme vous demande les numéros des chevaux, dans l'ordre proposé par vos journaux préférés.

On place alors dans le tableau P les points obtenus, le premier cheval obtenant le maximum (7 points s'il y a 7 chevaux par pronostic).

A l'initialisation, le tableau R contient les numéros des chevaux dans l'ordre de départ ($R(1)=1$, $R(2)=2$, ...).

Lignes 300 à 360 : tri

Cette partie du programme trie les deux tableaux en plaçant les chevaux par ordre croissant des points obtenus.

Le tableau R se trouve alors dans l'ordre d'arrivée (présumé !).

Il s'agit ici d'un tri par la méthode Bubble Sort. Il consiste à comparer le premier élément du tableau P à tous les autres éléments, en effectuant une inversion chaque fois que ceux-ci sont plus petits. Le plus petit élément est ainsi amené en début de table. On effectue la même chose avec le deuxième élément, et ainsi de suite jusqu'au classement complet. L'inversion de deux éléments du tableau est assurée par l'instruction SWAP.

Lignes 500 à 680 : résultats

Le programme affiche à l'écran la liste des chevaux dans l'ordre avec leurs numéros et le nombre de points qu'ils ont obtenus, pour peu que ce dernier soit supérieur à 1.

Les lignes 580 à 660 ne considèrent que les quatre premiers et proposent quatre solutions pour votre tiercé en effectuant des combinaisons.

LISTE DES VARIABLES :

A\$	dernière touche appuyée
CP	nombre de chevaux par pronostic
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
N	nombre de pronostics
NC	nombre de chevaux
NP	numéro du cheval
P()	points de chaque cheval
R()	ordre d'arrivée pronostiqué

LE LOTO : GRILLES ALEATOIRES

Seriez-vous de ces sceptiques qui ne jouent pas au loto parce que l'on n'y gagne jamais ?


Ou seriez-vous au contraire un de ces acharnés qui croient au hasard pur ?

Dans ce cas pourquoi ne pas tenter autre chose que votre date de naissance.

Voilà un programme de loto qui vous imprimera une grille avec un choix de six à dix numéros (pour les amateurs de grilles multiples), juste le temps de feuilleter la suite de votre livre.

L O T O				
	10	20	30	40
1	11	21	31	41
2	12	22	32	42
3	13	23	33	43
4	14	24	34	44
5	15	25	35	45
6	16	26	36	46
7	17	27	37	47
8	18	28	38	48
9	19	29	39	49

LE LOTO - CHIFFRES A TIRER



```

10 REM LOTO (ALEATOIRE)
20 REM -----
30 DIM A(49)
40 SCREEN 0:COLOR 4,1,1:KEY OFF
50 A=RND(-(TIME/10) MOD 100)
60 INPUT "NOMBRE DE CHIFFRES A TIRER";N
70 IF N<6 OR N>10 THEN 40
80 REM -----
90 REM TIRAGE
100 FOR I=1 TO N
110 A=INT(RND(1)*49)+1
120 IF A(A)=1 THEN 110
130 A(A)=1
140 NEXT I
150 OPEN "GRP:" FOR OUTPUT AS #1
160 REM -----
170 REM RESULTAT
200 SCREEN 2:DRAW "BM122,0":PRINT #1,"LOTO"
210 LINE (72,17)-(198,181),6,B
220 FOR I=0 TO 9
230 FOR J=0 TO 40 STEP 10
240 IF I+J=0 THEN 290
250 X=10+J/10*3:Y=3+I*2
260 DRAW "BM"+STR$(X*8)+", "+STR$(Y*8):PRINT #1,USING"##";I+J
270 LINE (X*8-2,Y*8-3)-(X*8+16,Y*8+9),6,B
280 IF A(I+J)=1 THEN LINE (X*8,Y*8-2)-(X*8+15,Y*8+8)
290 NEXT J
300 NEXT I
310 LINE INPUT A$:RUN

```

LE PROGRAMME

Lignes 30 à 70 : initialisation

Après le dimensionnement du tableau A(), on saisit au clavier le nombre de chiffres que l'on veut obtenir (6 pour les grilles simples, 7 à 10 pour les grilles multiples).

La ligne 70 refuse toutes autres valeurs.

Lignes 100 à 150 : tirage

Les lignes 100 à 140 remplissent le tableau A() avec des chiffres 1 pour les cases qui viennent d'être tirées au hasard. La ligne 120 vérifie le contenu de A() pour s'assurer que le nombre n'a pas déjà été tiré. Le fichier ouvert à la ligne 150 permet d'écrire des caractères sur l'écran graphique.

Lignes 200 à 310 : résultat

Après le dessin du cadre (ligne 210), on affiche avec deux boucles FOR/NEXT imbriquées les 49 chiffres du loto. Les chiffres qui ont été choisis sont barrés comme sur un bulletin normal.

LISTE DES VARIABLES :

A	nombre aléatoire
A()	liste des chiffres choisis
AS	dernière touche appuyée
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
N	nombre de chiffres à tirer
X et Y	coordonnées sur l'écran du chiffre qui va être affiché

LE LOTO : STATISTIQUES

Pourquoi ne pas jouer au Loto "scientifiquement" ?

Il faut pour cela garder tous les tirages depuis la création de ce jeu. En effet, un chiffre tiré 500 fois a beaucoup plus de chance de sortir qu'un chiffre déjà tiré 600 fois.

Lors de la première utilisation, le programme demande les tirages précédents des 49 chiffres. Si vous ne les connaissez pas, vous pouvez les trouver dans une des nombreuses brochures spécialisées sur le Loto.

Cette formalité étant remplie, donnez le tirage de la semaine et le nombre de chiffres dont vous avez besoin (6 pour la formule simple, de 7 à 10 pour les multiples). Après le tri, le bulletin s'affiche, les chiffres qui ont le plus de chance de sortir étant barrés.


Pour finir, le MSX vous invite à sauver ses précieux renseignements pour le prochain tirage soit sur cassette soit sur disquette suivant la configuration que vous possédez.

LISTE DES TIRAGES

(1) = 957	(18) = 1234	(35) = 1170
(2) = 1033	(19) = 1045	(36) = 939
(3) = 1265	(20) = 967	(37) = 1070
(4) = 988	(21) = 987	(38) = 1261
(5) = 1103	(22) = 1075	(39) = 1009
(6) = 1299	(23) = 1154	(40) = 1205
(7) = 990	(24) = 976	(41) = 1301
(8) = 1056	(25) = 1287	(42) = 1273
(9) = 1145	(26) = 1105	(43) = 999
(10) = 1302	(27) = 1271	(44) = 1041
(11) = 956	(28) = 934	(45) = 1056
(12) = 1134	(29) = 1256	(46) = 1206
(13) = 1206	(30) = 1193	(47) = 1159
(14) = 988	(31) = 1010	(48) = 1263
(15) = 1294	(32) = 982	(49) = 1298
(16) = 1001	(33) = 987	
(17) = 1102	(34) = 1034	

APPUYEZ SUR RETURN QUAND LE MAGNETO
K7 EST PRET POUR L'ENREGISTREMENT

LE LOTO STATISTIQUES



```

10 REM LOTO (STATISTIQUES)
20 REM -----
30 DIM A(49),B(49),C(49),L(49)
40 SCREEN 0:COLOR 15,1,1:KEY OFF
50 A=RND(-(TIME/10) MOD 100)
60 INPUT"PREMIERE UTILISATION (O/N)";R$
70 IF R$="O" THEN GOSUB 1000:GOTO 140
80 CLS:PRINT"METTEZ LE MAGNETO K7 EN LECTURE"
90 OPEN "LOTO" FOR INPUT AS #1
100 FOR I=1 TO 49
110 INPUT#1,L(I)
120 NEXT I
130 CLOSE
140 OPEN "GRP:" FOR OUTPUT AS #1
180 REM -----
190 REM LOTO DE LA SEMAINE
200 CLS
210 PRINT"ENTREZ LE LOTO DE LA SEMAINE":PRINT
220 FOR I=1 TO 7
230 IF I=7 THEN PRINT"COMPLEMENTAIRE ";:GOTO 250
240 PRINT "CHIFFRE";I;
250 INPUT C
260 IF C<1 OR C>49 THEN 230
270 L(C)=L(C)+1
280 NEXT I
380 REM -----
390 REM CHOIX DE LA SEMAINE
400 CLS:INPUT"NOMBRE DE CHIFFRES A TIRER ";N
410 IF N<6 OR N>10 THEN 400
420 FOR I=1 TO 49
430 C(I)=L(I):B(I)=I:A(I)=0
440 NEXT I
450 F=0
460 FOR I=1 TO 48
470 IF C(I)<=C(I+1) THEN 510
480 A=C(I):C(I)=C(I+1):C(I+1)=A
490 A=B(I):B(I)=B(I+1):B(I+1)=A
500 F=1
510 NEXT I
520 IF F=1 THEN 450
530 FOR I=1 TO N
540 A(B(I))=1
550 NEXT I
580 REM -----
590 REM RESULTATS
600 SCREEN 2:DRAW"BM122,0":PRINT#1,"LOTO"
610 LINE (72,17)-(198,181),6,B
620 FOR I=0 TO 9
630 FOR J=0 TO 40 STEP 10
640 IF I+J=0 THEN 690
650 X=10+J/10*3:Y=3+I*2
660 DRAW"BM"+STR$(X*8)+", "+STR$(Y*8):PRINT#1,USING"##";I+J
670 LINE (X*8-2,Y*8-3)-(X*8+16,Y*8+9),6,B
680 IF A(I+J)=1 THEN LINE (X*8,Y*8-2)-(X*8+15,Y*8+8)
690 NEXT J
700 NEXT I
710 DRAW"BM32,184":PRINT#1,"UNE AUTRE SEMAINE (O/N) ?"
720 R$=INPUT$(1)
730 IF R$="O" THEN 200
780 REM -----
790 REM SAUVEGARDE
800 SCREEN 0:CLOSE:CLS:PRINT"LISTE DES TIRAGES"
810 X=0:Y=2
820 FOR I=1 TO 49

```

```

830 LOCATE X,Y:PRINT "(";MID$(STR$(I)+ " ",2,2);")=";L(I)
840 Y=Y+1
850 IF I/17=INT(I/17) THEN X=X+12:Y=2
860 NEXT I
870 LOCATE 0,20:PRINT"APPUYEZ SUR RETURN QUAND LE MAGNETO K
7 EST PRET POUR L'ENREGISTREMENT"
880 LINE INPUT R$
890 OPEN "LOTO" FOR OUTPUT AS #1
900 FOR I=1 TO 49
910 PRINT#1,L(I)
920 NEXT I
930 CLOSE
940 CLS:PRINT"A LA SEMAINE PROCHAINE !"
950 END
980 REM -----
990 REM ENTREE DU NOMBRE DE TIRAGES
1000 CLS:INPUT"VOULEZ-VOUS ENTRER LES TIRAGES DES ANNEE
S PRECEDENTES (O/N) ";R$
1010 IF R$<>"O" THEN 1070
1020 CLS:PRINT" ENTREZ LES TIRAGES":PRINT
1030 FOR I=1 TO 49
1040 PRINT I;"A DEJA ETE TIRE ";
1050 INPUT L(I)
1060 NEXT I
1070 RETURN

```

LE PROGRAMME

Lignes 30 à 140 : initialisation

S'il s'agit de la première utilisation on exécute le sous-programme en 1000 pour entrer les tirages des années précédentes.

Dans le cas contraire on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 200 à 280 : loto de la semaine

Le programme demande les sept chiffres du dernier tirage. On incrémente en conséquence le tableau L().

Lignes 400 à 550 : choix de la semaine

Après la sauvegarde du tableau L() dans le tableau C() (lignes 420 à 440), les lignes 450 à 520 le trient en ordre croissant.

Les lignes 530 à 550 placent dans le tableau A() les N nombres qui ont été choisis, c'est-à-dire ceux qui ont été le moins souvent tirés.

Lignes 600 à 730: résultats

Les lignes 620 à 700 affichent une grille de loto en barrant les chiffres qui ont le plus de chance de sortir.

La ligne 730 renvoie à la saisie si l'on désire entrer le tirage d'une autre semaine avant la sauvegarde.

Lignes 800 à 950 : sauvegarde


Le programme affiche sur l'écran les 49 nombres du loto avec leurs tirages respectifs, puis sauve sur cassette ou sur disquette le fichier dès qu'une touche a été appuyée.

Lignes 1000 à 1070 : entrée du nombre de tirages

Ces lignes saisissent les tirages des années précédentes pour chacun des nombres de 1 à 49.

LISTE DES VARIABLES :

A()	liste des nombres choisis
B()	ordre des 49 nombres après le tri
C	chiffres de la semaine
C()	sauvegarde du tableau L() pour le tri
F	drapeau. Si F=1 le tableau n'est pas encore trié.
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
L()	liste du nombre de tirages
N	nombre de chiffres à tirer
R\$	réponse du joueur
X et Y	coordonnées du prochain nombre à afficher



THEME 5 LE TEMPS

ARBRE GENEALOGIQUE

Comment déterminer les liens de parenté entre tous les membres d'une famille en vue d'élaborer un arbre généalogique ?

C'est ce que va faire pour vous le MSX.

Trois options vous sont présentées :

- 1 – **SAISIE** : vous pouvez entrer le numéro d'une personne suivi de son nom, son sexe (M ou F), puis le numéro d'un autre membre de la famille en relation avec celui que vous désirez créer. Pour ce dernier vous saisirez tout de suite après le type de relation qu'il a avec le personnage en cours de création (E pour enfant, P pour parent, F pour frère et sœur, et M pour un mariage).

Par exemple, on crée la personne X : si Y, référence du lien avec X est l'enfant de X, vous devez donner la lettre E pour enfant.


A la 1^{re} création vous serez obligé de donner un numéro non identifié comme référence, et vous rentrerez ses coordonnées à la deuxième saisie.

- 2 – **RECHERCHE** : en tapant le numéro d'une personne, vous obtenez ses liens avec tous les autres membres de la famille.

- 3 – **SAUVEGARDE** : sauve le fichier sur cassette ou sur disquette.

Pour changer d'option il suffit de taper "Ø" comme numéro et l'on revient au menu.

ARBRE GENEALOGIQUE



```

10 REM ARBRE GENEALOGIQUE
20 REM -----
30 CLEAR 3000
40 DIM N$(40), S$(40), P(40,40), T$(40,40), G$(5,8,2), A$(40)
50 GOSUB 1700
60 SCREEN 0:COLOR 4,1,1:KEY OFF
70 INPUT"PREMIERE UTILISATION (O/N)";R$
80 IF R$="O" THEN GOTO 300
90 CLS:PRINT"METTEZ LE MAGNETO K7 EN LECTURE"
100 OPEN "ARBRE" FOR INPUT AS #1
110 INPUT#1,N
120 FOR I=1 TO N
130 INPUT#1,N$(I),S$(I)
140 FOR J=1 TO N
150 INPUT#1,P(J,I),T$(J,I)
160 NEXT J,I
170 CLOSE
280 REM -----
290 REM MENU
300 CLS
310 PRINT"1 SAISIE PERSONNE"
320 PRINT"2 RECHERCHE"
330 PRINT"3 SAUVEGARDE"
340 PRINT "VOTRE CHOIX ? ";
350 V=VAL(INPUT$(1)):IF V<1 OR V>3 THEN 350
360 ON V GOTO 400,700,1300
380 REM -----
390 REM SAISIE
400 GOSUB 1500
410 PRINT:NU=0:INPUT"NUMERO DE SAISIE ";NU
420 IF NU=0 THEN 300
430 IF NU>N+1 THEN 410
440 IF NU>40 THEN PRINT"FICHIER PLEIN":GOTO 410
450 IF N$(NU)<>" " THEN PRINT N$(NU):GOTO 500
460 INPUT"NOM";N$(NU)
470 IF N$(NU)=" " THEN 400
480 INPUT"SEXE";S$(NU)
490 IF S$(NU)<>"M" AND S$(NU)<>"F" THEN 480
500 I=NU:N=N+1
510 L=0:INPUT"LIAISON AVEC ";L:IF L=0 THEN 400
520 IF P(L,NU)=L THEN PRINT"JE LE SAIS DEJA":GOTO 510
530 P(L,NU)=L
540 INPUT"TYPE (E,P,F,M)";T$(L,NU)
550 IF T$(L,NU)<>"E" AND T$(L,NU)<>"P" AND T$(L,NU)<>"F" AND
   T$(L,NU)<>"M" THEN 540
560 P(NU,L)=NU:T$(NU,L)=T$(L,NU)
570 IF T$(NU,L)="E" THEN T$(NU,L)="P":GOTO 590
580 IF T$(NU,L)="P" THEN T$(NU,L)="E"
590 GOTO 400
680 REM -----
690 REM RECHERCHE
700 GOSUB 1500
710 PRINT:NU=0:INPUT"NUMERO DE RECHERCHE ";NU
720 IF NU=0 OR N$(NU)=" " THEN 300
730 FOR I=1 TO N:A$(I)=" ":NEXT
740 CLS:PRINT N$(NU); " A POUR: "
750 FOR I=1 TO N
760 IF T$(I,NU)=" " THEN 800
770 A$(I)=T$(I,NU)
780 IF T$(I,NU)="E" THEN A$(P(I,NU))="P":GOTO 800
790 IF T$(I,NU)="P" THEN A$(P(I,NU))="E"
800 NEXT I
810 FOR K=1 TO N

```

```

820 FOR I=1 TO N
830 IF I=NU OR N$(I)="" OR A$(I)<>"" THEN 880
840 FOR J=1 TO N
850 IF A$(J)="" THEN 870
860 IF P(J,I)=J THEN A$(I)=T$(J,I)+A$(J):GOTO 880
870 NEXT J
880 NEXT I
890 NEXT K
900 FOR I=1 TO N
910 IF A$(I)="" THEN 1230
920 FOR J=1 TO LEN(A$(I))
930 IF MID$(A$(I),J,2)="FF" THEN A$(I)=LEFT$(A$(I),J)+RIGHT$(A$(I),LEN(A$(I))-J-1):IF J>1 THEN J=J-2:GOTO 1000 ELSE J=J-1:GOTO 1000
940 IF MID$(A$(I),J,3)="EFE" THEN A$(I)=LEFT$(A$(I),J)+RIGHT$(A$(I),LEN(A$(I))-J-1)
950 IF MID$(A$(I),J,3)="PFP" THEN A$(I)=LEFT$(A$(I),J)+RIGHT$(A$(I),LEN(A$(I))-J-1)
960 IF MID$(A$(I),J,2)="FE" THEN A$(I)=LEFT$(A$(I),J-1)+RIGHT$(A$(I),LEN(A$(I))-J)
970 IF MID$(A$(I),J,2)="EM" THEN A$(I)=LEFT$(A$(I),J)+RIGHT$(A$(I),LEN(A$(I))-J-1)
980 IF MID$(A$(I),J,2)="PF" THEN A$(I)=LEFT$(A$(I),J)+RIGHT$(A$(I),LEN(A$(I))-J-1):J=J-1:GOTO 1000
990 IF MID$(A$(I),J,2)="MP" THEN A$(I)=LEFT$(A$(I),J-1)+RIGHT$(A$(I),LEN(A$(I))-J):J=J-1
1000 NEXT J
1010 V=0:F=0:M=0:C=0
1020 FOR J=1 TO LEN(A$(I))
1030 IF MID$(A$(I),J,1)="E" THEN V=V-1
1040 IF MID$(A$(I),J,1)="P" THEN V=V+1
1050 IF MID$(A$(I),J,2)="EF" THEN F=1
1060 IF MID$(A$(I),J,2)="FP" THEN F=1
1070 IF MID$(A$(I),J,2)="MF" THEN F=1
1080 IF MID$(A$(I),J,2)="FM" THEN F=1
1090 IF MID$(A$(I),J,1)="M" THEN M=2
1100 IF MID$(A$(I),J,2)="PE" AND MID$(A$(I)+A$(I),J,1)<>"E" AND MID$(A$(I)+A$(I),J+2,1)<>"P" THEN M=2
1110 IF MID$(A$(I),J,2)="EP" AND MID$(A$(I)+A$(I),J,1)<>"P" AND MID$(A$(I)+A$(I),J+2,1)<>"E" THEN F=1
1120 IF MID$(A$(I),J,3)="EFP" THEN C=3
1130 NEXT J
1140 IF C=3 AND M=2 THEN C=0
1150 IF S$(I)="M" THEN S=1 ELSE S=2
1160 IF V=0 THEN PRINT G$(1+F+M+C,4,S);:GOTO 1220
1170 FOR J=V TO SGN(V) STEP -SGN(V)
1180 IF J<>V THEN PRINT "- ";
1190 IF J+4<1 OR J+4>7 THEN PRINT"ARRIERE";:GOTO 1210
1200 PRINT G$(1+F+M+C,J+4,S);
1210 NEXT J
1220 PRINT " ";N$(I)
1230 NEXT I
1240 A$=INPUT$(1):GOTO 700
1280 REM -----
1290 REM SAUEGARDE
1300 CLS:PRINT"APPUYEZ SUR RETURN QUAND LE MAGNETO K7 EST PRET POUR L'ENREGISTREMENT"
1310 LINE INPUT R$
1320 OPEN "ARBRE" FOR OUTPUT AS #1
1330 PRINT#1,N
1340 FOR I=1 TO N
1350 PRINT#1,N$(I),"S$(I)
1360 FOR J=1 TO N
1370 PRINT#1,P(J,I),"T$(J,I)

```

```

1380 NEXT J,I
1390 CLOSE
1400 CLS:PRINT"A LA PROCHAINE NAISSANCE"
1410 END
1480 REM -----
1490 REM AFFICHE LES NOMS
1500 CLS
1510 FOR I=1 TO N
1520 IF N$(I)="" THEN 1630
1530 PRINT "(";I;" " ";N$(I);" ";
1540 FOR J=1 TO N
1550 IF T$(J,I)<>"" THEN 1570
1560 NEXT J:GOTO 1630
1570 M=0
1580 IF S$(I)="M" THEN S=1 ELSE S=2
1590 IF T$(J,I)="P" THEN V=1:GOTO 1620
1600 IF T$(J,I)="E" THEN V=-1:GOTO 1620
1610 IF T$(J,I)="F" THEN V=0 ELSE V=0:M=2
1620 PRINT G$(1+M,V+4,S):PRINT "      DE ";N$(J)
1630 NEXT I
1640 RETURN
1680 REM -----
1690 REM DONNEES GENEALOGIQUES
1700 FOR K=1 TO 5
1710 FOR I=1 TO 7
1720 READ G$(K,I,1),G$(K,I,2)
1730 NEXT I
1740 NEXT K
1750 RETURN
1760 DATA ARRIERE,ARRIERE,PETIT,PETITE,FILS,FILLE,FRERE,SOEU
R
1770 DATA PERE,MERE,GRAND,GRAND,ARRIERE,ARRIERE
1780 DATA ARRIERE,ARRIERE,PETIT,PETITE,NEVEU,NIECE,FRERE,SOE
UR
1790 DATA ONCLE,TANTE,GRAND,GRAND,ARRIERE,ARRIERE
1800 DATA ARRIERE,ARRIERE,PETIT,PETITE,GENDRE,BRU,EPOUX,EPOU
SE
1810 DATA BEAU-PERE,BELLE-MERE,GRAND,GRAND,ARRIERE,ARRIERE
1820 DATA ARRIERE,ARRIERE,PETIT,PETITE,NEVEU (PAR ALLIANCE),
NIECE (PAR ALLIANCE),BEAU-FRERE,BELLE-SOEUR
1830 DATA ONCLE (PAR ALLIANCE),TANTE (PAR ALLIANCE),GRAND,GR
AND,ARRIERE,ARRIERE
1840 DATA ARRIERE,ARRIERE,ARRIERE,ARRIERE,PETIT-COUSIN,PETIT
E-COUSINE,COUSIN,COUSINE
1850 DATA GRAND-COUSIN,GRANDE-COUSINE,ARRIERE,ARRIERE,ARRIER
E,ARRIERE

```

LE PROGRAMME

Lignes 30 à 170 : initialisation

On fait appel au sous-programme en 1700 pour placer dans le tableau G\$() toutes les données généalogiques (père, mère, frère, sœur, fils, fille, etc.).

Les tableaux N\$(), S\$(), etc., (ligne 40) correspondent aux noms, sexes, etc., des personnes dans le fichier. Si vous trouvez que ce n'est pas assez, il est possible de modifier leurs indices maximum (ici 40) par une autre valeur supérieure. Il faut pour cela modifier tous les chiffres 40 de la ligne 40 et celui de la ligne 440. Donnez des valeurs raisonnables, sinon il ne restera plus de place en mémoire pour exécuter le programme. Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 300 à 360 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur (variable V).

Lignes 400 à 590 : saisie

On fait d'abord appel au sous-programme en 1500 (affiche les noms et les liaisons avec d'autres personnes).

Puis le programme demande le numéro que l'on veut créer ou modifier. Si on répond 0, la ligne 420 renvoie au menu. Dans le cas contraire, les lignes 460 et 480 saisissent le nom et le sexe.

Les lignes 510 à 590 saisissent les liaisons avec d'autres personnes. On vérifie en ligne 520 si le renseignement donné n'est pas déjà dans le fichier. Par exemple si l'on dit : X est l'enfant de Y, et que l'on sait déjà que Y est le parent de X on refuse l'entrée.

Ces lignes créent aussi toutes les liaisons inverses. Par exemple : si X est l'enfant de Y, on crée dans l'enregistrement de Y la liaison Y est le parent X (c'est le tableau S\$() – le sexe – qui détermine s'il s'agit du fils ou de la fille, du père ou de la mère).

Lignes 700 à 1240 : recherches

Après l'appel du sous-programme en 1500, on demande le numéro du nom que l'on veut rechercher. La boucle FOR/NEXT des lignes 750 à 800 place dans le tableau A\$(,) toutes les liaisons directes avec le nom à rechercher (frère, père, etc.). Les lignes 810 à 890 placent dans le tableau A\$(,) les liaisons indirectes en utilisant les liaisons directes déjà dans le tableau (le père du père, le frère du père, etc.).

Ce sont les lignes 900 à 1230 qui déterminent les liaisons finales de la personne recherchée. La variable V donne le niveau dans le tableau G\$(,), la variable F un lien entre frères, la variable M un mariage et la variable C concerne les cousins.

Les lignes 1170 à 1220 affichent les liaisons complètes. Si V=2 par exemple, on affiche à la suite les mots "grand" puis "père". Si V=3, les mots "arrière", "grand" puis "père". Si V=-2, les mots "petit" et "fils", et ainsi de suite. Les variables F, M et C font passer à d'autres niveaux dans le tableau G\$(,). Si F=1 et V=-1 par exemple, on affiche le mot "neveu" à la place de "fils" (c'est le fils du frère). La combinaison des variables F, M et C permet d'utiliser toutes les solutions du tableau G\$(,).

Lignes 1300 à 1410 : sauvegarde

Ces lignes sauvent sur cassette ou sur disquette le fichier dès qu'une touche a été appuyée.

Lignes 1500 à 1640 : affichent les noms

Ce sous-programme affiche tous les noms et les liaisons correspondantes.

LISTE DES VARIABLES :

A\$	dernière touche appuyée
A\$()	liste des noms des liaisons (enfant, parent, etc.), pour la recherche
C	C=3 si on a trouvé une liaison entre cousins
F	F=1 si on a trouvé une liaison entre frères
G\$(,,)	liste des données généalogiques
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
K	indice de boucle FOR/NEXT
L	dernière liaison saisie
M	M=2 si on a trouvé une liaison entre époux
N	nombre de personnes dans le fichier
NU	numéro de la personne à rechercher
N\$()	liste des noms
P	P=1 si toutes les liaisons avec NU n'ont pas été trouvées
P\$()	liste des liaisons avec les autres personnes (numéros)
R\$	réponse de l'utilisateur
S	S=1 pour masculin, S=2 pour féminin
SS\$()	liste des sexes
TS\$()	liste des liaisons avec les autres personnes (enfant, parent, etc.).
V	niveau généalogique

MARYSE MONTAUT A POUR:
FILLE CATHERINE MONTAUT
FILS FRANCOIS MONTAUT
FILS DENIS MONTAUT
FILLE NICOLE MONTAUT
EPOUX CHRISTIAN MONTAUT
PETITE-FILLE FRANCOISE MONTAUT
BEAU-PERE ALEXANDRE MONTAUT
BRU CATHERINE MONTAUT
MERE HENRIETTE ROCHE
PETIT-FILS ERIC MONTAUT


ASTROLOGIE

Votre MSX va vous permettre de mieux étudier votre entourage grâce à son programme d'astrologie occidentale.


Dès que vous aurez rentré la date de naissance d'un proche, il vous affichera son signe zodiacal.

Il vous exposera aussi en quelques lignes les traits généraux qui caractérisent les natifs de ce signe.

A propos, de quel signe est votre MSX ?



ASTROLOGIE



```

10 REM ASTROLOGIE
20 REM -----
30 DIM J(12),S$(12):GOSUB 600
40 SCREEN 0
50 INPUT"JOUR ET MOIS DE NAISSANCE";J,M:IF M=1 THEN 70
60 FOR I=1 TO M-1:N=N+J(I):NEXT I
70 N=N+J
180 REM -----
190 REM CALCUL DU SIGNE
200 FOR I=1 TO 13
210 S=S+1
220 READ P:C=C+P
230 IF N<C THEN S=S-1:GOTO 260
240 NEXT I
250 STOP
260 IF S=0 THEN S=12
380 REM -----
390 REM SIGNE ASTROLOGIQUE
400 CLS:PRINT TAB(15)"ASTROLOGIE":PRINT:PRINT:PRINT
410 PRINT" VOUS ETES DU SIGNE ";
420 A$=RIGHT$(S$(S),2):IF A$="NS" OR A$="UX" THEN PRINT"DES
";:GOTO 440
430 IF A$="CE" OR A$="GE" THEN PRINT"DE LA "; ELSE PRINT"DU
";
440 PRINT S$(S)
450 PRINT
460 PRINT:RESTORE 660
470 IF A$="NS" OR A$="UX" THEN PRINT" LES ";S$(S); " SONT:":G
OTO 490
480 IF A$="CE" OR A$="GE" THEN PRINT" LA ";S$(S); " EST:" ELS
E PRINT"LE ";S$(S); " EST:"
490 FOR I=1 TO S:READ A$:NEXT I
500 PRINT:PRINT A$
510 A$=INPUT$(1):RUN
580 REM -----
590 REM DONNEES
600 FOR I=1 TO 12:READ J(I):NEXT I
610 DATA 31,28,31,30,31,30,31,31,30,31,30,31
620 FOR I=1 TO 12:READ S$(I):NEXT I
630 DATA VERSEAU,POISSONS,BELIER,TAUREAU,GEMEAUX,CANCER,LION
,VIERGE,BALANCE,SCORPION,SAGITTAIRE,CAPRICORNE
640 RETURN
650 DATA 20,30,30,30,31,31,32,31,31,30,30,30,29
660 DATA "SIGNE D'AIR. ALTRUISTE ET SENSIBLE, IL CHERCHE A S
E REALISER DANS LE DOMAINE MORAL OU SOCIAL. INDEPENDANT, SOU
VENT ANTICONFORMISTE, MAIS TOUJOURS LUCIDE, IL EST A LA RECH
ERCHE D'UN ABSOLU DANS SA VIE."
670 DATA "SIGNE D'EAU. SECRET ET IMAGINATIF, IL SE VEUT PART
IE INTEGRANTE DE L'UNIVERS. IL EST EMOTIF, IMPRESSIONNABLE,
SENSIBLE A TOUT CE QUI TOUCHE LES AUTRES. SA GENEROSITE L'EN
TRAINE VERS LES GRANDES CAUSES HUMANITAIRES."
680 DATA "SIGNE DE FEU. IL EST LE TYPE MEME DE L'HOMME OU DE
LA FEMME D'ACTION. PLEIN DE VITALITE ET DE TONUS, IL AIME S
E DEPENSER PHYSIQUEMENT ET SE SURPASER DANS L'EFFORT."
690 DATA "SIGNE DE TERRE. IL POSSEDE UNE PERSONNALITE PROFON
DE ET PASSIONNEE. C'EST UN ETRE CHARNEL ET SENSUEL. IL FAIT
PREUVE D'UNE ACTIVITE TENACE QUI LE POUSSE A ALLER JUSQU'AU
BOUT DE CE QU'IL ENTREPREND."
700 DATA "SIGNE D'AIR. LE NATIF DE CE SIGNE POSSEDE UNE GRAN
DE VIVACITE DE TEMPERAMENT QUI EN FAIT UN ETRE DOUE. SA PERS
ONNALITE MULTIPLE (SIGNE DOUBLE) LUI DONNE UNE GRANDE FACULT
E D'ADAPTATION ET BEAUCOUP DE RAPIDITE DANS LE TRAVAIL."

```

710 DATA "SIGNE D'EAU, IL EST DOTE D'UNE PERSONNALITE TRANQUILLE POSSEDANT LE CALME DE L'ESPRIT, UN CHARME SUBTIL ET UNE GRANDE DOUCEUR DE TEMPERAMENT. IL AIME FLANER, REVER, SE REFUGIER DANS L'IMAGINAIRE."

720 DATA "SIGNE DE FEU. D'UN TEMPERAMENT MAGNIFIQUE ET GENEREUX, IL DETESTE PASSER INAPERCU ET POSSEDE LE GOUT DE SE REALISER PLEINEMENT DANS SA VIE SOCIALE. VOLONTAIRE, C'EST L'HOMME OU LA FEMME DES GRANDES ENTREPRISES."

730 DATA "SIGNE DE TERRE. DE NATURE METHODIQUE ET METICULEUX, LE NATIF DE LA VIERGE RECHERCHE LE CALME DE L'ESPRIT ET LA SECURITE. IL FAIT CEPENDANT PREUVE DE CURIOSITE ET DE VIVACITE D'ESPRIT EN PROFONDEUR."

740 DATA "SIGNE D'AIR. OUVERT, SENSIBLE, AFFECTUEUX, LE NATIF DE LA BALANCE EST UN ETRE A LA RECHERCHE DE SON EQUILIBRE. IL SE CARACTERISE PAR SON CHARME, SA SUBTLITE ET SON SENS INTUITIF."

750 DATA "SIGNE D'EAU. C'EST UN PASSIONNE COMBATIF. ALTRUISTE ET TRAVAILLEUR ACHARNE, LE NATIF DE CE SIGNE POSSEDE UNE RESISTANCE PHYSIQUE CONSIDERABLE."

760 DATA "SIGNE DE FEU. LE SAGITTAIRE POSSEDE UNE PERSONNALITE BRILLANTE SOUTENUE PAR UN PROFOND EQUILIBRE ET UN TRES GRAND SERIEUX. IL EPROUVE LE BESOIN DE FRANCHIR SES LIMITES TANT MATERIELLES QUE SPIRITUELLES."

770 DATA "SIGNE DE TERRE. C'EST UN FLEGMATIQUE OU UN PASSIONNE A FROID. REFLECHI ET METHODIQUE, IL EST PROFOND, GRAVE ET DISCIPLINE. UNE FOIS SA DECISION PRISE, IL EST DIFFICILE DE LE FAIRE CHANGER D'AVIS."

LE PROGRAMME

Lignes 30 à 70 : initialisation

Après l'appel du sous-programme en 600 qui place dans le tableau J() le nombre de jours par mois et dans le tableau S\$() les douze signes du zodiaque, le programme saisit le jour et le mois de naissance et calcule le nombre de jours vécus depuis le 1^{er} janvier.

Lignes 200 à 260 : calcul du signe

Les DATAs de la ligne 650 donnent le nombre de jours d'écart entre chaque signe. La boucle FOR/NEXT des lignes 200 à 240 compare le total de ces jours avec ceux du nombre de jours vécus (variable N) pour déterminer le signe du zodiaque (variable S).

Lignes 400 à 510 : signe astrologique

Ces lignes affichent le signe du zodiaque pointé par S, ainsi que le commentaire correspondant (DATAs des lignes 660 à 770).

Les lignes 420, 430, 470 et 480 rétablissent l'orthographe s'il s'agit d'un signe au pluriel ou au féminin.

LISTE DES VARIABLES :

A\$	deux dernières lettres du signe
C	total du nombre de jours d'écart entre les signes
I	indice de boucle FOR/NEXT
J	jour de naissance
J()	nombre de jours par mois
M	mois de naissance
N	nombre de jours vécus depuis le 1 ^{er} janvier
P	dernier écart entre deux signes
S	numéro du signe
S\$()	liste des douze signes du zodiaque

ASTROLOGIE CHINOISE

Pour approfondir la connaissance des gens qui vous entourent, voilà un programme d'astrologie chinoise qui complète celui d'astrologie occidentale.

La saisie d'une date de naissance par votre MSX lui fait calculer le signe chinois de cette année et l'élément qui la caractérise.

En effet, le signe astrologique dépend de l'année de naissance (l'année chinoise bien sûr, différente par sa durée d'une année occidentale). Ce signe est complété par un élément associé au yin et au yang (année positive et négative). Ainsi, 1983 était l'année du cochon d'eau vive (du 13 février 83 au 2 février 84).

Maintenant vous saurez tout sur votre MSX !

ASTROLOGIE CHINOISE

VOUS ETES DU SIGNE DU SINGE

1968 EST UNE ANNEE DE TERRE INCULTE

VOUS ETES DONC UN SINGE DE TERRE


LE SINGE EST:

- SPIRITUEL - SUBTIL - ADROIT - PASSIONNE - FASCINANT -

MAIS AUSSI:

- RUSE - INFIDELE - MALHONNETE - VANITEUX - DISSIMULATEUR -

ASTROLOGIE CHINOISE



```

10 REM ASTROLOGIE CHINOISE
20 REM -----
30 DIM J(12),S$(12):GOSUB 700
40 SCREEN 0:COLOR 4,1,1:KEY OFF
50 INPUT"JOUR,MOIS,ANNEE DE NAISSANCE";J,M,A
60 IF A>1899 THEN A=A-1900
70 IF A<0 OR A>99 THEN 40
80 N=A*365+INT(ABS((A-1)/4))
90 IF A/4=INT(A/4) AND A<>0 THEN J(2)=29 ELSE J(2)=28
100 IF M=1 THEN 120
110 FOR I=1 TO M-1:N=N+J(I):NEXT I
120 N=N+J
130 REM -----
140 REM CALCUL DU SIGNE
150 FOR I=1 TO 101
160 S=S+1
170 READ P:C=C+T(P)
180 IF N<C THEN S=S-1:GOTO 260
190 NEXT I
200 STOP
210 E=S:S=S MOD 12:E=E MOD 10
220 IF S=0 THEN S=12
230 IF E=0 THEN E=10
240 REM -----
250 REM SIGNE ASTROLOGIQUE
260 CLS:PRINT TAB(11)"ASTROLOGIE CHINOISE":PRINT:PRINT
270 PRINT" VOUS ETES DU SIGNE DU ";S$(S)
280 PRINT
290 PRINT A+1900;"EST UNE ANNEE ";E$(E);" ";Y$(E)
300 PRINT
310 PRINT" VOUS ETES DONC UN ";S$(S);" ";E$(E)
320 PRINT:RESTORE 890
330 PRINT" LE ";S$(S);" EST: "
340 FOR I=1 TO (S-1)*10:READ A$:NEXT I
350 PRINT:PRINT" - ";
360 FOR I=1 TO 5
370 READ A$:PRINT A$;" - ";
380 NEXT I
390 PRINT:PRINT:PRINT" MAIS AUSSI: ":PRINT:PRINT" - ";
400 FOR I=1 TO 5
410 READ A$:PRINT A$;" - ";
420 NEXT I
430 PRINT:PRINT
440 A$=INPUT$(1):RUN
450 REM -----
460 REM DONNEES
470 FOR I=1 TO 12:READ J(I):NEXT I
480 DATA 31,28,31,30,31,30,31,31,30,31,30,31
490 FOR I=1 TO 12:READ S$(I):NEXT I
500 DATA RAT,BOEUF,TIGRE,CHAT,DRAGON,SERPENT,CHEVAL,BOUC,SIN
510 GE,COQ,CHIEN,COCHON
520 FOR I=1 TO 10 STEP 2:READ E$(I):E$(I+1)=E$(I):NEXT I
530 DATA DE METAL,D'EAU,DE BOIS,DE FEU,DE TERRE
540 FOR I=1 TO 10:READ Y$(I):NEXT I
550 DATA FROID,CHAUD,NATURELLE,VIVE,,ALLUME,LATENT,ALLUME,IN
560 CULTE,CULTIVEE
570 T(0)=353:T(1)=354:T(2)=355:T(3)=383:T(4)=384:T(5)=385:T(
580 6)=31
590 RETURN
600 DATA 6,4,1,2,3,1,2,4,1,2,4,1
610 DATA 4,1,1,4,1,2,4,2,4,1,1,4
620 DATA 1,2,4,1,2,4,1,3,1,2,4,2
630 DATA 1,4,1,4,1,1,4,2,1,5,1,1
640 DATA 4,1,4,1,2,4,1,2,4,1,3,2

```

850 DATA 1,4,2,1,4,2,0,4,2,4,1,2
 860 DATA 4,1,1,4,1,4,1,2,4,2,1,4
 870 DATA 1,4,1,1,4,2,2,4,1,1,3,2
 880 DATA 4,1,2,4,1
 890 DATA ENERGIQUE, METICULEUX, SOCIABLE, HUMORISTE, SENTIMENTAL
 900 DATA AGITE, JOUEUR, INQUIET, ENNUYEUX, SOUPCONNEUX
 910 DATA TRAVAILLEUR, FIER, ELOQUENT, ORIGINAL, TENACE
 920 DATA LENT, AUTORITAIRE, VINDICATIF, JALOUX, LOURDAUD
 930 DATA GENEREUX, COURAGEUX, SENSIBLE, PROFOND, ACTIF
 940 DATA VANITEUX, QUERELLEUR, TEMERAIRE, EMPORTE, ENTETE
 950 DATA DISCRET, RAFFINE, VERTUEUX, AMBITIEUX, INGENIEUX
 960 DATA PEDANT, SUSCEPTIBLE, DISTANT, SECRET, TORTUEUX
 970 DATA SCRUPULEUX, PERSPICACE, CHANCEUX, AUTONOME, INFLUENT
 980 DATA INFLEXIBLE, EXIGEANT, IRRITABLE, INFATUE, INSATISFAIT
 990 DATA SAGE, CULTIVE, INTUITIF, ELEGANT, PHILOSOPHE
 1000 DATA PRESOMPTUEUX, POSSESSIF, FLEGMATIQUE, EXTRAVAGANT, VOL
 AGE
 1010 DATA AIMABLE, HABILE, INDEPENDANT, SENSUEL, ENJOUE
 1020 DATA EGOISTE, FAIBLE, IMPATIENT, INSENSIBLE, IMPITOYABLE
 1030 DATA CREATIF, INTELLIGENT, DOUX, SIMPLE, ALTRUISTE
 1040 DATA CAPRICIEUX, INEXACT, INDISCRET, DEPENDANT, INDISCIPLIN
 E
 1050 DATA SPIRITUEL, SUBTIL, ADROIT, PASSIONNE, FASCINANT
 1060 DATA RUSE, INFIDELE, MALHONNETE, VANITEUX, DISSIMULATEUR
 1070 DATA FRANC, SINCERE, AMUSANT, ENTHOUSIASTE, AVENTUREUX
 1080 DATA MANIAQUE, MEFIANT, DEPENDS, VANTARD, ACERBE
 1090 DATA LOYAL, ATTENTIF, LUCIDE, FIDELE, MODESTE
 1100 DATA CRITIQUE, PESSIMISTE, MORALISATEUR, RESERVE, CYNIQUE
 1110 DATA INDULGENT, CONSCIENCIEUX, DECISIF, PACIFIQUE, AIMANT
 1120 DATA NAIF, CREDULE, ENTETE, DESARME, TERRE A TERRE

LE PROGRAMME

Lignes 30 à 120 : initialisation

Après l'appel du sous-programme en 700 qui place dans le tableau J() le nombre de jours par mois, dans S\$() les douze signes, dans E\$() les 5 éléments et dans Y\$() le yin et le yang, le programme saisit le jour, le mois et l'année de naissance et calcule le nombre de jours vécus depuis le 1^{er} janvier 1900.

Lignes 200 à 280 : calcul du signe

Les DATAs des lignes 800 à 880 donnent le nombre de jours de chaque année chinoise (totalement différent des années occidentales de 365 ou 366 jours) en se référant au tableau T() pour ne pas surcharger les données dans le programme. La boucle FOR/NEXT des lignes 200 à 240 compare le total de ces jours avec ceux du nombre de jours vécus (variable N) pour déterminer le signe chinois (variable S). Les signes se répétant tous les douze ans, on ne prend que le modulo 12 de S (modulo = reste de la division). Il en est de même pour les éléments (variable E) qui se répètent tous les dix ans.

Lignes 400 à 580 : signe astrologique

Ces lignes affichent le signe chinois pointé par S, l'élément pointé par E avec le yin et le yang correspondant (yang = positif, yin = négatif), ainsi que des commentaires sur le signe (DATAs des lignes 890 à 1120).

LISTE DES VARIABLES :

A	année de naissance
AS	commentaires sur le signe à afficher
C	total du nombre de jours des années chinoises
E	numéro d'un élément
ES()	liste des cinq éléments
I	indice de boucle FOR/NEXT
J	jour de naissance
J()	nombre de jours par mois
M	mois de naissance
N	nombre de jours vécus depuis le 1 ^{er} janvier 1900
P	nombre de jours dans l'année chinoise (pointe le tableau T)
S	numéro du signe
SS()	liste des douze signes chinois
T()	nombre de jours possibles dans une année chinoise
YS()	liste des yin et des yang

ESPERANCE DE VIE

Vous ne le savez peut-être pas, mais les voyantes modernes ne vous répondent plus à la légère quand vous leur demandez combien de temps vous vivrez.

Elles calculent consciencieusement votre espérance de vie grâce à un ordinateur discrètement relié à leur boule de cristal.

Lequel ordinateur utilise le test très sérieux qui vous est proposé dans les pages suivantes. Il a été mis au point par un médecin anglais d'après les renseignements d'ordre privé demandés par les assureurs-vie à leurs clients.

Ce test est basé sur le principe d'une espérance de vie correspondant à chaque âge (71 ans si l'on a 30 ans, 73 ans si l'on a 50 ans...). Mais des critères négatifs (abus du tabac, de l'alcool...) ou positifs (bon sommeil, longévité des parents) peuvent rallonger ou raccourcir cette espérance de vie de plusieurs années. Bien sûr, cela reste du domaine des statistiques, chaque histoire humaine étant unique et imprévisible ! Mais on peut méditer...


ESPERANCE DE VIE


QUESTION 14

- 1-VOUS ETES MARIE OU VIVEZ
MARITALEMENT
- 2-VOUS ETES SOLITAIRE
- 3-VOUS ETES VEUF
- 4-FEMME SEPARÉE OU DIVORCÉE
- 5-FEMME VEUVE

REPONDEZ PAR UN CHIFFRE
ENTRE 1 ET 5 ?

ESPERANCE DE VIE





```

10 REM ESPERANCE DE VIE
20 REM -----
30 SCREEN 0:PRINT"      ESPERANCE DE VIE":PRINT:PRINT
40 INPUT"VOTRE AGE ";AGE
50 IF AGE<15 OR AGE>80 THEN 30
60 IF AGE<41 THEN E=70.7+(AGE-15)*(1.8/25)
70 IF AGE>40 AND AGE<60 THEN E=72.6+(AGE-41)*(3.4/18)
80 IF AGE>59 THEN E=76.3+(AGE-60)*(9.4/20)
180 REM -----
190 REM INTERPRETATION DES QUESTIONS
200 FOR I=1 TO 14
210 CLS:PRINT:PRINT TAB(12);"ESPERANCE DE VIE"
220 READ N
230 FOR J=1 TO N
240 READ A(J),Q$(J)
250 NEXT J
260 PRINT:PRINT"QUESTION ";I:PRINT
270 FOR J=1 TO N
280 A$=STR$(J)+"-"+Q$(J):PRINT A$
290 NEXT J
300 IF N>1 THEN 360
310 PRINT:PRINT"REPONDEZ O/N ? ";
320 R$=INPUT$(1)
330 IF R$<>"O" AND R$<>"N" THEN 320
340 IF R$="O" THEN E=E+A(1)
350 GOTO 400
360 PRINT:PRINT"REPONDEZ PAR UN CHIFFRE":PRINT"ENTRE 1 ET";N
; "?"
370 R$=INPUT$(1)
380 R=VAL(R$):IF R<1 OR R>N THEN 370
390 E=E+A(R)
400 NEXT I
480 REM -----
490 REM RESULTATS
500 CLS:PRINT:PRINT TAB(12);"ESPERANCE DE VIE":PRINT:PRINT:P
RINT
510 PRINT TAB(7);"VOTRE ESPERANCE DE VIE EST":PRINT
520 PRINT TAB(15);"DE";INT(E);"ANS"
530 A$=INPUT$(1)
540 RUN
580 REM -----
590 REM DONNEES DES QUESTIONS
600 DATA 1,3,"ETES-VOUS UNE FEMME"
610 DATA 1,1,"UN DE VOS GRANDS-PARENTS A VECU      80 ANS
    OU PLUS"
620 DATA 4,4,"VOTRE MERE A VECU PLUS DE 80 ANS"
630 DATA 2,"VOTRE PERE A VECU PLUS DE 80 ANS"
640 DATA 6,"LES DEUX",0,"AUCUN"
650 DATA 6,-4
660 DATA "VOUS AVEZ UN PARENT DECEDE D'UNE      CRISE CARD
    IAQUE AVANT 50 ANS"
670 DATA -4,"D'UNE EMBOLIE OU ARTERIOSCLEROSE    AVANT 5
    0 ANS"
680 DATA -2,"DECEDE ENTRE 50 ET 60 ANS"
690 DATA -3,"DECEDE DE DIABETE AVANT 60 ANS"
700 DATA -2,"DECEDE DE CANCER DE L'ESTOMAC      AVANT 6
    0 ANS"
710 DATA 0,"AUCUN"
720 DATA 4,-9,"VOUS FUMEZ PLUS DE 40 CIGARETTES  PAR
    JOUR"
730 DATA -7,"ENTRE 20 ET 40 CIGARETTES"
740 DATA -2,"MOINS DE 20 CIGARETTES"
750 DATA 0,"VOUS NE FUMEZ PAS"
760 DATA 1,2

```

770 DATA "VOUS AVEZ DES RELATIONS SEXUELLES OIS PAR SEMAINE"	1 OU 2 F
780 DATA 1,2,"VOUS EFFECTUEZ UN EXAMEN MEDICAL T TOUS LES ANS"	COMPLE
790 DATA 1,-2,"VOUS DEPASSEZ VOTRE POIDS NORMAL"	
800 DATA 1,-2,"VOUS DORMEZ PLUS DE 10 HEURES PAR OU MOINS DE 5 H"	NUIT
810 DATA 3,3	
820 DATA "VOUS BUVEZ MODEREMENT (1/2 LITRE 4 VERRS DE BIERE)"	DE VIN OU
830 DATA 0,"VOUS NE BUVEZ PAS"	
840 DATA -8,"VOUS BUVEZ BEAUCOUP"	
850 DATA 1,3,"VOUS PRENEZ DE L'EXERCICE TROIS AR SEMAINE"	FOIS P
860 DATA 3,3,"VOUS ETES EX-UNIVERSITAIRE"	
870 DATA 2,"VOUS ETES BACHELIER"	
880 DATA 0,"NI L'UN NI L'AUTRE"	
890 DATA 2,1,"VOUS ETES CITADIN"	
900 DATA -1,"VOUS ETES CAMPAGNARD"	
910 DATA 5,1,"VOUS ETES MARIE OU VIVEZ LEMENT"	MARITA
920 DATA -9,"VOUS ETES SOLITAIRE"	
930 DATA -7,"VOUS ETES VEUF"	
940 DATA -4,"FEMME SEPARÉE OU DIVORCÉE"	
950 DATA -3,"FEMME VEUVE"	

LE PROGRAMME

Lignes 30 à 80 : initialisation

L'espérance de vie initiale (variable E) est calculée suivant une courbe statistique pour des âges allant de 15 à 80 ans. Pour suivre le plus possible cette courbe, on effectue un découpage en trois tranches : de 15 à 40 ans, de 41 à 59 ans et de 60 à 80 ans.

Lignes 200 à 400 : interprétation des questions

Toutes les questions sont placées dans les DATAs des lignes 600 à 950. La première donne le nombre de réponses possibles (variable N), et les N DATAs qui suivent les questions et le nombre d'années à retrancher ou à ajouter.

Dans le cas d'une seule question (réponse par OUI ou NON), le tableau A est pris en considération uniquement si la réponse est OUI.

Lignes 500 à 540 : résultats

Ces lignes affichent à l'écran votre espérance de vie (variable E).

Si vous obtenez un nombre d'années inférieur à votre âge réel, vous avez sûrement menti quelque part !

LISTE DES VARIABLES :

A\$	dernière touche appuyée
AGE	âge actuel de l'utilisateur
A()	années à ajouter ou à retrancher suivant la réponse aux questions
E	espérance de vie
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
N	nombre de réponses possibles
Q\$()	tableau des questions
R	réponse de l'utilisateur (numérique)
R\$	réponse de l'utilisateur (alpha-numérique)

BIORYTHME


Ce programme permet la représentation graphique du biorythme de chacun pour un mois donné.

Les courbes obtenues dessinent les trois principaux cycles qui influencent la vie humaine, avec leurs périodes positives et négatives : le cycle physique de 23 jours, le cycle émotionnel de 28 jours et le cycle intellectuel de 33 jours. Ces données, tout à fait sérieuses, ont fait l'objet de longues études statistiques.


Pendant les périodes positives, notre énergie se libère et nous sommes actifs. Pendant les périodes négatives, notre organisme récupère ses forces. Aux points zéro des courbes, nous vivons des jours critiques.

Ce ne sont pas forcément des jours dangereux, mais ce sont sûrement des jours où il faudra redoubler d'attention. Une grande prudence est nécessaire pendant les jours critiques doubles (deux courbes aux points zéro), et il est vivement conseillé de ne pas sortir de chez soi les jours critiques triples...

Pour calculer votre biorythme, le MSX demandera successivement le prénom, le jour, le mois et l'année de naissance, puis le mois et l'année du biorythme souhaité. Le biorythme se dessine alors sur l'écran.



BIORYTHME



```

20 REM -----
30 DIM M$(12),M(12)
40 PI=3.14159
50 FOR I=1 TO 12:READ M$(I),M(I):NEXT I
60 DATA JANVIER,31,FEVRIER,28,MARS,31,AVRIL,30,MAI,31,JUIN,30
70 DATA JUILLET,31,AOUT,31,SEPTEMBRE,30,OCTOBRE,31,NOVEMBRE,30,DECEMBRE,31
80 SCREEN 0:COLOR 4,1,1
180 REM -----
190 REM SAISIE DES DATES
200 INPUT"VOTRE PRENOM";P$
210 PRINT"JOUR, MOIS, ANNEE DE NAISSANCE"
220 INPUT J,M,A
230 PRINT"MOIS, ANNEE DU BIORYTHME"
240 INPUT V,W
250 IF W<A THEN PRINT"IMPOSSIBLE":GOTO 210
260 IF W=A AND M>=V THEN PRINT"IMPOSSIBLE":GOTO 210
280 REM -----
290 REM CALCULS
300 R=0
310 IF A/4=INT(A/4) THEN M(2)=29 ELSE M(2)=28
320 R=M(M)-J+1
330 IF A=W THEN D=V-1 ELSE D=12
340 FOR I=M+1 TO D:R=R+M(I):NEXT I
350 IF A=W THEN 410
360 FOR I=A+1 TO W-1
370 R=R+365:IF I/4=INT(I/4) THEN R=R+1
380 NEXT I
390 IF W/4=INT(W/4) THEN M(2)=29 ELSE M(2)=28
400 FOR I=1 TO V-1:R=R+M(I):NEXT I
410 PRINT R
480 REM -----
490 REM DESSIN DU BIORYTHME
500 SCREEN 2
510 OPEN "GRP:" FOR OUTPUT AS #1
520 DRAW"BM16,0":PRINT#1,LEFT$(P$,10);" NE LE";J;M$(M);A
530 DRAW"BM16,8":PRINT#1,"BIORYTHME POUR ";M$(V);W
540 DRAW"BM16,18":PRINT#1,"^"
550 LINE (18,23)-(18,180)
560 LINE (18,100)-(235,100)
570 FOR J=0 TO (M(V)-1)*7
580 P=100-INT(70*SIN((R+(J/7))*2*PI/23))
590 E=100-INT(70*SIN((R+(J/7))*2*PI/28))
600 I=100-INT(70*SIN((R+(J/7))*2*PI/33))
610 PSET(J+24,P),6:PSET(J+24,E),4:PSET(J+24,I),15
620 IF J>=M(V) THEN 650
630 IF J>9 THEN DRAW"BM"+STR$(7*J+22)+",176":PRINT#1,MID$(STR$(J+1),2,1)
640 DRAW"BM"+STR$(7*J+22)+",184":PRINT#1,RIGHT$(STR$(J+1),1)
650 NEXT J
660 DRAW"BM240,32":PRINT#1,"+":DRAW"BM240,160":PRINT#1,"-"
670 COLOR 6:DRAW"BM240,88":PRINT#1,"P"
680 COLOR 4:DRAW"BM240,96":PRINT#1,"E"
690 COLOR 15:DRAW"BM240,104":PRINT#1,"I"
700 A$=INPUT$(1)
710 RUN

```

LE PROGRAMME

Lignes 30 à 80 : initialisation

Après dimensionnement des tableaux M\$() et M(), on y place le nom et le nombre de jours de chaque mois que l'on lit dans les DATAs des lignes 60 et 70.

Lignes 200 à 260 : saisie des dates

On saisit dans ces lignes tous les renseignements nécessaires à l'élaboration du biorythme.

Lignes 300 à 410 : calcul

On calcule le nombre de jours vécus par l'intéressé, en tenant compte des années bissextiles évidemment.

Lignes 500 à 710 : édition

Avec les renseignements collectés ci-dessus, on peut éditer le biorythme. Après les axes, on dessine point par point les trois courbes en calculant leurs ordonnées par une fonction sinus. L'abscisse correspond aux numéros des jours concernés. Le fichier ouvert à la ligne 510 permet d'afficher des caractères directement sur l'écran haute résolution.

LISTE DES VARIABLES :

A	année de naissance
A\$	dernière touche appuyée
D	nombre de mois restants dans l'année de naissance
E	ordonnée du point "émotionnel"
I	ordonnée du point "intellectuel"
J	jour de naissance
M	mois de naissance
P	ordonnée du point "physique"
P\$	prénom
PI	valeur de PI
R	nombre de jours vécus
V	mois du biorythme
W	année du biorythme

JOUR DE LA SEMAINE

Etes-vous curieux de savoir si vous êtes né un jeudi ou un dimanche ?

Voilà un petit programme qui vous donnera, outre le jour de la semaine de votre date de naissance, le nombre de jours, d'heures et de minutes que vous avez déjà vécus (nombre parfois impressionnant !).

VOUS ETES NE(E) LE :

DIMANCHE 4 JANVIER 1959

VOUS AVEZ VECU :

9425 JOURS

226200 HEURES

13572000 MINUTES

814320000 SECONDES

JOUR DE LA SEMAINE



```

10 REM JOUR DE LA SEMAINE
20 REM -----
30 DIM M$(12),M(12)
40 FOR I=1 TO 12:READ M$(I),M(I):NEXT I
50 DATA JANVIER,31,FEVRIER,28,MARS,31,AVRIL,30,MAI,31,JUIN,30
60 DATA JUILLET,31,AOUT,31,SEPTEMBRE,30,OCTOBRE,31,NOVEMBRE,30,DECEMBRE,31
70 SCREEN 0:COLOR 6,1,1:KEY OFF
80 FOR I=1 TO 7:READ J$(I):NEXT I
90 DATA LUNDI,MARDI,MERCREDI,JEUDI,VENDREDI,SAMEDI,DIMANCHE
180 REM -----
190 REM SAISIE DES DATES
200 PRINT"JOUR, MOIS, ANNEE DE NAISSANCE"
210 INPUT J,M,A
220 PRINT"JOUR, MOIS, ANNEE ACTUELS"
230 INPUT U,V,W
240 IF A=W AND V<=M THEN PRINT"IMPOSSIBLE":GOTO 200
250 IF A>99 OR W>99 THEN 200
280 REM -----
290 REM CALCULS DU NB DE JOURS
300 R=0
310 IF A/4=INT(A/4) THEN M(2)=29 ELSE M(2)=28
320 R=M(M)-J+1
330 IF A=W THEN D=V-1 ELSE D=12
340 IF M+1>D THEN 360
350 FOR I=M+1 TO D:R=R+M(I):NEXT I
360 IF A=W THEN 440
370 IF A+1=W THEN 410
380 FOR I=A+1 TO W-1
390 R=R+365:IF I/4=INT(I/4) THEN R=R+1
400 NEXT I
410 IF W/4=INT(W/4) THEN M(2)=29 ELSE M(2)=28
420 IF V=1 THEN 440
430 FOR I=1 TO V-1:R=R+M(I):NEXT I
440 R=R+U-1
480 REM -----
490 REM CALCUL DU JOUR DE LA SEMAINE
500 S=(A*365)+INT((A-1)/4)
510 IF A=0 THEN S=0
520 IF A/4=INT(A/4) THEN M(2)=29 ELSE M(2)=28
530 IF M=1 THEN 550
540 FOR I=1 TO M-1:S=S+M(I):NEXT I
550 S=S+J-1
560 Y=S-(7*INT(S/7))+1
580 REM -----
590 REM RESULTATS
600 CLS
610 PRINT "VOUS ETES NE(E) LE ":PRINT
620 PRINT J$(Y);J;M$(M);1900+A
630 PRINT:PRINT:PRINT
640 PRINT"VOUS AVEZ VECU ":PRINT
650 PRINT USING"#####";R;:PRINT" JOURS"
660 PRINT USING"#####";R*24;:PRINT" HEURES"
670 PRINT USING"#####";R*1440;:PRINT" MINUTES"
680 PRINT USING"#####";R*86400;:PRINT" SECONDES"
690 PRINT:PRINT

```

LE PROGRAMME

Lignes 30 à 90 : initialisation

La ligne 40 remplit les tableaux M\$() et M() avec le nom et le nombre de jours par mois avec les DATAs des lignes 50 et 60.

La ligne 80 remplit, elle, le tableau J\$() avec le nom des jours de la semaine.

Lignes 200 à 250 : saisie des dates

Les lignes 200 à 230 saisissent successivement le jour, le mois et l'année de naissance, puis le jour, le mois et l'année actuelle.

En cas d'erreur, la ligne 240 affiche le message "IMPOSSIBLE".

Lignes 300 à 440 : calcul du nombre de jours

On calcule d'abord le nombre de jours entre la date de naissance et le 31 décembre de cette même année. Puis le nombre de jours entre l'année de naissance + 1 et l'année actuelle - 1. Pour terminer on ajoute le nombre de jours entre le 1^{er} janvier de l'année actuelle et la date désirée.

Lignes 500 à 560 : calcul du jour de la semaine

On calcule pour cela le nombre de jours entre le 1^{er} janvier 1900 et la date de naissance.

Lignes 600 à 690 : résultats

Les lignes 650 à 680 affichent sur l'écran les résultats des calculs précédents :

- La date de naissance
- Le nombre de jours vécus
- Le nombre d'heures (on multiplie par 24 le nombre de jours)
- Le nombre de minutes (on multiplie par 1440)
- Le nombre de secondes (on multiplie par 86400)

On utilise dans ces lignes l'instruction PRINT USING pour cadrer les chiffres sur l'écran.

LISTE DES VARIABLES :

A	année de naissance
I	indice de boucle FOR/NEXT
J	jour de naissance
J\$()	noms des jours de la semaine
M	mois de naissance
M\$()	noms des mois
M()	nombre de jours par mois
R	nombre de jours vécus
S	nombre de jours entre le 1/1/1900 et la date de naissance
U	jour actuel
V	mois actuel
W	année actuelle
Y	jour de la semaine (exemple : Y=1 → lundi)

CALENDRIER

Comme son titre l'indique, ce programme permet d'obtenir sur l'écran un calendrier complet d'une année du vingtième siècle.

En effet, après avoir tapé RUN, l'ordinateur demande l'année du calendrier que vous souhaitez.


Cette année doit être rentrée sous la forme de ses deux derniers chiffres. Exemple : entrez 84 pour 1984 ; cela limite les demandes de 00 à 99, c'est-à-dire des années 1900 à 1999.


L'ordinateur affiche le calendrier mois par mois dès que vous appuyez sur une touche. Il précise, à gauche et à droite de chaque mois, le nombre de jours passés et à passer.

De quoi jeter un coup d'œil discret vers 1990 ou 1999 !

181 JUILLET 85 184						
L	M	M	J	V	S	D
1	2	3	4	5	6	7
8	9	10	11	12	13	14
15	16	17	18	19	20	21
22	23	24	25	26	27	28
29	30	31				

CALENDRIER





```

10 REM CALENDRIER
20 REM -----
30 DIM M$(12),N(12)
40 DATA JANVIER,31,FEVRIER,28,MARS,31
50 DATA AVRIL,30,MAI,31,JUIN,30,JUILLET,31,AOUT,31
60 DATA SEPTEMB.,30,OCTOBRE,31,NOVEMBRE,30,DECEMBRE,31
70 FOR I=1 TO 12
80 READ M$(I),N(I)
90 NEXT I
180 REM -----
190 REM CALCUL JOUR
200 J=0:B=0
210 SCREEN 0:COLOR 6,1,1:KEY OFF
220 INPUT "ANNEE ";U
230 IF U>99 THEN 210
240 M=(U*365)+INT((U-1)/4)
250 IF U=0 THEN M=0
260 Y=M-(7*INT(M/7))+1
270 IF INT(U/4)=U/4 THEN B=1
280 REM -----
290 REM EDITION
300 FOR I=1 TO 12
310 NJ=N(I)
320 IF I=2 AND B=1 THEN NJ=29
330 CLS:PRINT
340 PRINT TAB(8);J;M$(I);U;365+B-J
350 PRINT:PRINT
360 PRINT TAB(9);"L M M J V S D"
370 PRINT:PRINT:PRINT
380 FOR K=1 TO NJ
390 K$=RIGHT$(STR$(K),2)
400 PRINT TAB((Y-1)*3+8);K$;
410 Y=Y+1
420 IF Y=8 THEN Y=1:PRINT:PRINT
430 NEXT K
440 J=J+NJ
450 A$=INPUT$(1)
460 NEXT I
470 GOTO 200

```

LE PROGRAMME

Lignes 30 à 90 : initialisation

Après dimensionnement des tableaux M\$ et N, on remplit ceux-ci avec la liste des noms des mois de l'année et le nombre de jours par mois.

Lignes 200 à 270 : calcul du jour de la semaine

On saisit l'année du calendrier dans la variable U. Cette variable est alors utilisée pour connaître le numéro du jour de la semaine du 1^{er} janvier de cette année (lundi=1, mardi=2, etc.).

En ligne 270, on regarde s'il s'agit d'une année bissextile ; si c'est le cas, on met la variable B à 1.

Lignes 300 à 470 : édition du calendrier

Pour chaque mois de l'année, on extrait du tableau N le nombre de jours par mois (variable NJ).

S'il s'agit du mois de février (I=2), on vérifie le contenu de la variable B (année bissextile). Si c'est le cas, on place 29 dans la variable NJ au lieu de 28 comme il est indiqué dans le tableau N().

La ligne 340 affiche le nombre de jours écoulés (variable J), le nom du mois (tableau M\$), l'année étudiée (variable U) et le nombre de jours restants ($365 + B - J$).

La variable Y donnant la position du jour suivant, on affiche le numéro de ce jour en utilisant l'instruction TAB(Y).

Quand $Y=8$ (le 8^e jour de la semaine !), on remet cette variable à 1, et ainsi de suite.

En ligne 440, on incrémente le nombre de jours écoulés du nombre de jours que l'on vient d'afficher.

LISTE DES VARIABLES :


A\$	dernière touche enfoncée
B	B=1 si année bissextile
I	indice de boucle FOR/NEXT
J	nombre de jours écoulés
K	indice de boucle FOR/NEXT pour l'édition des jours
K\$	valeur de K sous forme de chaîne de caractères
M	nombre de jours écoulés depuis le 1 ^{er} janvier 1900
M\$()	liste des mois
N()	nombre de jours par mois
NJ	nombre de jours du mois étudié
U	année à étudier
Y	numéro du jour de la semaine (lundi=1, mardi=2, ...)

HORLOGE


Un nouveau rôle pour votre MSX. Vous pourrez l'installer sur un meuble de votre salon ou pourquoi pas dans votre cuisine !

Entrez l'heure au clavier en laissant un peu d'avance. Dès que le dessin est terminé, appuyez sur n'importe quelle touche pour mettre la pendule en route.

Le compteur interne utilisé est d'une très grande précision. Il permet d'obtenir une variation inférieure à une seconde par jour.



HORLOGE



```

10 REM HORLOGE
20 REM -----
30 SCREEN 0:COLOR 15,1,1:KEY OFF
40 INPUT"H,M,S";H,M,S
50 H=H*5+INT(M/12)
60 DIM X(62),Y(62)
70 SCREEN 2
80 OPEN "GRP:" FOR OUTPUT AS #1
90 GOSUB 500
100 A$=INPUT$(1)
180 REM -----
190 REM MOUVEMENT DES AIGUILLES
200 TIME=0
210 A=TIME
220 IF A<B THEN B=B-65536!
230 IF B+50<A THEN B=B+50:S=S+1 ELSE 210
240 IF S<60 THEN 330
250 S=0:M=M+1:IF M<60 THEN 270
260 M=0
270 IF INT(M/12)<>M/12 THEN 320
280 H=H+1
290 IF H<60 THEN 310
300 H=0
310 LINE (140,100)-(140-30*X(H+1),100+30*Y(H+1)),1
320 LINE (140,100)-(140-40*X(M+1),100+40*Y(M+1)),1
330 LINE (140,100)-(140-48*X(S+1),100+48*Y(S+1)),1
340 LINE (140,100)-(140-30*X(H+2),100+30*Y(H+2))
350 LINE (140,100)-(140-40*X(M+2),100+40*Y(M+2))
360 LINE (140,100)-(140-48*X(S+2),100+48*Y(S+2))
370 GOTO 210
480 REM -----
490 REM DESSIN DE L'HORLOGE
500 CLS
510 CIRCLE(140,100),80,6
520 FOR I=-3.1415 TO 3.1515 STEP .104717
530 PSET(140-50*SIN(I),100+50*COS(I)),5
540 X(Z+2)=SIN(I):Y(Z+2)=COS(I)
550 IF INT(Z/5)<>Z/5 THEN 610
560 LINE -(140-55*SIN(I),100+55*COS(I)),5
570 IF Z=0 THEN 610
580 IF Z/5=3 OR Z/5=9 THEN DRAW"BM"+STR$(INT(140-65*SIN(I)))
+",""+STR$(INT((100+65*COS(I))-6)):PRINT#1,RIGHT$(STR$(Z/5),1)
):GOTO 610
590 DRAW "BM"+STR$(INT(140-65*SIN(I)))+",""+STR$(INT(100+65*
COS(I))):PRINT#1,RIGHT$(STR$(Z/5),1)
600 IF Z/5>9 THEN DRAW"BM"+STR$(INT(140-65*SIN(I))-6)+",""+ST
R$(INT(100+65*COS(I))):PRINT#1,MID$(STR$(Z/5),2,1)
610 Z=Z+1
620 NEXT I
630 X(1)=X(61):Y(1)=Y(61)
640 RETURN

```

LE PROGRAMME

Lignes 30 à 100 : initialisation

La ligne 40 saisit l'heure (heures, minutes et secondes). Pour plus de réalité, la petite aiguille est déplacée toutes les 12 minutes et non toutes les heures. On multiplie donc la variable H par 5 (12 minutes \times 5 = 60 minutes = 1 heure). On l'incrmente aussi du nombre de fractions de 12 minutes contenues dans la variable M.

Lignes 200 à 370 : mouvement des aiguilles

La variable TIME est incrémentée par le Basic 50 fois par seconde. Il suffit d'incrémenter le compteur de secondes S dès que cette mémoire augmente de 50.

Les lignes 250 à 300 modifient le contenu de M quand $S=60$, puis H toutes les 12 minutes.

Les lignes 310 à 360 affichent les trois aiguilles par LINE grâce aux deux tableaux X() et Y() définis en 500.

Lignes 500 à 670 : dessin de l'horloge

La ligne 510 dessine un cercle sur l'écran, tandis que les lignes 550 à 650 affichent les heures et des tirets pour les secondes. Les coordonnées de ces derniers sont conservées dans les tableaux X() et Y() ce qui facilitera l'affichage des aiguilles.

LISTE DES VARIABLES :

A	dernière valeur de TIME
AS	dernière touche appuyée
B	ancienne valeur de A
H	compteur des heures
I	indice de boucle FOR/NEXT
M	compteur des minutes
S	compteur des secondes
X() et Y()	gardent les coordonnées des aiguilles sur l'écran
Z	compteur des minutes dans l'affichage. On affiche les heures seulement toutes les 5 minutes.

AGENDA ELECTRONIQUE


Pour ne plus oublier de rendez-vous, ou consulter d'un seul coup le planning de votre semaine, utilisez donc ce programme d'agenda.

Il est composé de trois parties :

- **SAISIE** : permet de saisir la date, l'heure et les commentaires sur un rendez-vous ou un fait précis.
- **RECHERCHE** : en donnant une date de début et de fin, le programme affiche tous les enregistrements compris dans cet intervalle.
- **SAUVEGARDE** : sauve le fichier sur cassette ou sur disquette.

DATE	HH	COMMENTAIRES
9/ 9	13	RDV SA DUPONT
20/ 9	8	PENSER A PAYER LE GAZ
1/10	16	PASSER A LA BANQUE
20/10	12	DEJEUNER AVEC LE PATRON
22/10	15	RDV SA DURANT
3/11	16	PRENDRE LES ENFANTS A L'E COLE
15/11	20	DINER AVEC MICHEL
2/12	14	RDV SA DUPONT

AGENDA ELECTRONIQUE



```

10 REM AGENDA ELECTRONIQUE
20 REM -----
30 CLEAR 5000: DIM D(1000), H(1000), C$(1000), J(12)
40 E$=CHR$(5)
50 FOR I=1 TO 12: READ J(I): NEXT I
60 DATA 31,29,31,30,31,30
70 DATA 31,31,30,31,30,31
80 SCREEN 0
90 INPUT "PREMIERE UTILISATION (O/N)"; R$
100 IF R$="O" THEN GOTO 200
110 CLS: PRINT "METTEZ LE MAGNETO K7 EN LECTURE"
120 OPEN "AGENDA" FOR INPUT AS #1
130 IF EOF(1) THEN 170
140 N=N+1
150 INPUT #1, D(N), H(N): LINE INPUT #1, C$(N)
160 GOTO 130
170 CLOSE
180 REM -----
190 REM MENU
200 CLS: PRINT TAB(18) "MENU"
210 PRINT TAB(13) "1 SAISIE 1": PRINT TAB(13) "2 RECHERCHES
220 PRINT TAB(13) "3 SAUVEGARDE 3"
230 PRINT: PRINT TAB(13) "VOTRE CHOIX ?";
240 R$=INPUT$(1): R=VAL(R$)
250 IF R<1 OR R>3 THEN 230
260 ON R GOTO 300,400,800
270 REM -----
280 REM SAISIE
290 CLS: PRINT: PRINT TAB(10) "SAISIE D'UN RENDEZ-VOUS": PRINT
300 GOSUB 700
310 N=N+1
320 LOCATE 10,14: PRINT E$; "HEURE : ";: LINE INPUT H$: H(N)=IN
T(VAL(H$))
330 IF H(N)<0 OR H(N)>24 THEN 330
340 LOCATE 0,16: LINE INPUT "COMMENTAIRES: "; C$(N)
350 D(N)=M*100+J
360 GOTO 200
370 REM -----
380 REM RECHERCHES
390 CLS: PRINT: PRINT TAB(15) "RECHERCHES"
400 PRINT: PRINT TAB(14) "DATE DE DEBUT"
410 GOSUB 700
420 DD=M*100+J
430 CLS: PRINT TAB(15) "RECHERCHES"
440 PRINT: PRINT TAB(14) "DATE DE FIN"
450 GOSUB 700
460 DF=M*100+J
470 CLS
480 PRINT " DATE HH COMMENTAIRES"
490 PRINT "-----"
500 FOR I=1 TO N
510 IF D(I)<DD OR D(I)>DF THEN 600
520 M=INT((D(I))/100)
530 J=D(I)-M*100
540 PRINT USING "##/"; J;
550 PRINT USING "## "; M, H(I);
560 FOR J=0 TO LEN(C$(I))/25
570 PRINT TAB(9); MID$(C$(I), 1+J*25, 25)
580 NEXT J
590 NEXT I
600 R$=INPUT$(1)
610 GOTO 200
620 REM -----

```

```

690 REM SAISIE D'UNE DATE
700 LOCATE 10,10:PRINT E$;"JOUR   : ";:LINE INPUT J$:J=VAL(J$)
710 IF J>31 OR J<1 THEN 700
720 LOCATE 10,12:PRINT E$;"MOIS   : ";:LINE INPUT M$:M=VAL(M$)
730 IF M>12 OR M<1 THEN 720
740 IF J>J(M) THEN 700
750 RETURN
760 REM -----
770 REM SAUVEGARDE
780 CLS:PRINT"APPUYEZ SUR RETURN QUAND LE MAGNETO K7 EST PR
ET POUR L'ENREGISTREMENT"
790 LINE INPUT R$
800 OPEN "AGENDA" FOR OUTPUT AS #1
810 FOR I=1 TO N
820 PRINT#1,D(I),H(I),C$(I)
830 NEXT I
840 CLOSE
850 CLS:PRINT"A LA PROCHAINE !"
860 END

```

LE PROGRAMME

Lignes 30 à 170 : initialisation

On place dans le tableau J() le nombre de jours par mois.

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 200 à 250 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 300 à 370 : saisie

Pour saisir un rendez-vous, on fait appel au sous-programme en 700 (saisie des dates).

Les lignes 330 et 350 demandent à l'utilisateur l'heure et les commentaires éventuels sur ce rendez-vous.

Lignes 400 à 620 : recherches

Le programme demande les dates de début et de fin pour les recherches. On utilise aussi pour cela le sous-programme de saisie des dates en 700.

Les lignes 510 à 600 parcourent par une boucle FOR/NEXT l'intégralité du fichier, et affichent les enregistrements compris entre la date de début et la date de fin.

Lignes 700 à 750 : saisie d'une date

Ces lignes saisissent au clavier le jour et le mois. Les lignes 710, 730 et 740 vérifient les éventuelles erreurs.

Lignes 800 à 880 : sauvegarde

Ces lignes sauvent sur cassette ou sur disquette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

ES	caractère d'effacement d'une ligne (code ASCII 5)
C\$()	liste des commentaires
D()	liste des dates
DD	date de début des recherches
DF	date de fin des recherches
H()	liste des heures
H\$	heure du rendez-vous
I	indice de boucle FOR/NEXT
J	jour du rendez-vous (numérique)
J()	nombre de jours par mois
J\$	jour du rendez-vous
M	heure du rendez-vous (numérique)
M\$	heure du rendez-vous
N	nombre d'enregistrements dans le fichier
R\$	choix de l'utilisateur
R	valeur numérique de R\$

The following information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972. The information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972. The information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972.

The following information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972. The information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972.

Summary of Findings

The following information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972. The information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972.

The following information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972. The information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972.

The following information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972. The information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972.

The following information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972. The information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972.

The following information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972. The information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972.

The following information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972. The information was obtained from the records of the Department of the Interior, Bureau of Land Management, for the year 1972.



THEME 6 VOTRE FORME

POIDS IDEAL

Ce programme ne vous donnera pas votre poids idéal sur le plan esthétique, car cela dépend surtout de l'image physique que l'on veut présenter. Mais il calcule le poids idéal dont se servent les assureurs-vie pour juger de la longévité éventuelle de leurs clients selon le rapport poids-taille.

Donnez votre âge, votre taille et votre carrure : le MSX se charge des calculs et vous révélera le poids le plus favorable à votre santé.


VOTRE AGE ? 25
TAILLE (EN CMS) ? 176
CARRURE (ETROITE=1, MOY.=2, LARGE=3)? 2
SEXE FEMININ (O/N) ? N

VOTRE POIDS IDEAL EST DE:

71 KGS

A + OU - 2 KGS


POIDS IDEAL



```

10 REM POIDS IDEAL
20 REM -----
30 SCREEN 0
40 INPUT "VOTRE AGE "; A
50 IF A < 0 OR A > 100 THEN 40
60 INPUT "TAILLE (EN CMS) "; T
70 IF T < 0 OR T > 250 THEN 60
80 INPUT "CARRURE (ETROITE=1, MOY.=2, LARGE=3) "; C
90 IF C < 1 OR C > 3 THEN 80
100 INPUT "SEXE FEMININ (O/N) "; S$
110 IF S$ <> "O" AND S$ <> "N" THEN 100
120 IF S$ = "O" THEN S = .8 ELSE S = .9
180 REM -----
190 REM CALCULS
200 CLS
210 PRINT TAB(8); "VOTRE POIDS IDEAL EST DE: "
220 PRINT:PRINT:PRINT
230 PRINT TAB(15); INT((T-100+C+(A/20))*S); "KGS"
240 PRINT:PRINT
250 PRINT TAB(12); "A + OU -"; C; "KGS"

```



LE PROGRAMME

Lignes 30 à 120 : saisie des éléments

Le programme saisit successivement l'âge, la taille, la carrure et le sexe du sujet.

Les lignes 50, 70, 90 et 110 vérifient que les réponses soient vraisemblables.

Lignes 200 à 250 : calculs

Ces lignes calculent la fourchette de votre poids idéal en appliquant la formule :

$$\text{taille} - 100 + \text{carrure} + \text{âge}/20$$

le tout multiplié par 0.8 pour les femmes et 0.9 pour les hommes.

La fourchette est donnée par la carrure (entre 1 et 3 kilogrammes).

LISTE DES VARIABLES :

A	âge
C	carrure
SS	sexe féminin (oui ou non)
S	multiplicateur suivant le sexe
T	taille en cm


COURBES POIDS-TAILLE

Destiné aux jeunes parents, ce programme permet de tracer sur l'écran les courbes du poids et de la taille d'un enfant de moins de deux ans.


Le MSX traduit simplement sous forme graphique les renseignements portés sur le carnet de santé remis aux heureux parents le jour de la naissance.

Il saisit chaque semaine le poids et la taille, ou le poids et la taille à la naissance s'il s'agit de la première utilisation.

Après le tracé des courbes, le programme demande si vous désirez entrer une autre semaine avant la sauvegarde du fichier sur cassette. Vous pouvez ainsi entrer plusieurs semaines, voire plusieurs mois, en une seule fois.



COURBES POIDS/TAILLE



```

10 REM COURBES POIDS/TAILLE
20 REM -----
30 DIM P(104),T(104)
40 SCREEN 0
50 INPUT"PREMIERE UTILISATION (O/N)";R$
60 IF R$="O" THEN GOTO 140
70 CLS:PRINT"METTEZ LE MAGNETO K7 EN LECTURE"
80 OPEN "POIDS" FOR INPUT AS #1
90 FOR I=0 TO 104
100 INPUT#1,P(I),T(I)
110 IF S=0 AND P(I)=0 THEN S=I
120 NEXT I
130 CLOSE
140 OPEN "GRP:" FOR OUTPUT AS #1
180 REM -----
190 REM DONNEES DE LA SEMAINE
200 SCREEN 0
210 IF S=0 THEN PRINT"ENTREZ LES DONNEES A LA NAISSANCE":PRINT:GOTO 250
220 PRINT"ENTREZ LES DONNEES DE LA";S;
230 IF S=1 THEN PRINT "ere" ELSE PRINT"eme"
240 PRINT" SEMAINE":PRINT
250 INPUT"POIDS (EN KILOGRAMMES) ";P(S)
260 IF P(S)>13 THEN GOTO 250
270 INPUT"TAILLE (EN CENTIMETRES) ";T(S)
280 IF T(S)>130 THEN GOTO 270
290 IF S=0 THEN S=1:GOTO 200
380 REM -----
390 REM COURBES DE LA SEMAINE
400 SCREEN 2:DRAW"BM16,0":PRINT#1,"TAILLE/POIDS":DRAW"BM24,8":PRINT#1,"CMS KG"
410 DRAW "BM89,22":PRINT#1,"^":DRAW"BM252,129":PRINT#1,">"
420 LINE (91,23)-(91,132):LINE -(312,132)
430 FOR I=0 TO 13
440 DRAW"BM24,"+STR$(8*I+24):PRINT#1,USING"### ";130-I*10,13-I
450 LINE (92,I*8+28)-(88,I*8+28)
460 NEXT I
470 FOR I=1 TO 26
480 A$=RIGHT$(" "+STR$(I*4),3)
490 DRAW"BM"+STR$(6*I+90)+",144":PRINT#1,LEFT$(A$,1)
500 DRAW"BM"+STR$(6*I+90)+",152":PRINT#1,MID$(A$,2,1)
510 DRAW"BM"+STR$(6*I+90)+",160":PRINT#1,RIGHT$(A$,1)
520 LINE (I*6+92,131)-(I*6+92,135)
530 NEXT I
540 DRAW"BM190,172":PRINT#1,"SEMAINES"
550 PSET(91,132-P(0)*8)
560 FOR I=1 TO S
570 LINE -(91+I*1.5,132-P(I)*8)
580 NEXT I
590 DRAW"BM"+STR$(INT(94+I*1.5))+", "+STR$(INT(128-P(S)*8)):PRINT#1,"P"
600 PSET(91,132-(T(0)/10)*8)
610 FOR I=1 TO S
620 LINE -(91+I*1.5,132-(T(I)/10)*8)
630 NEXT I
640 DRAW"BM"+STR$(INT(94+I*1.5))+", "+STR$(INT(128-(T(S)/10)*8)):PRINT#1,"T"
650 DRAW"BM24,184":PRINT#1,"UNE AUTRE SEMAINE (O/N) ":R$=INPUT$(1)
660 IF R$="O" THEN S=S+1:GOTO 200
680 REM -----

```

```

690 REM SAUVEGARDE
700 SCREEN 0:PRINT"APPUYEZ SUR RETURN QUAND LE MAGNETO K7 E
ST PRET POUR L'ENREGISTREMENT"
710 LINE INPUT R$
720 CLOSE:OPEN "POIDS" FOR OUTPUT AS #1
730 FOR I=0 TO 104
740 PRINT#1,P(I),T(I)
750 NEXT I
760 CLOSE
770 CLS:PRINT"A LA SEMAINE PROCHAINE !"
780 END

```

LE PROGRAMME

Lignes 30 à 140 : initialisation

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

La ligne 140 ouvre un fichier dans le but d'écrire du texte sur l'écran haute résolution (SCREEN 2).

Lignes 200 à 290 : données de la semaine

Le programme demande le poids et la taille de la semaine S. S'il s'agit de la première utilisation ($S=0$), la ligne 210 affiche le message "données à la naissance".

Lignes 400 à 660 : courbes de la semaine

Ces lignes dessinent sur l'écran les courbes du poids et de la taille de l'enfant de la naissance à la semaine en cours.

Les lignes 400 à 460 affichent l'axe des ordonnées, les lignes 470 à 530 les valeurs des poids et des tailles, et les lignes 520 à 620 tracent les deux courbes.


La ligne 660 renvoie à la saisie si l'on désire entrer les données de la semaine suivante avant la sauvegarde.

Lignes 700 à 780 : sauvegarde

Ces lignes sauvent sur cassette ou sur disquette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

AS	poids/tailles à afficher (graduations de l'ordonnée)
I	indice de boucle FOR/NEXT
P()	liste des poids
RS	réponse de l'utilisateur
S	numéro de la semaine (taille du fichier)
T()	liste des tailles



THEME 7 BRICOLAGE

PAPIER PEINT

Ce programme permet de calculer en fonction de certaines caractéristiques le nombre de rouleaux à acheter pour pouvoir tapisser une pièce ainsi que le rendement et le coût de l'opération.

Au départ, l'ordinateur vous demande le prix du rouleau, sa largeur et sa longueur, puis le périmètre et la hauteur de la pièce.

Ces mesures sont à entrer en mètres pour les calculs de l'ordinateur, après lesquels il vous précise le nombre de rouleaux à acheter, le nombre de lés que vous pouvez découper dans chaque rouleau, la longueur des chutes pour les raccords éventuels et le reste inutilisé de papier.


De même il vous donne le coût total de votre achat et vous chiffre les pertes de papier.

Vous pouvez garder le MSX à côté de votre colle et de vos pinceaux : il sera lui aussi un outil de travail précieux.

PAPIER PEINT

NOMBRE DE ROULEAUX: 6
NOMBRE DE LÉS PAR ROULEAU: 4
IL Y A 5 CHUTES DE .8 METRES
PLUS 5.4 METRES SUR LE DERNIER
ROULEAU
IL RESTE DONC 9.4 METRES DE CHUTE
POUR LES RACCORDS
LE PRIX TOTAL EST DE 336 FRANCS
ET LES PERTES DE 15 %
SOIT 52 FRANCS

PAPIER PEINT



```

10 REM PAPIER PEINT
20 REM -----
30 SCREEN 0:COLOR 6,1,1:KEY OFF
40 PRINT TAB(15)"PAPIER PEINT"
50 PRINT:PRINT:PRINT
60 INPUT"PRIX DU ROULEAU";PR
70 IF PR<1 THEN PRINT"C'EST PAS CHER. DONNEZ MOI L'ADRESSE!"
:GOTO 60
80 INPUT"LARGEUR DU PAPIER";LP
90 IF LP<.1 THEN PRINT"C'EST DU PAPIER PEINT OU DU PAPIER
TOILETTE ?":GOTO 80
100 INPUT"LONGUEUR DES ROULEAUX";LR
110 IF LR<1 THEN PRINT"VOUS VOUS ETES FAIT ROULER !":GOTO 10
0
120 INPUT"PERIMETRE DE LA PIECE";CP
130 IF CP>200 THEN PRINT"VOUS HABITEZ VERSAILLES JE SUPPOSE
?":GOTO 120
140 IF CP<1 THEN PRINT"VOUS TAPISSEZ VOTRE POULAILLER ?":GOT
0 120
150 INPUT"HAUTEUR DES MURS";HM
160 IF HM<1 THEN PRINT"C'EST BAS DE PLAFOND CHEZ VOUS !":GOT
0 150
170 IF INT(LR/HM)=0 THEN PRINT"IMPOSSIBLE! LES MURS SONT TRO
P HAUTS":AS=INPUT$(1):RUN
280 REM -----
290 REM CALCULS
300 NL=INT(LR/HM)
310 LC=LR-NL*HM
320 NR=INT((CP/(NL*LP))+.99999)
330 NTL=INT((CP/LP)+.99999)
340 ND=NTL-((NR-1)*NL)
350 LD=LR-(ND*HM)
360 PT=NR*PR
370 TC=((NR-1)*LC)+LD
380 R=TC/((NR*LR)/100)
480 REM -----
490 REM RESULTATS
500 CLS
510 PRINT TAB(13)"PAPIER PEINT"
520 PRINT:PRINT:PRINT
530 PRINT"NOMBRE DE ROULEAUX: ";NR
540 PRINT"NOMBRE DE LES PAR ROULEAU: ";NL
550 PRINT"IL Y A ";NR-1; "CHUTES DE ";LC; "METRES"
560 PRINT"PLUS ";LD; "METRES SUR LE DERNIER":PRINT"ROULEAU"
570 PRINT"IL RESTE DONC ";TC; "METRES DE CHUTE":PRINT"POUR LES
RACCORDS"
580 PRINT"LE PRIX TOTAL EST DE ";PT; "FRANCS"
590 PRINT"ET LES PERTES DE ";INT(R); "%"
600 PRINT"SOIT ";INT(PT*R/100); "FRANCS"
610 AS=INPUT$(1)
620 RUN

```

LE PROGRAMME

Lignes 30 à 170 : saisie des paramètres

On saisit à la suite le prix, la largeur et la longueur d'un rouleau, puis le périmètre et la hauteur des murs de la pièce à tapisser.

Lignes 300 à 380 : calculs

Cette partie du programme calcule toutes les dimensions des lés, des chutes et la composition du dernier rouleau.

La valeur 0.99999 ajoute un rouleau de plus dans le cas où le dernier est incomplet.

Lignes 500 à 620 : résultats

Toutes les données calculées précédemment sont affichées à l'écran, avec en plus le prix des pertes sur les chutes et le dernier rouleau s'il est incomplet.

LISTE DES VARIABLES :

AS	dernière touche appuyée
CP	périmètre de la pièce
HP	hauteur des murs
LC	longueur de la chute par rouleau
LD	longueur de la chute sur le dernier rouleau
LP	largeur du papier
LR	longueur des rouleaux
ND	nombre de lés sur le dernier rouleau
NL	nombre de lés par rouleau
NR	nombre de rouleaux
NTL	nombre total de lés
PR	prix du rouleau
PT	prix total du papier peint
R	rendement
TC	longueur totale des chutes

DISPOSITION DES MEUBLES

Le mieux pour changer votre appartement est encore d'en modifier la disposition des meubles.

Le programme est composé de deux parties :

- **LE GENERATEUR DE SPRITES** : vous devez dessiner avec cet utilitaire la forme des meubles en utilisant un joystick. Le bouton de ce dernier permet de changer la couleur d'un point élémentaire.

Touches de fonction :

F1: efface complètement le dessin d'un sprite.



F2: change le numéro du sprite en cours.

F3: donne accès au deuxième programme (disposition des meubles).

F4: reporte dans la mémoire du MSX le dessin du sprite en cours.

- **DISPOSITION DES SPRITES SUR L'ECRAN** : le joystick permet de déplacer le sprite en cours dans toutes les directions et sur toute la surface de l'écran. Comme pour le programme de définition, la touche F2 permet de changer le numéro du sprite en cours et la touche F3 de passer à l'autre partie (générateur de sprites).

DISPOSITION DES MEUBLES



```

10 REM DISPOSITION DES MEUBLES
20 REM -----
30 CLEAR 10000
40 DEFINT A-Z
50 DIM S$(16,16),XX(16),YY(16),X(16),Y(16)
60 COLOR 5,1,1:SCREEN 2,3
70 OPEN "GRP:" FOR OUTPUT AS #1
80 KEY 3,"A":KEY OFF
90 Z$(1)="EFFAC LUTIN DISPO CALC."
100 Z$(2)="      LUTIN GENE."
110 FOR I=1 TO 16
120 YY(I)=I*17:IF YY(I)>136 THEN YY(I)=YY(I)-136
130 IF I>8 THEN XX(I)=196 ELSE XX(I)=146
140 X(I)=XX(I):Y(I)=YY(I)
150 FOR J=1 TO 16
160 S$(I,J)=STRING$(16,48)
170 NEXT J
180 NEXT I
190 DATA F8,90,A0,D0,88,04,02,01
200 FOR I=1 TO 8:READ A$:Z$=Z$+CHR$(VAL("&H"+A$)):NEXT I
210 SPRITE$(0)=Z$
220 N=1
230 Z=1:GOSUB 1600:GOSUB 1900:GOSUB 950
240 X=8:Y=8
380 REM -----
390 REM DEPLACEMENTS (GENERATEUR)
400 KEY(1) ON:KEY(2) ON:KEY(3) OFF:KEY(4) ON
410 PUT SPRITE 0,(8+X*8,8+Y*8),6,0
420 PUT SPRITE N,(100,145),8,N
430 D=STICK(1):IF D<>0 THEN 460
440 A$=INKEY$:IF A$="A" THEN 1300
450 GOTO 500
460 IF D=7 THEN X=X-1:IF X=0 THEN X=16
470 IF D=3 THEN X=X+1:IF X=17 THEN X=1
480 IF D=1 THEN Y=Y-1:IF Y=0 THEN Y=16
490 IF D=5 THEN Y=Y+1:IF Y=17 THEN Y=1
500 FOR I=1 TO 200:NEXT
510 D=STRIG(1):IF D=0 THEN 400
520 IF POINT(9+X*8,9+Y*8)=1 THEN 550
530 LINE(9+X*8,9+Y*8)-(15+X*8,15+Y*8),1,BF
540 MID$(S$(N,Y),X)="0":GOTO 400
550 PAINT(10+X*8,10+Y*8),5
560 MID$(S$(N,Y),X)="1":GOTO 400
680 REM -----
690 REM CALCUL
700 Z$=""
710 FOR I=1 TO 16 STEP 8
720 FOR J=1 TO 16
730 Z$=Z$+CHR$(VAL("&B"+MID$(S$(N,J),I,8)))
740 NEXT J,I
750 SPRITE$(N)=Z$
760 PUT SPRITE N,(XX(N),YY(N)),5,N
770 RETURN
880 REM -----
890 REM No DU LUTIN (GENERATEUR)
900 A$=INPUT$(2):IF VAL(A$)<1 OR VAL(A$)>16 THEN RETURN
910 IF VAL(A$)=N THEN RETURN
920 PUT SPRITE N,(XX(N),YY(N)),5,N
930 N=VAL(A$)
940 COLOR 1:DRAW"BM74,170":PRINT #1,CHR$(219);CHR$(219);CHR$(219)
950 COLOR 5:DRAW"BM74,170":PRINT #1,N
960 FOR I=1 TO 16
970 FOR J=1 TO 16

```

```

980 IF MID$(S$(N,I),J,1)="1" THEN 1000
990 LINE (9+J*8,9+I*8)-(15+J*8,15+I*8),1,BF:GOTO 1010
1000 LINE (9+J*8,9+I*8)-(15+J*8,15+I*8),5,BF
1010 NEXT J,I:RETURN
1080 REM -----
1090 REM EFFACE
1100 FOR I=1 TO 16
1110 S$(N,I)=STRING$(16,48)
1120 NEXT I
1130 GOTO 960
1140 GOTO 1140
1150 FOR I=1 TO 16:PRINT XX(I),YY(I):NEXT
1280 REM -----
1290 REM DEPLACEMENTS (DISPOSITION)
1300 Z=2:GOSUB 1600:GOSUB 2140
1310 PUT SPRITE 0,(0,0),0,0
1320 FOR I=1 TO 16
1330 PUT SPRITE I,(X(I),Y(I)),5,I
1340 NEXT I
1350 KEY(2) ON
1360 D=STICK(1):IF D<>0 THEN 1380
1370 A$=INKEY$:IF A$="A" THEN 230
1380 IF D=7 THEN X(N)=X(N)-1:IF X(N)=0 THEN X(N)=220
1390 IF D=3 THEN X(N)=X(N)+1:IF X(N)=220 THEN X(N)=1
1400 IF D=1 THEN Y(N)=Y(N)-1:IF Y(N)=0 THEN Y(N)=190
1410 IF D=5 THEN Y(N)=Y(N)+1:IF Y(N)=190 THEN Y(N)=1
1420 PUT SPRITE N,(X(N),Y(N)),5,N
1430 GOTO 1350
1440 Z=1:GOSUB 1600:GOSUB 1900:GOSUB 950:RETURN
1580 REM -----
1590 REM TOUCHES DE FONCTION
1600 CLS
1610 DRAW"BM18,182":PRINT#1,Z$(Z)
1620 FOR I=0 TO 4
1630 LINE (16+I*48,180)-(56+I*48,190),5,B
1640 NEXT I
1650 IF Z=1 THEN ON KEY GOSUB 1100,900,,700,700
1660 IF Z=2 THEN ON KEY GOSUB ,2100
1670 RETURN
1880 REM -----
1890 REM DESSIN GENERATEUR
1900 FOR I=0 TO 16
1910 LINE (16,16+I*8)-(144,16+I*8)
1920 LINE (16+I*8,16)-(16+I*8,144)
1930 IF I>0 THEN PUT SPRITE I,(XX(I),YY(I)),5,I
1940 IF I>0 THEN DRAW"BM"+STR$(XX(I)+30)+"", "+STR$(YY(I)+7):P
RINT#1,I
1950 NEXT I
1960 RETURN
2080 REM -----
2090 REM No DU LUTIN (DISPOSITION)
2100 A$=INPUT$(2):IF VAL(A$)<1 OR VAL(A$)>16 THEN RETURN
2110 IF VAL(A$)=N THEN RETURN
2120 N=VAL(A$)
2130 COLOR 1:DRAW"BM74,170":PRINT #1,CHR$(219);CHR$(219);CHR
$(219)
2140 COLOR 5:DRAW"BM74,170":PRINT #1,N
2150 RETURN

```

LE PROGRAMME

Lignes 30 à 240 : initialisation

La ligne 30 réserve des octets en mémoire pour le stockage des sprites (tableau S\$). On place dans le tableau Z\$ les noms des fonctions pour le générateur et la disposition des meubles. Les variables X et Y donnent les coordonnées du point de départ, N le numéro du sprite en cours et XX() et YY() la position des sprites dans le générateur (X() et Y() pour la disposition).

Lignes 400 à 560 : déplacement dans le générateur de sprites

Cette partie du programme teste le joystick et déplace le curseur sur l'écran en laissant derrière lui une trace de couleur noire si le bouton (testé par STRIG) est enfoncé.

Lignes 700 à 770 : calcul

Le programme calcule à partir du tableau S\$ la définition du sprite en cours.

Lignes 900 à 1010 : numéro du lutin en cours

Le programme saisit au clavier le nouveau numéro du sprite. Les lignes 960 à 1010 affichent à l'écran les anciennes valeurs.

Lignes 1100 à 1150 : efface

Ces lignes effacent le dessin du sprite sur l'écran et dans le tableau S\$.

Lignes 1300 à 1440 : disposition des sprites

On déplace sur l'écran le sprite en cours en testant la direction du joystick. Les coordonnées des seize objets sont conservées dans les tableaux X() et Y().

Lignes 1600 à 1670 : touches de fonction

On affiche en bas de l'écran les touches de fonction contenues dans le tableau Z\$.

Lignes 1900 à 1960 : dessin du générateur


Affiche avec une boucle FOR/NEXT le dessin du générateur. La ligne 1930 place les sprites aux coordonnées XX et YY, et la ligne 1940 indique leurs numéros.


Lignes 2100 à 2150 : numéro du lutin (disposition)

Identique aux lignes 900 à 1010, mais n'affiche pas les points du générateur.

LISTE DES VARIABLES :

AS	dernière touche appuyée
D	direction du joystick
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
N	numéro du sprite en cours
SS()	dessin binaire des sprites
X et Y	coordonnées du curseur
X() et Y()	coordonnées des sprites dans le générateur
XX() et YY()	coordonnées des sprites dans la disposition des meubles
Z	Z=1 générateur, Z=2 disposition
ZS	valeurs hexadécimales d'un sprite
ZS()	noms des touches de fonctions






THEME 8 UTILITAIRES

CRYPTOGRAMMES

Si vous aimez les messages secrets, les informations codées, les encres sympathiques, bref, si vous êtes un cryptologue averti, vous pourrez donner libre cours à votre passion en échangeant des textes indécodables... sauf pour celui qui en connaît la clef !


En effet, le MSX vous demande s'il s'agit d'un message à coder ou à décoder, puis le mot de passe (la clef du message). Vous devez taper un message quelconque, par exemple "IF THEN ELSE" ou "DE NORBERT A MARCEL". Sans cette clef il n'est pas possible de décoder un message.

Yrf xojqcrvozsse !



MESSAGE CODE:
MRNHZMEICORNMESEHH
ZVEQNAHPNEGZXGEBHCW
ZFHGEGAMOE

CRYPTOGRAMMES



```

10 REM CRYPTOGRAMMES
20 REM -----
30 DIM C(26)
40 SCREEN 0:COLOR 4,0,0:KEY OFF
50 INPUT"CODAGE OU DECODAGE (C/D) ";R$
60 IF R$="D" THEN 400
70 IF R$<>"C" THEN 40
180 REM -----
190 REM CODAGE
200 GOSUB 600
210 CLS
220 PRINT"MESSAGE CODE: "
230 FOR I=1 TO LEN(M$)
240 IF MID$(M$,I,1)<"A" OR MID$(M$,I,1)>"Z" THEN 260
250 PRINT CHR$(C(ASC(MID$(M$,I,1))-64)+64);
260 NEXT I
270 R$=INPUT$(1)
280 RUN
380 REM -----
390 REM DECODAGE
400 GOSUB 600
410 CLS
420 PRINT"MESSAGE DECODE: "
430 FOR I=1 TO LEN(M$)
440 IF MID$(M$,I,1)<"A" OR MID$(M$,I,1)>"Z" THEN 460
450 FOR J=1 TO 26
460 IF ASC(MID$(M$,I,1))-64=C(J) THEN PRINT CHR$(J+64);:GOTO 480
470 NEXT J
480 NEXT I
490 R$=INPUT$(1)
500 RUN
580 REM -----
590 REM CREATION CODES
600 CLS
610 INPUT"MOT DE PASSE";P$
620 IF P$="" THEN 600
630 D=1
640 FOR I=1 TO LEN(P$)
650 IF MID$(P$,I,1)<"A" OR MID$(P$,I,1)>"Z" THEN 690
660 IF C(ASC(MID$(P$,I,1))-64)<>0 THEN 690
670 C(ASC(MID$(P$,I,1))-64)=D
680 D=D+1
690 NEXT I
700 FOR I=26 TO 1 STEP -1
710 IF C(I)<>0 THEN 740
720 C(I)=D
730 D=D+1
740 NEXT I
750 CLS
760 INPUT"MESSAGE ";M$
770 IF M$="" THEN 750
780 RETURN

```

LE PROGRAMME

Lignes 30 à 70 : initialisation

La ligne 50 saisit le type d'opération à effectuer, codage ou décodage. Les lignes 60 et 70 envoient aux programmes concernés.

Lignes 200 à 280 : codage

Après l'appel du sous-programme de saisie du mot de passe, du message et de la création du tableau de transcription, on code une à une toutes les lettres du message M\$ à l'aide d'une boucle FOR/NEXT. Seules les lettres sont réellement codées, les autres signes sont ignorés.

Lignes 400 à 500 : décodage

Après l'appel du sous-programme de saisie du mot de passe, du message et de la création du tableau de transcription, on décode une à une toutes les lettres du message M\$ à l'aide d'une boucle FOR/NEXT.

Lignes 600 à 780 : création des codes

Dès que le mot de passe est saisi, on place dans le tableau C() la correspondance de chaque lettre (exemple : si C(1)=3, la lettre A devient C; si C(26)=5, la lettre Z devient E) en fonction du mot de passe P\$.

LISTE DES VARIABLES :

C()	tableau de transcription
D	nombre de lettres déjà dans le tableau C()
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
M\$	message à coder ou à décoder
P\$	mot de passe
R\$	choix de l'utilisateur

GESTION DES QSO

Votre MSX est aussi radio-amateur (il sait décidément tout faire, votre ordinateur !).

Si vous avez déjà utilisé ce programme, vous ne devez pas répondre à la question "Première utilisation" (tapez uniquement sur la touche "RETURN"). Il affiche alors un menu de six options :

- 1 – **SAISIE D'UN QRZ** : vous devez entrer le QRZ et le QTH d'un de vos correspondants.

Si vous n'avez plus rien à rentrer, appuyez sur "RETURN" à la question "QRZ:".

- 2 – **SAISIE D'UN QSO** : si vous désirez introduire dans le fichier un nouveau QSO, entrez le numéro du QRZ correspondant (si vous avez oublié ce numéro, utilisez l'option 4 ou 5). Le programme saisit alors la date de la liaison (on peut mettre 4 QSO par enregistrement). Vous devez taper "Ø" en réponse à la question "Numéro du QRZ:" pour revenir au menu.

- 3 – **SUPPRESSION** : en donnant le numéro d'un QRZ, le programme supprime l'enregistrement correspondant dans le fichier.


- 4 – **RECHERCHE** : elle permet d'afficher un enregistrement complet à partir du renseignement que vous lui donnez. Par exemple le mot "Cachan" pour obtenir tous les Cibistes de votre connaissance habitant la ville de Cachan.

- 5 – **LISTING** : édite simplement sur l'écran le fichier dans son intégralité.

- 6 – **SAUVEGARDE** : le programme sauve le fichier sur cassette ou sur disquette.

N.B. : le code Q utilisé est celui couramment employé par les Cibistes et différent des normes en vigueur.

GESTION DES QSO



```

10 REM GESTION DES QSO
20 REM -----
30 CLEAR 3000
40 DIM D$(6,100)
50 L$(1)="QRZ":L$(2)="QTH":L$(3)="QSO DU"
60 SCREEN 0
70 INPUT"PREMIERE UTILISATION (O/N)";R$
80 IF R$="O" THEN 200
90 CLS:PRINT"METTEZ LE MAGNETO K7 EN LECTURE"
100 OPEN "QSO" FOR INPUT AS #1
110 INPUT#1,N
120 FOR I=1 TO N
130 FOR J=1 TO 6
140 LINE INPUT#1,D$(J,I)
150 NEXT J,I
160 CLOSE
180 REM -----
190 REM MENU
200 CLS
210 PRINT"NB DE QRZ: ";N:PRINT:PRINT
220 PRINT"1 SAISIE QRZ"
230 PRINT"2 SAISIE QSO"
240 PRINT"3 SUPPRESSION"
250 PRINT"4 RECHERCHE"
260 PRINT"5 LISTING"
270 PRINT"6 SAUVEGARDE"
280 PRINT:PRINT"VOTRE CHOIX ?";
290 R$=INPUT$(1):R=VAL(R$)
300 IF R<1 OR R>6 THEN 290
310 ON R GOTO 400,800,1000,600,1200,1400
380 REM -----
390 REM SAISIE D'UN QRZ
400 CLS:PRINT"SAISIE D'UN QRZ":PRINT
410 IF N=100 THEN PRINT"FICHIER PLEIN";:R$=INPUT$(1):GOTO 20
0
420 N=N+1
430 FOR I=1 TO 2
440 PRINT L$(I);": ";:LINE INPUT D$(I,N)
450 IF I=1 AND D$(1,N)="" THEN N=N-1:GOTO 200
460 NEXT I
470 GOTO 400
580 REM -----
590 REM RECHERCHE
600 CLS:F=0
610 PRINT "DONNEE A RECHERCHER: ";:LINE INPUT D$:PRINT
620 IF D$="" THEN 200
630 FOR I=1 TO N
640 FOR J=1 TO 6
650 IF INSTR(D$(J,I),D$)<>0 THEN 670
660 NEXT J:GOTO 690
670 PRINT USING"###";I;:FOR K=1 TO 2:PRINT " ";LEFT$(D$(K,I)+
" ",10);:NEXT K:PRINT
680 PRINT TAB(2);:FOR K=3 TO 6:PRINT LEFT$(D$(K,I)+
" ",8); " ";:NEXT K:PRINT:F=1
690 NEXT I:PRINT
700 IF F=0 THEN PRINT"JE N'AI RIEN TROUVE"
710 R$=INPUT$(1):GOTO 600
780 REM -----
790 REM SAISIE DES QSO
800 CLS:PRINT"SAISIE DES QSO":PRINT
810 NU=0:INPUT"NUMERO DU QRZ";NU
820 IF NU<1 OR NU>N THEN 200

```

```

830 FOR I=3 TO 6
840 IF D$(I,NU)="" THEN 870
850 NEXT I
860 PRINT:PRINT"IL N'Y A PLUS DE PLACE";:A$=INPUT$(1):GOTO 8
80
870 PRINT L$(3);":":LINE INPUT D$(I,NU)
880 GOTO 800
980 REM -----
990 REM SUPPRESSION
1000 CLS:PRINT"SUPPRESSION":PRINT
1010 NU=0:INPUT"NUMERO DU QRZ";NU
1020 IF NU<1 OR NU>N THEN 200
1030 PRINT
1040 FOR I=1 TO 3
1050 PRINT I;L$(I);":":D$(I,NU)
1060 NEXT I:PRINT
1070 R$="":INPUT"VOUS ETES SUR DE SUPPRIMER CE QRZ";R$
1080 IF R$<>"O" THEN 1000
1090 N=N-1
1100 FOR I=NU TO N
1110 FOR J=1 TO 6
1120 D$(J,I)=D$(J,I+1)
1130 NEXT J,I
1140 GOTO 1000
1180 REM -----
1190 REM LISTING
1200 CLS:F=0
1210 FOR I=1 TO N
1220 PRINT USING"###";I;:FOR K=1 TO 2:PRINT " ";LEFT$(D$(K,I)
+"",10);:NEXT K:PRINT
1230 PRINT TAB(2);:FOR K=3 TO 6:PRINT LEFT$(D$(K,I)+
",8);":NEXT K:PRINT:F=F+3
1240 IF F>20 THEN PRINT:PRINT"APPUYEZ SUR UNE TOUCHE POUR LA
SUITE";:R$=INPUT$(1):CLS:F=0
1250 NEXT I
1260 PRINT:PRINT"APPUYEZ SUR UNE TOUCHE POUR LE MENU";
1270 R$=INPUT$(1):GOTO 200
1380 REM -----
1390 REM SAUVEGARDE
1400 CLS:PRINT"APPUYEZ SUR RETURN QUAND LE MAGNETO K7 EST P
RET POUR L'ENREGISTREMENT"
1410 LINE INPUT R$
1420 OPEN "QSO" FOR OUTPUT AS #1
1430 PRINT#1,N
1440 FOR I=1 TO N
1450 FOR J=1 TO 6
1460 PRINT#1,D$(J,I)
1470 NEXT J,I
1480 CLOSE
1490 CLS:PRINT"A LA PROCHAINE !"
1500 END

```

Lignes 1400 à 1500 : sauvegarde

Ces lignes sauvent sur cassette ou sur disquette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

D\$	donnée à rechercher
D\$(,)	tableau des données du fichier
F	drapeau. Si F=0, aucun enregistrement n'a été trouvé.
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
K	indice de boucle FOR/NEXT
L\$()	libellés des données
N	nombre de QRZ dans le fichier
NU	numéro d'enregistrement choisi par l'utilisateur
R\$	choix de l'utilisateur
R	valeur numérique de R\$

1 ASPIRINE PARIS X
10/01/84 05/06/84 15/10/84

2 BACCARA BAGNEUX
12/02/84

3 DRACULA PARIS
05/05/84

4 ALTOS CACHAN
31/01/84 06/09/84

5 GAI-LURON PARIS XI
06/06/84 07/06/84 08/06/84

6 ICARE CACHAN
10/09/84 11/12/83 16/04/84 09/10/84

7 SIDERAL BICETRE
16/11/83

APPUYEZ SUR UNE TOUCHE POUR LA SUITE

LE PROGRAMME

Lignes 30 à 160 : initialisation

On place dans le tableau L\$() les libellés des trois données nécessaires à l'exploitation du fichier.

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 200 à 310 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 400 à 470 : saisie d'un QRZ

On saisit par une boucle FOR/NEXT le QRZ et le QTH. Si le nom de la station est une chaîne vide, on revient au menu (ligne 450).

Lignes 600 à 710 : recherche

Le programme demande d'abord la donnée sur laquelle on désire effectuer la recherche (variable D\$). Puis les lignes 630 à 690 parcourent par une boucle FOR/NEXT l'intégralité du fichier, et affichent les enregistrements où l'on a trouvé la chaîne D\$.

Lignes 800 à 880 : saisie d'un QSO

Ces lignes saisissent la date d'un QSO dans la limite de quatre.

Lignes 1000 à 1140 : suppression

On supprime du fichier l'enregistrement choisi par l'utilisateur (variable NU). Après l'avoir affiché, le programme demande confirmation. Les lignes 1100 à 1130 décalent le tableau vers le bas.

Lignes 1200 à 1270 : listing

On affiche par une boucle FOR/NEXT l'intégralité du fichier. Le programme est stoppé par la ligne 1240 dès que l'écran est plein.

GESTION D'UNE CAVE

Vous désirez un sommelier particulier ? En voici un tout désigné : votre MSX.

Après la rituelle question "Première utilisation", le programme affiche un menu pour les options suivantes :

- 1 – **SAISIE** : vous devez entrer tous les renseignements sur un cru que vous avez dans votre cave. Les dates "Avant" et "Après" correspondent aux deux années (ou à une date complète si vous le désirez) entre lesquelles le vin sera le meilleur.

Si vous ne voulez plus entrer de crus, tapez uniquement sur "RETURN" à la question "CRU:".

- 2 – **MODIFICATION** : si vous désirez modifier un cru, entrez son numéro (si vous avez oublié ce numéro, utilisez l'option 4 ou 5).

Le programme vous demande alors le type de donnée que vous voulez modifier. Si vous n'avez plus de modification à ajouter, taper "Ø" en réponse à la question "Numéro du cru:".


- 3 – **SUPPRESSION** : si vous avez malheureusement épuisé toutes les bouteilles d'un cru particulier, cette option permet de le supprimer du fichier. Comme pour la modification, taper "Ø" pour revenir au menu.

- 4 – **RECHERCHE** : c'est la partie la plus intéressante du programme. Elle permet d'afficher tous les crus à consommer entre deux dates fixes, de connaître toutes les bouteilles de Bordeaux ou de champagne qui vous attendent impatiemment dans votre cave ou celles qui vous restent de l'année 76. Il n'est pas nécessaire de donner un nom entier : le mot "BEAU" suffit à retrouver les crus d'appellation "BEAUjolais" ou les bouteilles de St-Emilion Château "BEAUséjour".

- 5 – **LISTING** : édite simplement sur l'écran le fichier dans son intégralité.

- 6 – **SAUVEGARDE** : le programme sauve le fichier sur cassette ou sur disquette.

GESTION D'UNE CAVE



```

10 REM GESTION D'UNE CAVE
20 REM -----
30 CLEAR 5000
40 DIM D$(6,100)
50 L$(1)="CRU":L$(2)="APPELLATION":L$(3)="MILLESIME":L$(4)="
NB DE BOUTEILLES":L$(5)="MEILLEUR APRES":L$(6)="MEILLEUR AVA
NT"
60 SCREEN 0:COLOR 6,1,1:KEY OFF
70 INPUT"PREMIERE UTILISATION (O/N)";R$
80 IF R$="O" THEN 200
90 CLS:PRINT"METTEZ LE MAGNETO K7 EN LECTURE"
100 OPEN "CAVE" FOR INPUT AS #1
110 INPUT#1,N
120 FOR I=1 TO N
130 FOR J=1 TO 6
140 LINE INPUT#1,D$(J,I)
150 NEXT J,I
160 CLOSE
180 REM -----
190 REM MENU
200 CLS
210 PRINT"NB DE CRUS:";N:PRINT:PRINT
220 PRINT"1 SAISIE"
230 PRINT"2 MODIFICATION"
240 PRINT"3 SUPPRESSION"
250 PRINT"4 RECHERCHE"
260 PRINT"5 LISTING"
270 PRINT"6 SAUVEGARDE"
280 PRINT:PRINT"VOTRE CHOIX ?";
290 R$=INPUT$(1):R=VAL(R$)
300 IF R<1 OR R>6 THEN 290
310 ON R GOTO 400,900,1100,600,1300,1500
380 REM -----
390 REM SAISIE
400 CLS:PRINT"SAISIE D'UN CRU":PRINT
410 IF N=100 THEN PRINT"FICHER PLEIN";:R$=INPUT$(1):GOTO 20
0
420 N=N+1
430 FOR I=1 TO 6
440 PRINT L$(I);":":LINE INPUT D$(I,N)
450 IF I=1 AND D$(1,N)="" THEN N=N-1:GOTO 200
460 NEXT I
470 GOTO 400
580 REM -----
590 REM RECHERCHE
600 CLS:F=0
610 PRINT"RECHERCHE PAR : "
620 FOR I=1 TO 6:PRINT I;L$(I):NEXT
630 PRINT:PRINT"QUEL TYPE DE DONNEE ?":PRINT
640 R$="":R$=INPUT$(1):R=VAL(R$)
650 IF R<1 OR R>6 THEN 200
660 IF R<5 THEN 700
670 INPUT"DATE DEBUT ";DD
680 INPUT"DATE FIN ";DF
690 GOTO 710
700 D$="":PRINT L$(R);":":LINE INPUT D$:PRINT
710 PRINT
720 FOR I=1 TO N
730 IF R<5 THEN IF INSTR(D$(R,I),D$)=0 THEN 790 ELSE 750
740 IF VAL(D$(R,I))<DD OR VAL(D$(R,I))>DF THEN 790
750 F=1:PRINT USING"### ";I;
760 PRINT D$(1,I);" (";D$(2,I);")"
770 PRINT TAB(4);"MILLESIME:";D$(3,I);TAB(20);"NB DE B.:";D$
(4,I)

```

```

780 PRINT TAB(4); "    APRES: "; D$(5,1); TAB(23); "AVANT: "; D$(6,
I):PRINT
790 NEXT I:PRINT
800 IF F=0 THEN PRINT"JE N'AI RIEN TROUVE"
810 R$=INPUT$(1):GOTO 600
880 REM -----
890 REM MODIFICATION
900 CLS:PRINT"MODIFICATION":PRINT
910 NU=0:INPUT"NUMERO DU CRU";NU
920 IF NU<1 OR NU>N THEN 200
930 FOR I=1 TO 6
940 PRINT I;L$(I); " ";D$(I,NU)
950 NEXT I
960 PRINT:INPUT"MODIFICATION DU";R
970 IF R<1 OR R>6 THEN 200
980 PRINT L$(R); " ";:LINE INPUT D$(R,NU)
990 GOTO 900
1080 REM -----
1090 REM SUPPRESSION
1100 CLS:PRINT"SUPPRESSION":PRINT
1110 NU=0:INPUT"NUMERO DU CRU";NU
1120 IF NU<1 OR NU>N THEN 200
1130 PRINT
1140 FOR I=1 TO 6
1150 PRINT I;L$(I); " ";D$(I,NU)
1160 NEXT I:PRINT
1170 R$="":INPUT"VOUS ETES SUR DE SUPPRIMER CE CRU";R$
1180 IF R$<>"O" THEN 1100
1190 N=N-1
1200 FOR I=NU TO N
1210 FOR J=1 TO 6
1220 D$(J,I)=D$(J,I+1)
1230 NEXT J,I
1240 GOTO 1100
1280 REM -----
1290 REM LISTING
1300 CLS:F=0
1310 FOR I=1 TO N
1320 F=F+4:PRINT USING"### ";I;
1330 PRINT D$(1,I); " (";D$(2,I); ")"
1340 PRINT TAB(4); "MILLESIME: ";D$(3,I); TAB(20); "NB DE B. ";D
$(4,I)
1350 PRINT TAB(4); "    APRES: ";D$(5,I); TAB(23); "AVANT: ";D$(6
,I):PRINT
1360 IF F>16 THEN PRINT:PRINT"APPUYEZ SUR UNE TOUCHE POUR LA
SUITE";R$=INPUT$(1):CLS:F=0
1370 NEXT I
1380 PRINT:PRINT"APPUYEZ SUR UNE TOUCHE POUR LE MENU";
1390 R$=INPUT$(1):GOTO 200
1480 REM -----
1490 REM SAUVEGARDE
1500 CLS:PRINT"APPUYEZ SUR RETURN QUAND LE MAGNETO K7 EST P
RET POUR L'ENREGISTREMENT"
1510 LINE INPUT R$
1520 OPEN "CAVE" FOR OUTPUT AS #1
1530 PRINT#1,N
1540 FOR I=1 TO N
1550 FOR J=1 TO 6
1560 PRINT#1,D$(J,I)
1570 NEXT J,I
1580 CLOSE
1590 CLS:PRINT"A LA PROCHAINE BOUTEILLE !"
1600 END

```

LE PROGRAMME

Lignes 30 à 160 : initialisation

On place dans le tableau L\$() les libellés des six données nécessaires à l'exploitation du fichier.

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 200 à 310 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur.

Lignes 400 à 470 : saisie d'un cru

On saisit par une boucle FOR/NEXT les six données du fichier. Si aucun cru n'a été entré, on revient au menu (ligne 450).

Lignes 600 à 810 : recherche

Le programme demande d'abord sur quel type de donnée on désire effectuer la recherche, puis la donnée elle-même (variable D\$ pour les quatre premières, dates de début DD et de fin DF pour une période).

Les lignes 720 à 790 parcourent par une boucle FOR/NEXT l'intégralité du fichier, et affichent les enregistrements où l'on a trouvé la chaîne D\$ ou comprise entre les deux dates DD et DF.

Lignes 900 à 990 : modification

Ces lignes modifient l'enregistrement choisi par l'utilisateur (variable NU).

Lignes 1100 à 1240 : suppression

On supprime du fichier l'enregistrement choisi par l'utilisateur (variable NU). Après l'avoir affiché, le programme demande confirmation. Les lignes 1200 à 1230 décalent le tableau vers le bas.

Lignes 1300 à 1390 : listing

On affiche par une boucle FOR/NEXT l'intégralité du fichier. Le programme est stoppé par la ligne 1360 dès que l'écran est plein.

Lignes 1500 à 1600 : sauvegarde

Ces lignes sauvent sur cassette ou sur disquette le fichier dès qu'une touche a été appuyée.

LISTE DES VARIABLES :

DS	donnée à rechercher
DS(,)	tableau des données du fichier
DD	date de début des recherches
DF	date de fin des recherches
F	drapeau. Si F=0, aucun enregistrement n'a été trouvé.
I	indice de boucle FOR/NEXT
J	indice de boucle FOR/NEXT
LS()	libellés des données
N	nombre d'opérations dans le fichier
NU	numéro d'enregistrement choisi par l'utilisateur
RS	choix de l'utilisateur
R	valeur numérique de RS

RECHERCHE PAR :

- 1 CRU
- 2 APPELLATION
- 3 MILLESIME
- 4 NB DE BOUTEILLES
- 5 MEILLEUR APRES
- 6 MEILLEUR AVANT

QUEL TYPE DE DONNEE ?

DATE DEBUT ? 80
DATE FIN ? 82

- 1 DOM PERIGNON (CHAMPAGNE)
MILLESIME: 76 NB DE B.: 4
APRES: 82 AVANT: 86
- 2 CH. MARGAUX (BORDEAUX)
MILLESIME: 73 NB DE B.: 5
APRES: 81 AVANT: 93

REPERTOIRE TELEPHONIQUE

Pourquoi ne pas simplifier votre annuaire et vous réserver quelques numéros de téléphone ?

Ce fichier vous permettra de garder les numéros d'une quarantaine de personnes.


Le programme affiche un menu de trois options :

- 1 – **SAISIE** : avant de saisir un nom et un numéro de téléphone, le MSX affiche sur l'écran la totalité du fichier. Vous devez donner en priorité un numéro de libre pour une nouvelle saisie ou le numéro d'enregistrement que vous désirez modifier. Il faut répondre "Ø" à la question "Numéro:" pour revenir au menu.
- 2 – **RECHERCHE** : dès que vous aurez donné le nom à rechercher, cette option affiche l'intégralité du fichier et s'arrête après un bip sonore sur les noms correspondants. Il n'est pas nécessaire de donner un nom en entier. Ceci permet de retrouver toutes les personnes d'une même famille.
- 3 – **SAUVEGARDE** : sauve le fichier sur cassette ou sur disquette.



1	DUPONT	: 678.90.54
2	DURANT	: 456.89.07
3	GOMEZ	: 008.85.96
4	JOSEPH	: 456.76.03
5	LENOIR	: 709.45.23
6	LEROY	: 665.87.96
7	MARIE	: 567.89.01
8	MARTIN	: 445.83.19
9	NAPOLEON	: 567.48.97
10	PEROT	: 906.87.65
11	PREVOST	: 673.95.84
12		
13		
14		
15		
16		
17		
18		
19		
20		

AGENDA TELEPHONIQUE



```

10 REM AGENDA TELEPHONIQUE
20 REM -----
30 DIM N$(40),T$(40)
40 SCREEN 0
50 INPUT"PREMIERE UTILISATION (O/N)";R$
60 IF R$="O" THEN GOTO 200
70 CLS:PRINT"METTEZ LE MAGNETO K7 EN LECTURE"
80 OPEN "TEL" FOR INPUT AS #1
90 FOR I=1 TO 40
100 INPUT#1,N$(I),T$(I)
110 NEXT I
120 CLOSE
180 REM -----
190 REM MENU
200 CLS
210 PRINT"1 SAISIE PERSONNE"
220 PRINT"2 RECHERCHE"
230 PRINT"3 SAUVEGARDE"
240 PRINT:PRINT" VOTRE CHOIX ? ";:V$=INPUT$(1)
250 V=VAL(V$):IF V<1 OR V>3 THEN 200
260 ON V GOTO 300,500,600
280 REM -----
290 REM SAISIE
300 GOSUB 800
310 LOCATE 0,21
320 NU$="":INPUT"NUMERO";NU$:NU=VAL(NU$)
330 IF NU=0 THEN 200
340 LOCATE 0,20:PRINT:PRINT N$(NU); " :";T$(NU); "
350 N$="":INPUT"NOM";N$
360 IF N$<>" THEN N$(NU)=N$
370 T$="":INPUT"TELEPHONE";T$
380 IF T$<>" THEN T$(NU)=T$
390 GOTO 300
480 REM -----
490 REM RECHERCHE
500 CLS
510 INPUT"NOM A RECHERCHER";NR$
520 GOSUB 800
530 GOTO 200
580 REM -----
590 REM SAUVEGARDE
600 CLS:PRINT"APPUYEZ SUR RETURN QUAND LE MAGNETO K7 EST PR
ET POUR L'ENREGISTREMENT"
610 LINE INPUT R$
620 OPEN "TEL" FOR OUTPUT AS #1
630 FOR I=1 TO 40
640 PRINT#1,N$(I),"",T$(I)
650 NEXT I
660 CLOSE
670 CLS:PRINT"A LA PROCHAINE FOIS !"
680 END
780 REM -----
790 REM AFFICHE LES NOMS
800 CLS:X=0:Y=0
810 FOR I=1 TO 40
820 LOCATE X,Y
830 IF N$(I)="" THEN PRINT I:GOTO 860
840 IF V=1 THEN PRINT I;LEFT$(N$(I),6); " ";LEFT$(T$(I),10) E
LSE PRINT I;N$(I); " :";T$(I)
850 IF V=2 AND LEFT$(N$(I),LEN(NR$))=NR$ THEN PLAY"DO":A$=IN
PUT$(1)
860 Y=Y+1:IF Y=20 AND V=1 THEN Y=0:X=20
870 IF Y=20 THEN A$=INPUT$(1):CLS:Y=0
880 NEXT I
890 RETURN

```

LE PROGRAMME

Lignes 30 à 120 : initialisation

Les tableaux N\$() et T\$() correspondent aux noms et numéros de téléphone du fichier. Si vous trouvez que ce n'est pas assez, il est possible de modifier leurs indices maximum (ici 40), par une autre valeur supérieure. Il faut pour cela modifier les chiffres 40 aux lignes 30, 90, 630 et 810.

Si ce n'est pas la première utilisation, on lit sur la cassette ou sur la disquette le fichier déjà créé antérieurement.

Lignes 200 à 250 : menu

Ces lignes ventilent sur le reste du programme suivant le choix de l'utilisateur (variable V).

Lignes 300 à 390 : saisie

On fait d'abord appel au sous-programme en 800 (affiche les noms).

Puis le programme demande le numéro que l'on veut créer ou modifier. Si on répond 0, la ligne 330 renvoie au menu. Dans le cas contraire, les lignes 350 et 370 saisissent le nom et le numéro de téléphone.

Lignes 500 à 530 : recherches

Le programme place dans NR\$ le nom que l'on veut rechercher. Le sous-programme en 800 affiche alors tous les noms correspondants.

Lignes 600 à 680 : sauvegarde

Ces lignes sauvent sur cassette ou sur disquette le fichier dès qu'une touche a été appuyée.

Lignes 800 à 890 : affichent les noms

Ce sous-programme affiche tous les noms et numéros de téléphone du fichier pour la création (V=1), ou uniquement ceux commençant par la chaîne de caractères NR\$ pour la recherche (V=2).

LISTE DES VARIABLES :

AS	dernière touche appuyée
I	indice de boucle FOR/NEXT
NU	numéro d'enregistrement dans le fichier (numérique)
N\$	dernier nom saisi
N\$()	liste des noms
NR\$	nom à rechercher
NU\$	numéro d'enregistrement dans le fichier
RS	réponse de l'utilisateur
T\$	dernier numéro de téléphone saisi
T\$()	liste des numéros de téléphone
V	choix de l'utilisateur
X et Y	coordonnées sur l'écran du prochain nom à afficher

Achevé d'imprimer
sur les presses de l'imprimerie IBP
à Rungis (94 Val-de-Marne) 686.73.54
Dépôt légal - Décembre 1984

N° d'impression: 4828
N° d'édition: 86595-166-1
N° d'ISBN: 2-86595-166-9

Que faire d'un ordinateur chez soi?
Peut-on gérer son compte bancaire?
Prévoir le coût de sa voiture chaque année?
Apprendre à compter à ses enfants?
Calculer son signe astrologique chinois?

C'est en réponse à ces questions
que Jean-François Sehan vous propose une sélection
de quarante programmes à caractère familial.
Ce livre est composé de huit parties :

- les finances
- la pédagogie,
- la cuisine,
- les jeux nationaux,
- le temps,
- votre forme,
- le bricolage
et pour finir
- quelques utilitaires.

Dans un but pédagogique, chaque programme
est accompagné d'un organigramme,
d'une liste des variables et d'une explication
de chaque ligne Basic pour mieux
assimiler les techniques de programmation
ou les adapter à d'autres ordinateurs.



EDITIONS DU P.S.I.
BP 86

FRANCE 77402 LAGNY/MARNE CÉDEX

ISBN 2-86595-166-5
Imprimé en France